

An Introduction to Prolog

Helen Pain and Robert Dale

October 30, 2002

Contents

1	Introduction	7
2	Prolog Basics	9
2.1	What is Prolog?	9
2.2	Main concepts	9
2.3	An example Prolog program	9
2.4	Facts, Questions and Variables	11
2.4.1	Facts	11
2.4.2	Questions	11
2.4.3	Variables	12
2.5	Prolog Terminology and Syntax	13
2.6	Expressing Relationships in Prolog	15
2.7	Unifying Terms	16
2.8	Conjunctions	17
2.8.1	Asking Questions that Contain Conjunctions	17
2.8.2	Using Variables in Conjunctions	18
2.9	Summary	18
2.10	Exercises	19
3	How Prolog Works	23
3.1	Introduction	23
3.2	Running, Consulting and Editing Prolog Programs	23
3.2.1	Starting Prolog	23

3.2.2	Consulting a File	23
3.2.3	Seeing What's in the Database	24
3.2.4	Getting Out of Prolog	24
3.2.5	Writing and Modifying Programs	24
4	Rules	26
4.1	Introduction	26
4.2	Rules	26
4.2.1	Expressing Rules in Prolog	26
4.2.2	Returning to the 'Drinks' Example	28
4.2.3	Recursion	30
4.2.4	Getting multiple answers	32
4.3	Summary	34
4.4	Exercises	34
4.5	Practical 1	36
4.5.1	Getting Started	36
4.5.2	Running Prolog	36
4.5.3	Getting out of Prolog	38
4.5.4	Editing the Program	39
4.5.5	Testing the Program	39
4.5.6	More Things to Add	40
4.5.7	Harder Things	40
4.6	Practical 2: Writing a Program to Find Who is Where	42
4.6.1	Complicating the Program	43
5	Backtracking	44
5.1	Getting more than one answer	44
5.2	Representing Prolog's behaviour using trees	45
5.3	Summary	49
5.4	Exercises	51
6	Built-in System Predicates	53
6.1	Introduction	53

6.2	The Example Program	53
6.3	Asking the Database Some Questions	53
6.4	Using Rules	55
6.5	Adding Built-in or System Predicates for User Interaction	55
6.6	Looking for Multiple Responses	57
6.7	Dealing with Arithmetic Operators	58
6.8	Summary	61
6.9	Practical 3: consulting a simple database	63
6.9.1	The Basic <code>books</code> Program	63
6.9.2	The <code>books</code> Program—Simple Interactive Version	63
6.9.3	Writing your own Database	66
7	Lists	67
7.1	Introduction	67
7.2	The <code>books3</code> Program	67
7.2.1	The Example Database	67
7.2.2	The Rules	67
7.2.3	How It Works	67
7.2.4	The <code>member</code> Predicate	70
7.2.5	The <code>prlist</code> Predicate	70
7.3	Lists	70
7.3.1	Basics	70
7.4	Manipulating Lists	71
7.4.1	Matching Lists	71
7.4.2	Constructing and Destructing Lists	72
7.5	Summary	74
7.6	Exercises	74
7.7	Practical 4: Simple Sentence Generation	77
8	Manipulating lists	80
8.1	Introduction	80
8.2	The predicate <code>member/2</code>	80
8.2.1	Building the predicate	80

8.2.2	Cleaning Up	81
8.2.3	Recursion in List Processing	82
8.2.4	And/Or Trees for Recursive Predicates	82
8.3	Other List Processing Predicates	83
8.3.1	Printing a list of elements	83
8.3.2	Checking that no element of a list of letters is a consonant	84
8.4	More List Processing Predicates	85
8.4.1	Finding the maximum of a list of numbers	85
8.4.2	Building new list structures	85
8.5	Exercises	86
9	How Programs Work	91
9.1	Tracing	91
9.2	The Byrd Box Model of Execution	91
9.3	Debugging using the Tracer	95
9.4	Loading Files	97
9.5	Some common mistakes	98
9.6	Summary	99
10	Further list processing predicates	100
10.1	Changing one sentence into another: the predicate <code>alter/2</code>	100
10.2	Deleting the first occurrence of an element from a list: the predicate <code>delete/2</code>	103
10.3	Reversing a list: the predicates <code>rev/2</code> and <code>rev/3</code>	104
10.4	Joining two lists together: the predicates <code>append/3</code>	105
10.5	Exercises	106
10.6	Practical 5: List Processing	112
10.6.1	Introduction	112
10.6.2	Basic List Processing	112
10.6.3	More Advanced List Processing	114
11	Parsing in Prolog	116
11.1	Introduction	116
11.2	Simple English Syntax	116

11.3	The Parse Tree	117
11.4	Prolog Grammar Rules	117
11.5	Using the Grammar Rules	119
11.6	How to Extract a Parse Tree	120
11.7	Adding Arbitrary Prolog Goals	120
11.8	Practical 6: Definite Clause Grammars	122
11.8.1	Introduction	122
11.8.2	The Basic Grammar	122
11.8.3	Structure Building	124
11.8.4	Adding Number Agreement	124
11.8.5	Extending the Coverage of the Grammar	125
12	Input/Output	126
12.1	Basic input/output facilities	126
12.2	File input/output	129
12.3	Translating atoms and strings	132
12.4	Practical 7: Input/Output	134
12.4.1	Introduction	134
12.4.2	Basic Terminal Input/Output	134
12.4.3	File Input/Output	135
12.4.4	Atoms and Strings	136
13	Morphology: A List Processing Application	138
13.1	Summary	142
13.2	Prolog Practical 8: Morphology: An Exercise in List Processing	143
13.2.1	Introduction	143
13.2.2	Basic Operations on Atoms and Strings	143
13.2.3	Morphological Processing	143
14	Top-down design: The Missionaries and Cannibals Problem	146
14.1	The Problem	146
14.2	Viewing the Problem as State Space Search	146
14.3	A Prolog Program to Solve the Problem	147

14.4	Problem Solution	148
14.5	Where We Are So Far	151
14.6	Prolog Practical 9: The Missionaries and Cannibals Problem	153
14.6.1	Introduction	153
14.6.2	Making it Work	153
14.6.3	Adding a Commentary	153
15	Eliza in Prolog	155
15.1	Implementing Eliza	155
15.2	Some Comments on Prolog Layout	156
15.3	Prolog Practical 10: Eliza	158
16	Problem Solving in Prolog: The Monkey and the Bananas	160
16.1	The Problem	160
16.2	The General Approach to a Solution	160
16.3	Representational Considerations	160
16.4	Doing All This in Prolog	163
16.5	Using Strips in Another Domain	166
16.6	Prolog Practical 11: A Simple Version of STRIPS	167
16.6.1	Basics	167
16.6.2	Extensions	167
17	Answers	171
17.1	Chapter 2	171
17.2	Chapter 4	174
17.3	Chapter 5	176
17.4	Chapter 7	179
17.5	Chapter 8	180
17.6	Chapter 10	187

1 Introduction

What This Course is About

This course is an introduction to the programming language Prolog. By the end of the course you should be able:

- to understand and run Prolog programs;
- to edit and modify Prolog programs; and
- to write simple Prolog programs.

The course assumes no previous knowledge of programming: even if you have previous programming experience in other programming languages, you will find that Prolog is quite different from conventional programming languages.

Access to the Computer

You can access the computer from a variety of different terminals around the university (see the AI1 Systems notes). In general, however, you will be using those in the School of Informatics. Those used by the AI1 course are situated in the undergraduate terminal room (C1) on the first floor at 80 South Bridge.

Practical sessions take place at various times in the week: you will have already been assigned to one of the practical groups. Demonstrators will be available during these sessions to help you with any problems that arise.

You can also use the terminals in C1 outside of your practical session during working hours (9am–5pm), as long as no other class is having a practical session. Outside of these hours you can get access with a key and your smart card (see the course secretary in the ITO office, 80 South Bridge, to get a key).

Terminals are also available elsewhere (at George Square Library, Appleton Tower, Pollock Halls, and JCMB). Also see the Systems page for further information.

The Rest of These Notes

These notes are organized in sections which correspond to material covered in the lectures, plus some additional topics and exercises.

The contents of each of the sections are as follows:

Section 2: This section contains an introductory overview to Prolog. We describe the main concepts in Prolog, and show an example Prolog program. We also introduce the concepts of **facts**, **questions** and **variables**; the conventions embodied in Prolog syntax; Prolog relations and unification of terms, and how Prolog queries can contain **conjunctions**.

- Section 3:** We review the basic things you need to know in order to use Prolog under Unix.
- Section 4:** The bulk of the section looks at the notion of **rules** in Prolog. Recursion is also introduced in this section.
- Section 5:** This section discusses in detail the use of **backtracking** in Prolog's search strategy.
- Section 6:** This section concentrates on Prolog **built-in system predicates**. To motivate the use of these, we introduce a new example. It also describes how arithmetic operators work in Prolog.
- Section 7:** This section focusses on the use of **lists** as a basic data structure in Prolog programming. We modify the example introduced in the previous section to show how lists can be used.
- Section 8:** In this section, we look more closely at **list manipulation**, and discuss a number of useful list processing predicates.
- Section 9:** This section explains how we can find out how Prolog programs work by **tracing** them.
- Section 10:** Further **list processing predicates** are described in this section.

The material from here on is optional for AI1Ah.

- Section 11:** In this section of these notes, we show how Prolog can be used to parse sentences in a natural language such as English, using **dcg's**.
- Section 12:** Basic **input/output** facilities provided by Prolog are described in this section.
- Section 13:** Further input/output facilities are described in relation to an application to **morphology**.
- Section 14:** **Top-down** design is illustrated through a problem solving example: **Missionaries and Cannibals**.
- Section 15:** An example of how to implement a simple **Eliza** program is described.
- Section 16:** This section describes a further problem solving example: the **Monkey and the Bananas** problem.

Answers: Answers to exercises in the earlier sections are given in the final section.

As you proceed through this course, you will find that you hear a lot of jargon; you will gradually come to understand what you need.

2 Prolog Basics

2.1 What is Prolog?

Prolog is a language; more precisely, it is a **computer language** or **programming language**. Programming languages are to be distinguished from **natural languages** such as English and French; however, as with natural languages, there are many different programming languages (for example, Prolog, Basic, and Fortran), and there are different dialects of each of these.

Computer languages such as Prolog are described as **high-level languages**: they are closer to natural language than the **lower-level languages** that the computer itself uses (these are usually known as **machine code** and **assembler**).

The basic mode in which you use Prolog is highly interactive. You ask Prolog a question, this is interpreted by the computer, and a response is provided. There is a program that is responsible for interpreting Prolog, called the **Prolog interpreter**.

Prolog is one of a family of languages called **logic programming** languages, which are based on logic. A Prolog program is a series of logical statements or assertions which are evaluated by the Prolog interpreter.

2.2 Main concepts

Some of the main Prolog concepts that you will be introduced to are:

- facts;
- questions;
- logical variables;
- matching (or ‘unification’);
- conjunctions; and
- rules.

Each of these is described in more detail below.

2.3 An example Prolog program

Figure 1 shows an example of a Prolog program, along with some of the dialogue that might take place between the user and the Prolog interpreter. Alongside (on the right) comments tell you what each line ‘means’; each comment begins with a ‘%’.

```
?- listing.                % lists the program

hates(heather, whisky).    % heather hates whisky

likes(alan, coffee).       % alan likes coffee
likes(alan, whisky).       % alan likes whisky
likes(heather, gin).       % heather likes gin
likes(heather, coffee).    % heather likes coffee

drinks(alan, beer).        % alan drinks beer
drinks(heather, lager).    % heather drinks lager

yes
?- likes(alan, coffee).    % is it true that alan likes coffee?

yes
?- drinks(heather, lager). % is it true that heather drinks lager?

yes
?- drinks(alan, lager).   % is it true that alan drinks lager?

no                          % it is not true that alan drinks lager
```

Figure 1: An interaction with the Prolog interpreter




Figure 2: The structure of a fact

2.4 Facts, Questions and Variables

2.4.1 Facts

A **fact** asserts some property of an object, or states some relation between two (or more) objects; it states that something is known to be true.

A fact is made up of a **predicate** (which states the relation or property) and a number of **arguments** (which are the objects): see Figure 2.

A fact can contain any number of arguments; so, for example:

<code>drinks(alan,beer,export,lorimers).</code>	has four arguments
<code>likes(alan,coffee).</code>	has two arguments
<code>listing.</code>	has no arguments

A Prolog program may have any number of facts. When a program is given to the Prolog interpreter (loaded into Prolog, or **consulted**), the set of facts that are part of the program represent all that is known to be true. They are really a set of **logical assertions**. Together they are often referred to as the **database**. They have no real ‘meaning’, however: the person writing the program defines their own interpretation of what they mean. Whilst it might seem silly to interpret

```
drinks(beer,alan)
```

as *beer drinks alan*, there is nothing to stop us using this interpretation— it’s all the same to Prolog. What is important is to be consistent, so we have to keep arguments that are meant to refer to the same objects in the same argument slots.

2.4.2 Questions

If a question is asked by the user, the Prolog interpreter looks at the database of facts¹ that it has to see if there is enough information to answer it. Given the example program we saw in Figure 1, we might ask questions like the following:

¹The database can also contain **rules**, as we will see later.

- What does Heather drink?
- Does Alan like coffee?
- Who drinks whisky?

The Prolog interpreter doesn't understand English, however, so we have to re-express these questions in Prolog itself. Taking the second of these questions as an example, we ask:

```
?- likes(alan, coffee).
```

and the Prolog interpreter replies:

```
yes
```

The Prolog interpreter matches the question to each fact (or assertion) in the database, as follows:

1. First, Prolog finds a fact that matches the predicate in the question.
2. If this match succeeds, Prolog then matches the first argument to the predicate.
3. If this match succeeds, Prolog matches the second argument, and so on for the rest of the arguments.
4. If the match fails at any point, Prolog looks for the next assertion that the predicate matches and tries again to match the arguments.
5. If the predicate and all the arguments are successfully matched, the process stops and the interpreter prints **yes**, meaning 'there is a match—I can show this to be true'. The goal of finding a match has been satisfied.
6. If there is no match at all then **no** is printed, meaning 'there is no match': the goal of finding a match cannot be satisfied.

2.4.3 Variables

Suppose the question we want to ask is *What does Heather like?*; some way has to be found of asking the 'what'. The goal is to find some 'what' such that Heather likes it; anything that satisfies the question will do. In Prolog the question becomes:

```
?- likes(heather,What).
```

Note that the 'What' begins with a capital letter. This is to indicate that it is a **variable**. It performs a similar function to the word *thing* in English, or *x* in algebra (a mathematical language). It can represent any object; its meaning can vary, and so it is called a variable.

The other arguments encountered until now represent particular people or specific objects:

```
alan
heather
whisky
gin
coffee
lager
beer
```

These do not change: they always represent the same object, and so are called **constants**.

Both variables and constants are examples of structures called **atoms**. They cannot be broken down into smaller objects that mean anything to Prolog.

When a match succeeds, any variables are given the value of any constant that they match to. This matching process is one of the very powerful facilities that comes free with Prolog; the Prolog interpreter does this matching automatically and remembers what is matched to what.

When a variable has no value, we refer to it as an **uninstantiated variable**. When it gets given a value by the matching process, it is referred to as an **instantiated variable**.

So, above, if we match

```
?- likes(heather, What).
```

to

```
likes(heather, gin).
```

then we say that `What` has become instantiated to `gin`. The Prolog interpreter prints out the values of any variables that have become instantiated in the process of matching:

```
What = gin
```

Note that the same variable can appear in different questions and provide different answers. This is because the variable names only apply to each question: this is referred to as the **scope** of the variable. In Prolog, the scope of the variables is said to be **local**, that is, they only have the same value in a limited (local) environment. All the variables in the same question with the same name become instantiated at the same time to the same value.

2.5 Prolog Terminology and Syntax

Prolog programs consist of three kinds of **clauses**:

- facts
- questions
- rules

Terms:

- a constant is a term
- a variable is a term
- a compound term is a term

Constants:

- an atom is a constant
- an integer is a constant
- a real number is a constant

Atoms are made up of:

- letters
- digits
- the underscore
- symbols (+, -, *, /, \, ^, <, >, =, ~, :, .., ?, @, #, \$, &)

A quoted string is an atom.

Constants are used to refer to objects. Constants begin with lower case letters. Note that all predicate names are constants—don't use a variable for a predicate:

```
X(jane, jim).
```

Certain conventions are used when writing Prolog—in other words, Prolog, like other languages, has a **syntax**:

- All predicates start with a lower case letter.
- All variables begin with an upper case letter.²
- The format of each fact or assertion is:
 - a predicate followed by any number of arguments;
 - the arguments are separated by commas and enclosed by round brackets;
 - there is no space between the predicate and the opening bracket; and
 - a full stop follows the enclosing round bracket.

This is shown diagrammatically in Figure 3.

²Later, you will discover other ways of writing variables.




Figure 3: The elements of a fact

- The predicate can be a string like `son_of`, `drinks`, `likes`, and so on.
- The arguments can be **constants**, **variables**, or even other assertions.

The Prolog interpreter prints the characters `| ?-` to indicate that it is waiting for the user to ask a question or to set a goal to be solved; we refer to these characters as the Prolog **prompt**.

2.6 Expressing Relationships in Prolog

Suppose I want to say:

the capital of paris is France

In Prolog we might write this as:

```
has_capital(france, paris).
```

where we are expressing a relationship `has_capital` between 2 arguments, `france` and `paris`. Note that the full stop is used to terminate the clause, and that objects are referred to with words beginning with lower case letters. We can express a relationship with more than two things:

```
ate(robert, curry, breakfast).
```

Or one argument relations (usually called predicates):

```
ate_curry_for_breakfast(robert).
robert_ate_for_breakfast(curry).
robert_ate_curry(breakfast).
```

Or 0-argument relations:

It is all a matter of how you choose to represent the relationships, which will depend on what you want to do with them.

2.7 Unifying Terms

The process of matching is also referred to as **unification**. When two terms match we say that they **unify**. It is by the process of matching or unification that variables get instantiated. Unification is a two-way matching process. It operates on any pair of Prolog **terms**. For example, unifying `loves(john, X)` and `loves(Y, mary)` results in `loves(john, mary)`. There is a predicate defined for us in Prolog that allows us to test whether or not two things unify. Because it is provided by the system it is known as a **system predicate**. It is the infix predicate `=/2`. Note that **infix** means that it is placed between the two things that are being unified, and the `/2` is known as the **arity** of the predicate, and indicates how many arguments the predicate has. So, the predicate `=/2` takes two arguments and tries to unify them.

For example, we can unify the following sets of terms, resulting in the outcomes shown. Some comments are given in `()` after each match.

	Pairs of terms	Outcome
1a.	<code>?- fred=X.</code> (the variable X unifies with the constant fred)	<code>X=fred</code> yes
1b.	<code>?- c=letter(c).</code> (a constant cannot unify with a one argument predicate)	no
1c.	<code>?- f(tee)=f(S).</code> (the predicate names are the same; the variable S unifies with the constant tee)	<code>S=tee</code> yes
1d.	<code>?- father(john,tom)=father(tom,Who).</code> (the constants john and tom cannot unify)	no
1e.	<code>?- centre(a,X,c)=centre(Y,b,c).</code> (the constant a unifies with variable Y, constant b with X)	<code>Y=a X=b</code> yes
1f.	<code>?- colour(N,N)=colour(green,X).</code> (variable N unifies with constant green; variables N and X match, so X becomes instantiated to green also)	<code>N=green X=green</code> yes
		<small>\footnote{A variable will always unify with another variable. We refer to them then as shared variables. If one then unifies with a constant or term, the other shares that value also.}</small>
1g.	<code>?- first(sue,dave,bob)=first(N,dave,N).</code>	no

(variable N unifies with constant sue, but it then will not match in the 3rd argument with another constant bob)

- 1h. ?- drink(beer(lorimers,eighty),Pub)=drink(What,mathers).
 What=beer(lorimers,eighty) Pub=mathers
(the first argument here is a term, which will unify with the variable What)
- 1i. ?- havecar(metro)=havecar(Yes). Yes=metro
(variable Yes and constant metro unify)
- 1j. ?- f(X,Y,z)=f(z,a,Z). X=z Y=a Z=z yes
(X unifies with z in the first argument; Y with a in the 2nd and Z with z in the 3rd.)

2.8 Conjunctions

2.8.1 Asking Questions that Contain Conjunctions

Earlier we saw how to ask Prolog simple questions. Suppose we wanted to ask more complex questions like the following:

Is it true that both Alan and Heather like coffee?
Is there anything that Heather hates but Alan likes?

To answer these questions, we have to break them down into simpler questions, and ask each of them, one after the other. In Prolog we do this as follows:

```
| ?- <question1>, <question2>.
```

So, for example, we might write:

```
| ?- likes(heather, coffee), likes(alan, coffee).
```

yes

```
| ?- likes(heather, gin), likes(alan, whisky).
```

yes

```
| ?- likes(alan, beer), likes(heather, beer).
```

no

Prolog takes each subgoal in the query (one at a time, going from left to right) and an attempt is made to satisfy it. As each subgoal succeeds, the next is tried. If all the subgoals

succeed, then the question as a whole succeeds. If at any point one of the subgoals fails, then the whole question fails.

The comma in the query here is read as *and*, and is referred to as the **conjunction** (the subgoals being **conjoined subgoals**). So, if we have two assertions together that we want to show to be true, then they each must be true. In logic, this is equivalent to saying that:

A and B is true if A is true and B is true.

2.8.2 Using Variables in Conjunctions

As with simpler questions, variables can appear in conjunctions too. However, if a variable appears more than once in a series of conjoined goals it will always match to the same value: i.e., if the variable `What` is instantiated to `beer` in one subgoal it will also be instantiated to `beer` in all other subgoals in the same question.

```
| ?- likes(heather, What), likes(alan, What).  
  
X = coffee ?  
yes  
| ?- likes(alan, Something), hates(heather, Something).  
  
Something = whisky ?  
yes
```

Variables with different names may or may not share the same instantiations, depending on what they match to:

```
| ?- drinks(alan, X), likes(heather, S).  
  
X = beer  
S = gin ?  
yes  
| ?- drinks(X, Y), hates(X, Y).  
  
no
```

Note that a query can contain any number of conjoined goals.³

2.9 Summary

The basic mode of operation of Prolog is thus as follows.

1. We provide Prolog with a question.

³This is why when you forget the full stop and go on to the next line it doesn't matter: you might have wanted to have a lot of subgoals that would not all fit on one line.

2. Prolog matches the question with assertions in a database,

- looking for exact matches of predicates and constants; and
- looking for variables to match to anything (including other variables).

In the next sections, we look at how Prolog works, then go on to introduce the notions of **rules** and **backtracking**.

2.10 Exercises

4

Question 2.1 The predicate `=/2` takes 2 arguments and tries to unify them. For each of the following:

- specify whether the query submitted to Prolog succeeds or fails;
- if it succeeds, specify what is assigned to any variables.

2.1a `?- Pear = apple.`

2.1b `?- car = beetle.`

2.1c `?- likes(beer(murphys),john) = likes(Who,What).`

2.1d `?- f(1) = F.`

2.1e `?- name(Family) = smith.`

2.1f `?- times(2,2) = Four.`

2.1g `?- 5*3 = 15.`

2.1h `?- f(X,Y) = f(P,P).`

2.1i `?- a(X,y) = a(y,Z).`

2.1j `?- a(X,y) = a(z,X).`

⁴Answers to all exercises are given in the final chapter of these notes.

Question 2.2 The following Prolog program is consulted by the Prolog interpreter.

```
vertical(seg(point(X,Y),point(X,Y1))).  
horizontal(seg(point(X,Y),point(X1,Y))).
```

What will be the outcome of each of the following queries?

2.2a ?- vertical(seg(point(1,1),point(1,2))).

2.2b ?- vertical(seg(point(1,1),point(2,Y))).

2.2c ?- horizontal(seg(point(1,1),point(2,Y))).

2.2d ?- vertical(seg(point(2,3),P)).

2.2e ?- vertical(S), horizontal(S).

Question 2.3 The following Prolog program is consulted by the Prolog interpreter.

```
parent(pat,jim).  
parent(pam,bob).  
parent(bob,ann).  
parent(bob,pat).  
parent(tom,liz).  
parent(tom,bob).
```

What will be the outcome of each of the following queries?

2.3a ?- parent(bob,pat).

2.3b ?- parent(liz,pat).

2.3c ?- parent(tom,ben).

2.3d ?- parent(Pam,Liz).

2.3e ?- parent(P,C),parent(P,C2).

Question 2.4 The following Prolog program is consulted by the Prolog interpreter.

```
colour(b1,red).  
colour(b2,blue).  
colour(b3,yellow).
```

```
shape(b1,square).
shape(b2,circle).
shape(b3,square).
size(b1,small).
size(b2,small).
size(b3,large).
```

What will be the outcome of each of the following queries?

2.4a ?- shape(b3,S).

2.4b ?- size(W,small).

2.4c ?- colour(R,blue).

2.4d ?- shape(Y,square),colour(Y,blue).

2.4e ?- size(X,large),colour(X,yellow).

2.4f ?- shape(BlockA,square),shape(BlockB,square).

2.4g ?- size(b2,S),shape(b2,S).

2.4h ?- colour(b1,Shape),size(X,small),shape(Y,circle).

Question 2.5 The following Prolog program is consulted by the Prolog interpreter.

```
film(res_dogs,dir(tarantino),stars(keitel,roth),1992).
film(sleepless,dir(ephron),stars(ryan,hanks),1993).
film(bambi,dir(disney),stars(bambi,thumper),1942).
film(jur_park,dir(spielberg),stars(neill,dern),1993).
```

What will be the outcome of each of the following queries?

2.5a ?- film(res_dogs,D,S,1992).

2.5b ?- film(F,dir(D),stars(Who,hanks),Y).

2.5c ?- film(What,Who,stars(thumper),1942).

2.5d Write the query that would answer the question:

”Who directed Jurassic Park (jur_park)?”

and give the outcome of the query.

2.5e Write the query that would answer the question:

”What film did hanks appear in in 1993 and who was the other star?”

and give the outcome of the query.

3 How Prolog Works

3.1 Introduction

This section deals with the basics of running Prolog programs.

3.2 Running, Consulting and Editing Prolog Programs

3.2.1 Starting Prolog

Giving the command `prolog` to Unix will get you into Prolog:

```
snoopy[18]: prolog
```

You'll see something like this:

```
SICStus 3 #5: Tue Aug 26 10:14:51 BST 1997
| ?-
```

This tells you that the dialect of Prolog that you are using is SICStus Prolog.

3.2.2 Consulting a File

When you get the `| ?-` prompt, type in the name of the file you want to load into the Prolog database (i.e., the file to be **consulted**):

```
| ?- consult(family).
{consulting family...}
{file1 consulted, 20 msec 1015 bytes}

yes
| ?-
```

or use the following abbreviated form instead (but don't do both):

```
| ?- [family].
```

You will now have the contents of the file `family` in Prolog's database. The Prolog interpreter will know everything that is in this file.

3.2.3 Seeing What's in the Database

You can see what facts and rules (i.e., what clauses) Prolog knows about at any point by **listing** the contents of the prolog database:

```
| ?- listing.
```

Instead of listing the whole program, you can just list parts of it by using the predicate **listing** and the name of the predicate to be listed:

```
| ?- listing(parent).
```

```
parent(A, B) :-  
    father(A, B).  
parent(A, B) :-  
    mother(A, B).
```

```
yes
```

3.2.4 Getting Out of Prolog

Type **halt** to get out of Prolog:

```
| ?- halt.
```

```
snoopy[19]:
```

3.2.5 Writing and Modifying Programs

We can write new programs or alter existing ones using the editor **emacs**. We can add some new clauses to the file **family**; then if we go back into Prolog, and consult this file, the new clauses will be known by the Prolog interpreter.

In general, when writing prolog programs or altering existing ones, you should do the following:

- Write the clauses out on paper first.
- Get into the editor.
- Add to or change the text of the program.
- Go into Prolog.
- Consult the file containing the program.
- List the contents of the database.
- Test the program.
- Work out any changes you need to make.

- Exit Prolog and start again.

Although it may seem a waste of time to work out the clauses on paper first, if you don't do this you may degenerate into hacking away at the program without having a clear idea of what is really going on.

4 Rules

4.1 Introduction

In previous sections, we looked at a simple Prolog program, and introduced **facts** (or assertions), **questions**, **variables**, **syntax** and **conjunctions**. In this section, we go on to discuss **rules**.

4.2 Rules

4.2.1 Expressing Rules in Prolog

As we saw earlier, from the information that the Prolog interpreter has it can answer the question:

```
| ?- drinks(alan, beer).
```

```
yes
```

If you had the same information—that is, that *Alan drinks beer*—and were asked the question *Does Alan like beer?*, you would probably answer *Yes*. Knowing that *Alan drinks beer* allows you to infer that *Alan likes beer* (on the grounds that people don't usually drink things that they don't like). To see whether or not the Prolog interpreter is able to make the same inference—i.e., to see if it is able to reason that *If Alan drinks beer, then he must like it*—the question we must ask is:

```
| ?- likes(alan, beer).
```

The Prolog interpreter would reply:

```
no
```

meaning 'I cannot prove that Alan likes beer'. To allow Prolog to make the same inferences as we do, we have to provide the information about the relationship between *liking* and *drinking* explicitly; i.e., we have to find some way of encoding the rule that would be expressed in English as

If someone drinks beer then we can infer that that person likes beer.

or

We can say that someone likes beer if we know can prove that they drink it.

In Prolog this becomes:

```
likes(Person, beer):-                % a Person likes beer if ...
    drinks(Person, beer).           % ... that Person drinks beer
```

Some things to note here:

- The first predicate-argument structure here is what you are trying to show to be true (to prove), in the same way as you would try to satisfy any other assertion.
- The ‘:-’ is read as *if*.
- Whatever comes after the *if* and before the full stop is what has to be satisfied for the whole rule to succeed.

In the present example, there is one subgoal to satisfy. So, to satisfy the goal `likes(Person, beer)`, the subgoal `drinks(Person, beer)` must be satisfied.

The whole structure is generally referred to as a **rule**. A rule is one type of **clause**, just as a fact or assertion is also a clause; in particular, a rule is a clause with a **head** (the left-hand side of the ‘:-’) and a **body** (everything to the right-hand side). There can be any number of subgoals in the body of the clause.

A fact or assertion is then a clause with no body.

Restrictions on the form of a rule:

- Only one goal may appear in the head.
- Any number of goals may appear in the body, separated by commas.

So, we can’t have:

```
happy(fred), powerful(fred):-
    rich(fred).
```

We can have more than one goal in the body of a rule:

A man is happy if he is rich and famous.

can be expressed in Prolog as:

```
happy(Person):-
    man(Person),
    rich(Person),
    famous(Person).
```

Here we have three **conjoined subgoals**.

To express:

Someone is happy if they are healthy, wealthy or wise.

in Prolog, we first rewrite it as

```
Someone is happy if they are healthy or
Someone is happy if they are wealthy or
Someone is happy if they are wise.
```

Then:

```
happy(Person):-
    healthy(Person).
happy(Person):-
    wealthy(Person).
happy(Person):-
    wise(Person).
```

So the query "is someone happy" translates as `?- happy(Someone).` and will succeed if any of the three rules succeed i.e. if either `healthy(Someone).` or `wealthy(Someone).` or `wise((Someone).` can be proved. We may want to be more specific:

A woman is happy if she is healthy, wealthy or wise.

In Prolog this becomes:

```
happy(Person):-
    healthy(Person), woman(Person).
happy(Person):-
    wealthy(Person), woman(Person).
happy(Person):-
    wise(Person), woman(Person).
```

Here we have to check in every rule that Someone is a woman. We could have avoided this by using another rule in addition to the one above:

```
happy_woman(P):-
    woman(P), happy(P).
```

4.2.2 Returning to the 'Drinks' Example

If you want to write some conjunction of goals and use it generally (that is, you want to be able to vary what the variables match to), then it would be more convenient to write a rule for it (with all the conjoined subgoals as its body) than to keep rewriting all the conjoined goals separately each time.

If, for example, you wanted to know about the drinks that two people both like, we could write a rule `bothlike(Person, Other, Drink)` that is satisfied if both `Person` and `Other` like the same `Drink`:

```

bothlike(Person , Other, Drink):-      % Person and Other bothlike Drink if
    likes(Person, Drink),              % .. Person likes Drink and
    likes(Other, Drink).               % .. Other likes Drink

```

We would have to add this to the file in which we keep the rest of the clauses about what people like and drink. We would use an editor to do this, and then go back into Prolog and consult the amended file. The following goal (or query) could then be tried:

```
| ?- bothlike(alan, heather, S).
```

This would match to the head of the rule, with `Person` instantiated to `alan` (we will write this as `alan/Person`), `heather/Other` and `S/Drink`.⁵ The first subgoal to satisfy will then be:

```
likes(alan, Drink)
```

which will match with `S/coffee`. The next subgoal to satisfy is then

```
likes(heather, coffee)
```

which succeeds. There are no more subgoals, so `bothlike(alan, heather, S)` succeeds with `S` instantiated to `coffee`.

```
| ?- bothlike(alan, heather, S).
```

```
S = coffee ?
```

You may also have noticed that, by this rule, if we asked what two people both like coffee we would also find out that both Alan and Alan like coffee! Whilst this may seem silly (it is not quite what we intended), it is perfectly sensible logically.

```
| ?- bothlike(A, B, coffee).
```

```
A = alan
B = alan ?
yes
```

If we want to also say that we don't want to have both people being the same person, then we have to explicitly state that they are not the same person. This means that we have to say that the variables that match to the first two arguments of `bothlike(Person, Other, Drink)` must not be instantiated to the same value; in other words, `Person` must not be the same as `Other`. We add a subgoal to the rule in order to say this:

⁵Note that neither `X` nor `S` have a value at this point, but that they will share the same value as soon as one or the other of them becomes instantiated by matching.

```

bothlike(Person , Other, Drink):- % Person and Other bothlike Drink if
    likes(Person, Drink), % .. Person likes Drink and
    likes(Other, Drink). % .. Other likes Drink and
    Per\==Other. % .. Person and Other are not the same

```

For the remainder of this section, however, we will leave this clause with the possibility of `Person` and `Other` being the same.

4.2.3 Recursion

Suppose we decide that, given two people A and B, A likes B if A and B both like the same drink. So we are saying:

```

some person likes some other person if
    the first person likes some drink and
    the other person likes the same drink.

```

We can express this in Prolog as follows:

```

likes(Person, Other):- % Person likes Other if ..
    likes(Person, Drink), % .. Person likes Drink and
    likes(Other, Drink). % .. Other likes Drink.

```

This program is **recursive**, that is, it calls itself. Some recursive statements:

- An ancestor is a parent or a parent's ancestor.
- A string of characters is a single character or a single character followed by a string of characters.

An example recursive program:

```

talks_about(A,B):-
    knows(A,B).
talks_about(P,R):-
    knows(P,Q),
    talks_about(Q,R).

```

In English:

You talk about someone if you know them or you know someone who talks about them.

Given the database:

```

talks_about(A,B):-
    knows(A,B).
talks_about(P,R):-
    knows(P,Q),
    talks_about(Q,R).
knows(bill,jane).
knows(jane,pat).
knows(jane,fred).
knows(fred,bill).

```

and the goal:

```
| ?- talks_about(X,Y).
```

what do we get? Try this for yourself and see.

Another example: Given the database:

1. has_flu(X):- infected(X).
2. has_flu(X):- kisses(X,Y), has_flu(Y).
3. infected(john).
4. kisses(sue,john).
5. kisses(john,ann).
6. kisses(sally,ann).
7. kisses(fred,sue).

the outcome of each of the following queries is given, together with some explanation. Note that the above predicates are numbered for ease of reference.

a. ?- has_flu(john). yes

```

(rule 1:has_flu(john):-infected(john).
subgoal infected(john) matches 3.
all subgoals succeed, rule 1 succeeds)

```

b. ?- has_flu(sue). yes

```

(rule 1:has_flu(sue):-infected(sue).
  fails.
rule 2: has_flu(sue):- kisses(sue,Y),has_flu(Y).
subgoal kisses(sue,Y) matches 4, Y=john
subgoal has_flu(john)
  rule 1:has_flu(john):-infected(john).
  subgoal infected(john) matches 3.
  all subgoals succeed, rule 1 succeeds
all subgoals succeed, rule 2 succeeds)

```

c. ?- has_flu(sally). no

```
(rule 1:has_flu(sally):-infected(sally).
  fails.
rule 2: has_flu(sally):- kisses(sally,Y),has_flu(Y).
subgoal kisses(sally,Y) matches 6, Y=ann
subgoal has_flu(ann)
  rule 1:has_flu(ann):-infected(ann).
  fails
  rule 2:has_flu(ann):-kisses(ann,Y2),has_flu(Y2).
  subgoal kisses(ann,Y2)
    fails)
```

d. ?- has_flu(fred). yes

```
(rule 1:has_flu(fred):-infected(fred).
  fails.
rule 2: has_flu(fred):- kisses(fred,Y),has_flu(Y).
subgoal kisses(fred,Y) matches 7, Y=sue
subgoal has_flu(sue)
  rule 1:has_flu(sue):-infected(sue).
  fails.
  rule 2: has_flu(sue):- kisses(sue,Y1),has_flu(Y1).
  subgoal kisses(sue,Y1) matches 4, Y1=john
  subgoal has_flu(john)
    rule 1:has_flu(john):-infected(john).
    subgoal infected(john) matches 3.
    all subgoals succeed, rule 1 succeeds
  all subgoals succeed, rule 2 succeeds
all subgoals succeed, rule 2 succeeds)
```

4.2.4 Getting multiple answers

Returning to our previous example, the entire program that results from adding the recursive likes/2 rule is shown in Figure 4; here are some examples of queries and responses given this database:

```
| ?- likes(alan, beer).

yes
| ?- likes(alan, heather).

yes
| ?- likes(alan, alan).

yes
| ?-
```

```

drinks(alan, beer).
drinks(heather, lager).
likes(alan, coffee).
likes(alan, whisky).
likes(heather, gin).
hates(heather, whisky).
likes(heather, coffee).
bothlike(Person, Other, Drink):-
    likes(Person, Drink),
    likes(Other, Drink).

likes(Person, beer):-
    drinks(Person, beer).

likes(Person, Other):-
    likes(Person, Drink),
    likes(Other, Drink).

```

Figure 4: The complete program so far

Consider the query *What are all the things that Alan likes?*. In Prolog we would ask this by saying:

```
| ?- likes(alan, What).
```

and in response we would get:

```
What = coffee ? ;
```

The question mark here is Prolog's way of saying 'what do you want me to do next?'. If we just type return, Prolog will say *yes* and return us to the Prolog prompt. However, if we type a semi-colon before we hit the return key, the Prolog interpreter will look for the next solution and print that:

```
What = whisky ? ;
```

We can keep asking for more answers:

```
What = beer ? ;
```

```
What = alan ? ;
```

```
What = heather
```

```
yes
```

If we decide that this is enough we can stop here.

4.3 Summary

The behaviour we have just seen, where Prolog can provide us with more than one answer to a query, raises some interesting questions:

- How does Prolog get these solutions?
- What order does it get them in?
- What are we really doing (or what is the Prolog interpreter doing) when we ask it to find the next solution?
- Does it matter what order the clauses are in?

These are all questions that will be answered in the next section where we will be looking at **backtracking** in Prolog, and the use of trees to represent what the Prolog interpreter is doing.

4.4 Exercises

Question 4.1 The following Prolog program is consulted by the Prolog interpreter.

```
big(bear).
big(elephant).
small(cat).
brown(bear).
black(cat).
grey(elephant).
dark(Animal):- black(Animal).
dark(Animal):- brown(Animal).
```

What will be the outcome of each of the following queries?

- 4.1a ?- dark(X), big(X).
- 4.1b ?- big(X), grey(Y).
- 4.1c ?- dark(D), small(D).
- 4.1d ?- big(Animal), black(Animal).
- 4.1e ?- small(P), black(P), dark(P).

Question 4.2 The following Prolog program is consulted by the Prolog interpreter.

```
knows(A,B):-
    friends(A, B).
```

```
knows(A,B):-  
    friends(A, C),  
    knows(C, B).
```

```
friends(john, alice).  
friends(alice, tom).  
friends(sue, john).  
friends(sue, clive).  
friends(fred, tom).  
friends(tom, sue).
```

State whether the following queries succeed or fail. If a query fails, explain why.

4.2a ?- knows(alice, john).

4.2b ?- knows(clive, sue).

4.2c ?- knows(alice, fred).

4.2d ?- knows(sue, john).

4.5 Practical 1

4.5.1 Getting Started

1. Log on and check that you have the file `famtree.pl`:
2. Copy it if you haven't:

```
snoopy[14] cp /home/infteach/prolog/code/famtree.pl family
```

This makes a copy of the file `famtree.pl` in your area and calls it `family`.

4.5.2 Running Prolog

1. Give the command `prolog` to get into Prolog:

```
snoopy[15] prolog
```

You'll see something like the following:

```
snoopy[15] prolog
SICStus 3 #5: Tue Aug 26 10:14:51 BST 1997
| ?-
```

2. When you get the `| ?-` prompt, type in the name of the file to be loaded into the Prolog database (i.e., the file to be consulted):

```
| ?- consult(family).
{consulting /hame/helen/family.pl...}
{/hame/helen/family.pl consulted, 10 msec 1552 bytes}
```

```
yes
| ?-
```

or use the abbreviation for this instead (but don't do both):

```
| ?- [family].
```

You will now have the contents of the `family` file in the Prolog database. The **Prolog interpreter** will know everything that is in this file.

3. Look to see all the facts and rules (i.e., clauses) that Prolog knows about at the moment by typing `listing`:

```
| ?- listing.
```

Don't forget the full stop.

4. Instead of listing the whole program, just list parts of it by using the predicate `listing` and the name of the predicate to be listed:

```
| ?- listing(parent).
```

```
parent(A, B) :-  
    father(A, B).
```

```
parent(A, B) :-  
    mother(A, B).
```

yes

```
| ?- listing(son).
```

```
son(A, B) :-  
    parent(B, A),  
    male(A).
```

yes

```
| ?- listing(daughter).
```

```
daughter(A, B) :-  
    parent(B, A),  
    female(A).
```

yes

```
| ?-
```

5. Now try the following **goals**, typing each in response to the ‘| ?-’ prompt. Guess what you think will happen. See if it does. If it doesn’t quite do as you expect, can you see why?

```
| ?- mother(mary,fred).
```

```
| ?- father(tom,sue).
```

```
| ?- father(cecil,fred).
```

```
| ?- male(jim).
```

```
| ?- father(tom,Who).
```

```
| ?- mother(Mother,fred).
```

```
| ?- mother(X,Y).
```

```
| ?- parent(fred,cecil).
```

```
| ?- daughter(jane,Who).
```

```
| ?- son(What,How).
```

```
| ?- son(cecil,jane).
```

6. Try seeing if there is more than one match for some of the above goals: by typing ‘;’, you ask Prolog to look for another solution.

```
| ?- mother(mary, S).
```

```
S = tom ? ;
```

```
S = jane ? ;
```

```
S = fred ? ;
```

```
no
```

```
| ?-
```

‘no’ here means ‘no more solutions’.

Try this with other goals.

7. Think of some goals of your own for Prolog to satisfy. For example, how would you ask the following?

- Is Tom the father of Jim and of Sue?
- Who is the father of Cecil and the son of Mary?
- Is there anyone who has a son and is themselves the son of someone?
- Do Jane and Fred have the same mother?

A clue: remember the **conjunction**. You can ask if goal A is true and if goal B is true by typing

```
A, B.
```

8. Think how you would write rules for sister, brother, aunt and uncle.

4.5.3 Getting out of Prolog

Type `halt` to get out of Prolog:

```
| ?- halt.
```

You’ll see that you get the unix prompt again:

```
| ?- halt.
```

```
snoopy[16]
```

4.5.4 Editing the Program

We can alter the file `family` by using the editor `ue` (Emacs).

We can add some new clauses to the file. Then if we go back into Prolog and consult this file, the new clauses will be known by the Prolog interpreter.

Because we have made the changes in the file itself, we can save them to be used again, and we can also change them easily (for example, if we make a mistake in writing the clauses).

1. Get into the editor by typing the following:

```
snoopy[16] emacs family
```

2. Add some more people to the database.
3. We could also add the `grandparent` rule. You can put this anywhere in the file you like.

```
grandparent(Grandparent, Grandchild):-  
    parent(Grandparent, Parent),  
    parent(Parent, Grandchild).
```

4. Save the file by typing `CONTROL-X`, `CONTROL-C`, or select save from the emacs menu.

You are now back at Unix command level.

4.5.5 Testing the Program

1. Look at the file to see your latest version.

```
snoopy[17] more family
```

2. Get into Prolog, and consult your new version of `family`.
3. Try out some goals that make use of the new clauses you've added. If you spot any errors, you will need to edit the file again to alter it.

```
snoopy[19] prolog  
SICStus 2.1 #9: Mon Jul 11 10:16:09 BST 1994  
| ?- consult(family).  
{consulting /usr/local/dai/docs/dai/teaching/modules/1-prolog/code/famtree.pl...}  
{/usr/local/dai/docs/dai/teaching/modules/1-prolog/code/famtree.pl  
consulted, 10 msec 2784 bytes}  
  
yes  
| ?- listing.
```

4.5.6 More Things to Add

Try adding brother/sister and aunt/uncle clauses:

- write them out on paper first;
- get into the editor;
- add to or change the text;
- get back into Prolog;
- consult the file;
- list the database;
- test the new clauses;
- look for any changes needed; and
- start again.

4.5.7 Harder Things

If you get this far, and are quite happy, then try something harder.

1. How would you add facts and rules (i.e., clauses) to the database to give information about people's ages, and to be able to answer questions like *Is Fred older than Jim?* or *How old is Sue?*

Some hints:

- You could have a clause called `age` with two arguments, one for a person's name and the other for that person's age; for example:

```
age(fred,30).  
age(mary,72).
```

- You can compare ages using '>', which means 'is greater than'. $X > Y$ is true (the goal succeeds) if whatever matches to X is greater than whatever matches to Y . So, if X becomes instantiated to 72 and Y to 30, then the goal $X > Y$ would succeed.

Try:

```
| ?- age(cecil,X), age(mary,Y), Y>X.
```

(you have to add some ages to the database first).

2. There might be a rule

```
is_older(OldPerson, YoungPerson)
```

which is true when the `OldPerson` is older than the `YoungPerson`.

The whole rule, if we were comparing Mary and Cecil's ages, would be

Mary is older than Cecil if
we know Mary's age and
we know Cecil's age and
Mary's age is greater than Cecil's age.

Try to write this rule in Prolog.

3. We could have different rules for cases where we don't know everyone's age, but we know that if Fred is the parent of Cecil then Fred must be older than Cecil. See if you can write the corresponding Prolog rule.
4. We might try to write a rule that looks like this:

Mary is older than Cecil if
Mary is older than someone and
that someone is older than Cecil.

We are getting into more hairy ground here with clauses that have subgoals with the same name as themselves. If a clause calls a clause of the same name (i.e., if it has itself as a subgoal), we say that it is **recursive**.

4.6 Practical 2: Writing a Program to Find Who is Where

This practical is about writing a program to incorporate information about the location of people and their phone numbers. Your program should be able to answer questions like *Where is Fred?* and *What is Fred's phone number?* (given in appropriate Prolog terms, of course). Your program will also know about people visiting others and how to contact them.

1. Write predicates to incorporate the following information:

Helen has room F5.
Frank has room E6.
Paul has room F9.
Han has room E6.
Robert has room F9.
Dave has room E10.
Henry has room E12.
Janet has room E11.
Graeme has room E13.

For each fact, use a predicate `room` with two arguments: the first argument will be the person, and the second the room number.

2. Use MicroEmacs (`ue`) to put these clauses into a file on your area. Check that there are no errors. Then go into Prolog and consult this file.
3. Test the program by asking questions such as:

Which is Dave's room?
Who has room F9?

You have to rephrase the questions in Prolog, of course.

4. Add some more clauses to store the telephone number for each room. Use a predicate `phone` with two arguments (hereafter referred to as `phone/2`): the first argument should be the room number and the second should be the extension number. The extensions are as follows:

Room	Tel No.
E10	231
E12	233
E11	244
E13	237
F5	242
F9	239
E6	247

5. Again, use the editor to edit your original file, then go back into Prolog and consult the file and test it. Ask questions like:

Which room has extension 239?

What is room E6's extension?

6. Add a rule that will allow you to find out a person's phone number, if you know their room number and the phone number for that room. Use a predicate `ring/2`, whose first argument should be the person and second argument their phone number. So, a declarative reading for the rule you have to write might be something like

P can be rung at number N if P is in room R and room R has number N.

7. Again, edit the file, consult it and test it.

4.6.1 Complicating the Program

Suppose there are other people we know about, but instead of knowing their room numbers, we know whether or not they are visiting the office of someone else we know about. In this section, we'll incorporate this kind of information into the program.

1. To find out where someone is, first you would check to see if you have their room number, and if not, you would then check to see if they are visiting someone else and you have that person's room number.

Add the following information, using a predicate `visiting/2`:

Alan is visiting Helen.

Liam is visiting Paul.

Jane is visiting Alan.

Also, add a rule that tells you how to find people: you can find person P if they are in their room, or (if that fails) you can find person P if they are visiting person Q and you can find person Q.

Use a predicate `find/2`, the first argument being the person you want to find and the second being the room that they are found in (even if it is the room of the person they are visiting).

2. Edit the program, run it and test it by asking the following questions and some others of your own:

Who is Alan visiting?

Where can you find Dave?

Where can you find Jane?

Who is in room F9?

3. If you alter the predicate `ring`, so that it has the subgoal of `find(Person,Room)` instead of `room(Person,Room)`, you should be able to get the phone number you need to know in order to contact any person, even if they are visiting somebody else (or even if they are visiting someone who is already visiting someone!).

Do this and test it: find everyone's phone numbers.

5 Backtracking

This section deals with Prolog's search strategy, explaining how the Prolog interpreter backtracks to get more than one solution to a problem. The idea of using trees to represent the way that the Prolog interpreter searches for solutions is also introduced.

5.1 Getting more than one answer

The version of the `drinks` program that we will use below is shown in Figure 5. The clauses are numbered here to make it easier to refer to them later. Note that we are using a more general version of clause 9 here than that introduced earlier: we have substituted a variable for the occurrences of `beer` in the corresponding rule.

Given this database, the question *What are all the things that Alan likes?* will cause the following answers to be generated:

```
| ?- likes(alan, What).  
  
What = coffee ? ;  
  
What = whisky ? ;  
  
What = beer ? ;  
  
What = alan ? ;  
  
What = heather ? ;
```

After this solution we get no more; instead, the program seems to stop working, and we hear nothing back from Prolog. In some implementations of Prolog, we will be presented with an error message like the following:

```
%%% Local stack overflow - forced to abort %%%
```

In SICStus Prolog, the program doesn't come back at all. To restore things to normal, you have to type a Control-C character (hold down the key marked `CONTROL` or `CTRL`, and press the 'c' key while the `CONTROL` key is depressed).

Prolog will then respond with

```
^C  
Prolog interruption (h for help)?
```

At this point, you should type a to abort the program. Prolog will respond with

```
{ Execution aborted }  
| ?-
```

Why this problem occurs will become clear below; for the moment, we will focus on how the Prolog interpreter gets all these solutions to the question.

```

1.  drinks(alan, beer).
2.  drinks(heather, lager).
3.  likes(alan, coffee).
4.  likes(alan, whisky).
5.  likes(heather, gin).
6.  hates(heather, whisky).
7.  likes(heather, coffee).
8.  bothlike(Person, Other, X):-
        likes(Person, X),
        likes(Other, X).
9.  likes(Person, Drink):-
        drinks(Person, Drink).
10. likes(Person, Other):-
        likes(Person, Drink),
        likes(Other, Drink).

```

Figure 5: The complete drinks program

5.2 Representing Prolog's behaviour using trees

The first solution to the query used above comes from the first match of `likes(alan, What)` with `likes(alan, coffee)`, so we have the instantiation `What/coffee`.

If we take all the clauses in the database in order, this is the first clause to match (clause 3 in the table above). We can represent this as a tree with the root node being the top level goal (`likes(alan, What)`) and its daughter nodes being either:

- subgoals that must be satisfied for this top level goal to succeed;
- an indication of success (shown as \bigcirc) where this goal can succeed directly; or
- an indication of failure (shown as \otimes) where there is no match that can be made, either directly or if other subgoals are satisfied.

The arcs of the tree will be labelled with the number of the matching clause, and alongside will be written any instantiations that are made in the process of matching. Our first solution is thus as shown in Figure 6.

If we want to find other solutions, what we are doing (in effect) is failing the match that we have just found and saying 'Look for the next one'. Whenever a match is made, the clause that matches is noted and any instantiations are also noted by the interpreter. If we say 'Go back and redo that match' (by typing ';') then the search for a solution continues from where it left off. All instantiations made by this last match (that has now been made to fail) are forgotten. So here, the Prolog interpreter 'forgets' `What/coffee`, then looks for the next clause (after clause 3) that will match.

This will be clause number 4, with `What/whisky` (`What` instantiated to `whisky`); this is shown in Figure 7.




Figure 6: The first solution




Figure 7: The second solution




Figure 8: The third solution

With game playing programs, we can draw a tree to show all the possible or potential paths to a solution (all the legal moves); here, we use a slightly different kind of tree here to show the solutions to the query.

After the match with clause 4, there are no more simple assertions that match. The next match is with clause 9: in this clause, `alan` matches `Person` and `What` matches `Drink`. This is a rule, however, so now we have to prove the subgoal in the right-hand side of the rule.

We represent the subgoal explicitly in the tree, and look for matches for it. This subgoal is treated as if it is a completely new goal, so a match is attempted for the subgoal `drinks(alan, What)`.

Prolog will look through the clauses in the database from the beginning until a match is found; in the present case, we get a match with clause 1. So, the subgoal `drinks(alan, What)` succeeds with `What/beer`. Since all of its subgoals have been satisfied, `likes(alan, What)` also succeeds. The corresponding tree is shown in Figure 8.

On redoing after this match, the interpreter will look for another match for `drinks(alan, What)` first, forgetting the match of `What` to `beer`.

There is no other match for this goal, so Prolog will go back further up the tree and see if it can redo the previous goal. This means looking for another match for `likes(alan, What)` after clause 9.

There is a match in clause 10. However, to satisfy clause 10, two new subgoals must be satisfied first. These are `likes(alan, Drink)` and `likes(What, Drink)`.

Note here that because `What` in the goal matches to `Other` in clause 10, both variables will share the same value. The variable called `Drink` in the rule is a new one, so the Prolog interpreter generates some new name for it. This will eventually get a value on matching, but will not be passed back to the top level goal (simply because it doesn't appear in it).




Figure 9: The fourth solution

Because there are now conjoined subgoals to match (`likes(alan, Drink)` and `likes(What, Drink)`), these are both drawn in the tree and joined together to indicate that they both have to be satisfied before the goal above them (their parent) can be satisfied; see Figure 9.

So, the first subgoal to be matched from clause 10 is `likes(alan, Drink)`, which succeeds with `D/coffee` (as the first call of `likes(alan, What)`). The second subgoal now becomes `likes(What, coffee)`. The first match to this new subgoal will be `What/alan`. So `likes(What, coffee)` succeeds with `What/alan`.

It may not make to much sense in English to say *Alan likes himself if he and himself both like coffee*, but logically it is fine.

If we redo once more, we go back to the last goal that succeeded, and redo it (that is, we fail the last goal and look for the next match). In the present case, this means we forget `What/alan` and look for another solution to `likes(What, coffee)`.

This will match to clause 7 with `What/heather`, as shown in Figure 10.

Things get messier from here on. The last goal to succeed was `likes(What, coffee)` with `What/heather` (from clause 7); we try and redo this. The next clause that may give a solution to `likes(What, coffee)` is clause 9, which could be paraphrased here as *What likes coffee if What drinks coffee*. So, `likes(What, coffee)` succeeds if subgoal `drinks(What, coffee)` succeeds. This fails altogether and we redo again.

Next we try the match with clause 10, meaning that *What likes coffee if What likes SomeD and coffee likes SomeD*.⁶ We get a first match to this with `What/alan` and `Somedrink/coffee`; so the second subgoal `likes(D, SomeD)` is now instantiated to `likes(coffee, coffee)`.

⁶We will introduce new variables names from here on to make it easier to see what is happening.




Figure 10: The fifth solution

Clauses 1 to 8 fail, and clause 9 is tried: `likes(coffee, coffee)` if `drinks(coffee, coffee)`. This fails, so clause 10 is tried again.

`likes(coffee, coffee)` if `likes(coffee, D2)` and `likes(coffee, D2)`

This means we try again on `likes(coffee, D2)`; again, clauses 1–9 all fail, and so we try clause 10:

`likes(coffee, D2)` if `likes(coffee, D3)` and `likes(D2, D3)`

Our new subgoal is `likes(coffee, D3)`, which again fails to match clauses 1–9, and so once more we have clause 10:

`likes(coffee, D3)` if `likes(coffee, D4)` and `likes(D3, D4)`

which creates a new subgoal ...

As you may have realised by now, we are never going to get a solution to this, and the program will keep on trying to find new subgoals to match for clause 10, which each in turn call clause 10, and so on. We have an **infinite loop** here.

5.3 Summary

You should have some idea now about how different solutions to a goal are achieved, and how we can use tree to represent ‘getting all the solutions’. We will call this particular kind of tree an augmented AND/OR tree, referred to hereafter as an AND/OR tree.




Figure 11: The infinite loop

In the next section, we will use a different example that includes a simple database and clauses to help us answer questions about the contents of the database. Over the next few sections, we'll extend the program in various ways to make it easier to use.

5.4 Exercises

For each of the following programs, say if the query given fails or succeeds. Give any bindings made as a consequence.

Question 5.1

```
a:-b,c.  
b.  
c:-d.  
d:-e.  
  
?- a.
```

Question 5.2

```
a:-b,c.  
c:-e.  
b:-f,g.  
b:-n.  
e.  
f.  
n.  
  
?- a.
```

Question 5.3

```
do(X):-a(X),b(X).  
a(X):-c(X),d(X).  
a(X):-e(X).  
b(X):-f(X).  
b(X):-c(X).  
b(X):-d(X).  
c(1).  
c(3).  
d(3).  
d(2).  
e(2).  
f(1).
```

5.3a ?- do(1).

5.3b ?- do(2).

5.3c ?- do(3).

5.3d ?- do(A).

6 Built-in System Predicates

6.1 Introduction

In this section, we introduce a new example. This is a simple program that represents knowledge about books, their publishers and the shops that stock these publishers. Over the next sections, this program will be extended in a number of ways to make it easier to use. In order to do so, we'll make use of **system predicates** or **built-in predicates**.

6.2 The Example Program

The program shown in Figure 12 will be referred to as `books1`, and will be the simplest version of the program that we will use.

The program contains a number of different predicates:⁷

- The predicate `stocks/2` has two arguments: the first represents the name of a bookshop, and the second the name of a publisher stocked by that shop.
- The first argument of `book/2` is a book title, and the second is the book's publisher.
- The `open/1` and `closed/1` predicates indicate whether particular shops are open or closed.

6.3 Asking the Database Some Questions

We can ask Prolog questions about this database. The sort of questions we might want to ask, in English, are things like:

- Which shop stocks *Virago*?
- Who publishes *ET Rides Again*?
- Where can I buy *I Claudius*?

In Prolog, the first two of these questions would be:

```
| ?- stocks(Shop, virago).  
  
| ?- book(et_rides_again, Publisher).
```

The third question is a little more complicated. To answer *Where can I buy I Claudius?*, we need to answer a number of questions as follows:

⁷Note: a predicate is a collection of clauses with the same predicate name and the same number of arguments; the number of arguments of a predicate is the **arity** of the predicate.

```
stocks(james_thin, sfipubs).
stocks(james_thin, virago).
stocks(james_thin, penguin).
stocks(menzies, sfipubs).
stocks(menzies, sams).
stocks(better_books, penguin).
stocks(better_books, virago).
stocks(edinbooks, sfipubs).
stocks(edinbooks, virago).
stocks(edinbooks, sams).

book(son_of_et, sfipubs).
book(et, sfipubs).
book(i_was_a_teenage_robot, sfipubs).
book(et_rides_again, sfipubs).
book(biggles_and_wendy, virago).
book(freda_the_fire_engine, virago).
book(dict_of_computing, penguin).
book(i_claudius, penguin).
book(of_mice_and_men, penguin).
book(cookbook, sams).

open(menzies).
open(better_books).
open(edinbooks).
closed(james_thin).
```

Figure 12: The books1 program

- Who publishes *I Claudius*?
- What shop stocks this publisher?
- Is this shop open?

To do this in Prolog, we need a conjunction of goals:

```
| ?- book(i_claudius, Publisher), stocks(Shop, Publisher), open(Shop).
```

and in this case the answer would be:

```
Publisher = penguin,
Shop = better_books ?
```

Note that the first instantiation of `Shop` would be `james_thin`, but this would fail in the attempt to match the third goal; the Prolog interpreter would then backtrack and redo the second goal, instantiating `Shop` to `better_books`, after which `open(better_books)` succeeds.

6.4 Using Rules

If questions about where books can be purchased are to be asked often, it is sensible to write a rule to save repetition:

```
canbuy(Book, Shop):-
    book(Book, Publisher),
    stocks(Shop, Publisher),
    open(Shop).
```

We can now ask simply:

```
| ?- canbuy(i_claudius,Shop).
```

```
Shop = better_books ?
```

```
yes
| ?-
```

6.5 Adding Built-in or System Predicates for User Interaction

We can now go on to make the program more interesting and interactive by using a number of predicates provided by the system. These come for free and do not have to be defined by the user. They are called **system predicates** or **built-in predicates**. One example of a system predicate that we have already encountered is the predicate `listing/0`.

```
| ?- go.  
What book would you like to buy?  
|: et.  
You can buy et at menzies.  
Would you like another book?  
|: yes.  
What book would you like to buy?  
|: noddy_and_big_ears.  
I don't know where you can buy that book, sorry.  
Would you like another book?  
|: yes.  
What book would you like to buy?  
|: biggles_and_wendy.  
You can buy biggles_and_wendy at better_books.  
Would you like another book?  
|: no.  
I hope I was of some help to you. Have a nice day.  
yes | ?-
```

Figure 13: An interactive session with the `books` program

```
| ?- listing.
```

This shows all clauses currently in the Prolog database. `listing/1`, on the other hand, takes a predicate name as argument and lists all the clauses with that predicate name: so, for example

```
| ?- listing(book).
```

shows all the clauses that have the predicate name `book`.

The Prolog interpreter prints out the values of all variables appearing in the top level goal, but does not print the values of any other variables which are instantiated during the process of satisfying the top level goal. The printing of variables and their values is really just a side-effect of the matching process; it would be better if we had explicit control of the printing of variable values. The system predicate `write/1` gives us this control. There is also a system predicate `read/1` which allows a value to be given to a variable by being typed in by a user. Both these predicates take a single argument that can be instantiated to any term: that is, a variable, a constant, a number or any other atom, or a clause. **Atoms** include characters enclosed in single quotes, such as ‘What book would you like?; this permits strings of characters commencing with capital letters and containing spaces to be written out.

So if we wanted another user to use the `books` program, we could make it interact with the user, having the program ask questions and read the user’s responses. The dialogue might go something like that shown in Figure 13 (program output is printed here in *italics*).

To carry out this dialogue, we need to augment the program by adding a predicate that:

- asks the user what book they would like;

- reads the reply;
- finds where the book can be bought; and
- tells the user.

We might call this predicate `askbook`, and it might look something like the following:

```
askbook:-
    write('What book would you like to buy?'), nl,
    read(Book), nl,
    canbuy(Book, Shop),
    write('You can buy '),
    write(Book),
    write(' at '),
    write(Shop),
    write('.').
```

The system predicates used here are `write/1`, `read/1`, and `nl/0`. `nl/0` has the effect of printing a new line.

When the `read/1` predicate is called, the interpreter prints a new prompt, `|:`, and waits for the user to type a term terminated by a full stop. The value of the variable argument of `read/1` becomes instantiated to the term the user types. This provides a way to input values to a program.

If the book whose name was input does not exist or cannot be bought—that is, if the `canbuy/2` predicate fails—the Prolog interpreter will attempt to backtrack through `nl`, `read`, and `write`. These predicates cannot be re-satisfied (or redone), so an attempt will then be made to redo the predicate `askbook/2`. If there were no other `askbooks/2` clause, the whole program would fail here. However, we can add a second clause that simply writes a message to the user (telling them that the book is unknown) and succeeds:

```
askbook:-
    write('I don''t know where you can buy that book, sorry.').
```

Note the use of the double single quote here to allow us to print a single quote as part of a string that is itself delimited by single quotes.

6.6 Looking for Multiple Responses

We could extend the program further, allowing the user to ask about more books. This can be done by putting `askbook/0` as a subgoal to another clause `go/0`; the predicate `go/0` then becomes the top level predicate, and will do the following:

- call `askbook/0`, asking the user which book they want and responding accordingly;
- offer to find another book;
- check the user's reply;

```

go:-
    askbook, nl,
    write('Would you like another book?'), nl,
    read(Reply), nl,
    check(Reply), nl.

check(Reply):-
    Reply = yes,
    go.

check(Reply):-
    write('I hope I was of some help to you.  Have a nice day.').

```

Figure 14: The Prolog code to allow repeated requests

- if the user types `yes`, then `go/0` will be called again;
- if anything else is typed in response, the program stops.

The corresponding Prolog code looks like that shown in Figure 14.

6.7 Dealing with Arithmetic Operators

Arithmetic operators are another type of system predicate. The operators then enable us to do arithmetic in Prolog are:

`+` `-` `*` `/` add minus times divide

There is also a system predicate that the results of applying these operators, called `is/2`. All of the arithmetic operators can be used as infix (between arguments) or as prefix (like a predicate), for example:

e.g.

```
?- X is 3+7.
```

```
X=10
```

```
?- B is +(2,99).
```

```
B=101
```

```
?- B is 8 + 3.
```

```
B=11
```

Note that you need to put spaces either side of is/2:

```
Dis7+2.
```

```
no
```

```
?- 3 is 2 + 1.
```

```
yes
```

```
?- 4 is 4.
```

```
yes
```

```
?- 4 = 4.
```

```
yes
```

Note also that = and is mean different things: = means will unify with whereas is/2 means evaluates to.

```
?- 4 = 3 + 1.
```

```
no
```

```
?- 4 is 3 + 1.
```

```
yes
```

Prolog must be able to evaluate the right hand side of is/2: If there are variables on the right hand side that are uninstantiated then it will fail.

```
?- S is H+2.
```

```
*** Error: uninstantiated variable in arithmetic expression: _68
```

```
no
```

```
?- X is 3+3.
```

```
    X=6
```

```
yes
```

```
| ?- S is X+3.
```

```
*** Error: uninstantiated variable in arithmetic expression: _68
```

```
no
```

```
?- X is 3+3,Y is X+2.
```

```
    X=6
```

```
    Y=8
```

```
yes
```

Expressions can be complex, in the same way as ordinary arithmetic expressions.

```
| ?- Sum is 4+3-6+2.
```

```
Sum=3
```

```
yes
```

```
| ?- Sum is 3*4.
```

```
Sum=12
```

```
yes
```

```
| ?- Ans is 3*4-5.
```

```
Ans=7
```

```
yes
```

The left hand side is never evaluated:

```
| ?- 3+4 is 7.
```

```
no
```

```
| ?- 3+4 is 3+4.
```

```
no
```

```
| ?- P is 5/8.
```

```
P=0.625
```

```
yes
```

Comparisons of the sort $<$ (less than), $>$ (greater than), \geq (greater than or equal to), \leq (less than or equal to) and \neq (not equal to) can also be made:

```
| ?- 4>3.
```

```
yes
```

```
| ?- 3>4.
```

```
no
```

```
| ?- 5<10.
```

```
yes
```

```
| ?- 5<3+8.
```

```
yes
```

```
| ?- 5*6>9+10.
```

```
yes
```

```
| ?- 5>=5.
```

```
yes
```

```
| ?- 8=<16.
```

```
yes
```

```
| ?- 4\==8.
```

```
yes
```

```
| ?- 4\==4.
```

```
no
```

```
| ?- 3-1+6*4-2<7*8+8-10.
```

```
yes
```

Note that system predicates cannot be traced ⁸:

```
| ?- trace,3-1+6*4<7*8+8-10.
```

```
yes
```

Prefix notation can also be used instead of the more common infix notation:

```
| ?- <(+(-3,1),*(6,4)),+(*(7,8),-(8,10)))>.
```

```
yes
```

```
| ?- T is +(-3,1),*(6,4)).
```

```
T=26
```

```
yes
```

```
| ?- P is +(*(7,8),-(8,10)).
```

```
P=54
```

```
yes
```

```
| ?- 26<54.
```

```
yes
```

6.8 Summary

Whilst this extended program (which will be called `books2`) is an improvement on the basic program we started out with, it is still a little limiting in a number of respects:

1. There is little flexibility in the set of responses that can be given to the question *Would you like another book?*. In particular, note that responses other than `yes` have the following effects:

- `|: y.` fails to match.

⁸see later section on Tracing and Debugging

- |: Yes. is a variable, and so becomes bound to **yes** and succeeds.
- |: No. also becomes bound to **yes** and succeeds.

2. In each case, the `canbuy/2` predicate only finds the first solution, rather than offering all possible solutions to the query.

In the next section we consider how these limitations can be avoided by introducing data structures called **lists** that can be used to represent a collection of objects (for example, all the possible shops that a particular book can be bought from).

6.9 Practical 3: consulting a simple database

6.9.1 The Basic books Program

We start by using the first version of the `books` program. More complicated versions of this program will be used later.

1. Copy the file that contains the `books` program from the `/home/infteach/prolog/code` area.

```
cp /home/infteach/prolog/code/books1.pl books1
```

2. Start Prolog and consult the file:

```
consult(books1).
```

3. List the contents of the database and see if you can work out what sort of questions you could ask. A complete listing is provided in Figure 15.
4. Work out how to ask the following questions:
 - (a) Which publisher publishes *ET*?
 - (b) Which shop stocks Penguin books?
 - (c) Where can you buy *I Claudius*?
 - (d) Is there any publisher who publishes both *Freda the Fire Engine* and *Biggles and Wendy*?
 - (e) Are there two shops who both stock *Of Mice and Men* and *Biggles and Wendy*?
5. Try a few questions of your own. Test the program fully to see exactly how it works, and where it fails.

6.9.2 The books Program—Simple Interactive Version

Once you are quite happy and sure of what is going on with the first version of the `books` program, you should explore the next version.

1. Get out of Prolog. Copy the file that contains the interactive version:

```
cp /home/infteach/prolog/code/books2.pl books2
```

2. Get back into Prolog again and then consult your new file.

This is a more complicated program; a partial listing is provided in Figure 16. Try and see what it does and how it works by asking some questions.

```
stocks(james_thin,sfipubs).
stocks(james_thin,virago).
stocks(james_thin,penguin).
stocks(menzies,sfipubs).
stocks(menzies,sams).
stocks(better_books,penguin).
stocks(better_books,virago).
stocks(edinbooks,sfipubs).
stocks(edinbooks,virago).
stocks(edinbooks,sams).

book(son_of_et,sfipubs).
book(et,sfipubs).
book(i_was_a_teenage_robot,sfipubs).
book(et_rides_again,sfipubs).
book(biggles_and_wendy,virago).
book(freda_the_fire_engine,virago).
book(dict_of_computing,penguin).
book(i_claudius,penguin).
book(of_mice_and_men,penguin).
book(cookbook,sams).

open(menzies).
open(better_books).
open(edinbooks).
closed(james_thin).

canbuy(Book, Shop):-
    book(Book, Publisher),
    stocks(Shop, Publisher),
    open(Shop).
```

Figure 15: The books program, first version

```

stocks(james_thin,sfipubs).
stocks(james_thin,virago).
...
stocks(edinbooks,sams).

book(son_of_et,sfipubs).
book(et,sfipubs).
...
book(cookbook,sams).

open(menzies).
open(better_books).
open(edinbooks).
closed(james_thin).

canbuy(Book,Shop):-
    book(Book,Pub),
    stocks(Shop,Pub),
    open(Shop).

go:-
    askbook, nl,
    write('Would you like another book?'), nl,
    read(Reply), nl,
    check(Reply), nl.

askbook:-
    write('What book would you like to buy?'), nl,
    read(Book), nl,
    canbuy(Book, Shop),
    write('You can buy '),
    write(Book),
    write(' at '),
    write(Shop),
    write('.').

askbook:-
    write('I don''t know where you can buy that book, sorry.').

check(Reply):-
    Reply = yes,
    go.

check(Reply):-
    write('I hope I was of some help to you. Have a nice day.').

```

Figure 16: The second version of the books program

6.9.3 Writing your own Database

Once you have made sure you know exactly what is going on in the `books2` program (and you might want to play a little with the `read` and `write` predicates to check this), then write a database of your own.

Extend it in the same way the `books` program was extended, so that someone else can use it.

For example, you could put your timetable in it, and allow the user to query when you are free. You might want to include information about time and place of lectures, tutorials and practicals.

Test it on your neighbour. Explain how it works and then let them try it.

7 Lists

7.1 Introduction

In this section, we introduce a third version of the `books` program. This version will collect together all solutions to the `canbuy(Book, Shop)` goal and then print out the list of open shops.

In order to understand how this program works, this section is largely devoted to presenting the `list` data structure.

7.2 The `books3` Program

7.2.1 The Example Database

The basic database that the `books3` program uses is as before, and repeated in Figure 17.

7.2.2 The Rules

This version of the program uses the same predicates as before, as shown in Figure 18. Note, however, that the `askbook/0` and `canbuy/2` clauses do more in this version of the program: we have added subgoals that use the new predicates `prlist/1`, `possible/2`, and `filter/2`, and changed the behaviour of `canbuy/2` and `check/1` a little.⁹

7.2.3 How It Works

The basic structure of the `books3` program is as follows:

- ask what book the user wants;
- see where it can be bought:
 - find out its publisher;
 - see which shops stock this publisher (make a list of these);
 - make a new list of the open shops that stock that publisher; and
 - print out the list.
- ask the user if she would like another book:
 - if the reply is positive, do the whole thing again;
 - otherwise, stop.

⁹The `possible/2` and `filter/2` predicates will not be discussed here.

```
stocks(james_thin, sfipubs).
stocks(james_thin, virago).
stocks(james_thin, penguin).
stocks(menzies, sfipubs).
stocks(menzies, sams).
stocks(better_books, penguin).
stocks(better_books, virago).
stocks(edinbooks, sfipubs).
stocks(edinbooks, virago).
stocks(edinbooks, sams).

book(son_of_et, sfipubs).
book(et, sfipubs).
book(i_was_a_teenage_robot, sfipubs).
book(et_rides_again, sfipubs).
book(biggles_and_wendy, virago).
book(freda_the_fire_engine, virago).
book(dict_of_computing, penguin).
book(i_claudius, penguin).
book(of_mice_and_men, penguin).
book(cookbook, sams).

open(menzies).
open(better_books).
open(edinbooks).
closed(james_thin).
```

Figure 17: The books3 database

```
go:-
    askbook, nl,
    write('Would you like another book?'), nl,
    read(Reply), nl,
    check(Reply), nl.

askbook:-
    write('What book would you like to buy?'), nl,
    read(Book), nl,
    canbuy(Book, Shops),
    write('You can buy '),
    write(Book),
    write(' at '),
    prlist(Shops),
    write('.').

askbook:-
    write('I don''t know where you can buy that book, sorry.').

canbuy(Book, OpenShops):-
    book(Book, Publisher),
    possible(Publisher, ListOfShops),
    filter(ListOfShops, OpenShops).

check(Reply):-
    member(Reply, ['Yes', yes, 'Y', y]),
    go.

check(Reply):-
    write('I hope I was of some help to you. Have a nice day.').
```

Figure 18: The rules used in the books3

7.2.4 The member Predicate

In our new rules, notice that `check/1` includes the goal

```
member(Reply, ['Yes',yes,'Y',y])
```

The `member/2` predicate is used here to check whether the `Reply` is one of a list of possibilities. More generally, `member/2` tests whether an element is a member of a list; the Prolog code for `member` is as follows:

```
member(X, [X|_]).

member(X, [_|Rest]):-
    member(X,Rest).
```

10

7.2.5 The prlist Predicate

The `prlist/1` predicate prints out each element of a list on a separate line. The Prolog code is as follows:

```
prlist([]).

prlist([Head|Rest]):-
    nl, write(Head),
    prlist(Rest).
```

7.3 Lists

7.3.1 Basics

If we need to put a number of items together in one structure, perhaps in order to manipulate them as a single structure, we can build a data structure called a **list**. An example would be a list of all the shops in the `books` program:

```
[better_books, menzies, edinbooks, james_thin]
```

A list is defined as follows:

A list is an ordered sequence of elements.

¹⁰Note the use here of the variable `_`. This is referred to as the anonymous variable and is used in place of a named variable (such as `A` or `_73`) in cases where there is no further need to refer to this variable at any point later in the clause. In this case, once we have matched the head of the list to the element in the first argument, we no longer need the rest of the list, so use `_` to denote it.

List	Head	Tail
[a, b, c, d]	a	[b, c, d]
[a]	a	[]
[]	fails	fails
[[the, cat], sat]	[the, cat]	[sat]
[the, cat]	the	[cat]
[the, [cat, sat]]	the	[[cat, sat]]
[the, [cat, sat], down]	the	[[cat, sat], down]
[X, Y, Z]	X	[Y, Z]

Figure 19: Some lists and their heads and tails

Lists can be used to represent practically any kind of structure. In fact, lists are such a general data structure that computing languages have been built based entirely upon them; this is true of the language Lisp, for example.

A list can be either

- an empty list, written as ‘[]’; or
- a list containing one or more elements, each separated by commas and all enclosed in square brackets (‘[’ and ‘]’).

Here are some examples of lists:

- []
- [a, b]
- [a, X, j]

7.4 Manipulating Lists

A list can be manipulated by splitting it into its **head** and its **tail**. The head of the list is the first element of the list, and the tail of the list is the rest of the list. Further examples of lists, together with their heads and tails, are shown in Figure 19.

In order to split lists into their heads and tails, we use a **list destructor**. This is represented by the character ‘|’. So, the list with head X and tail Y is represented as

[X|Y]

7.4.1 Matching Lists

By using the list destructor together with Prolog’s unification (Prolog’s matching mechanism), we can easily split any list in order to access any of its elements (or to transform lists in some way, or build new ones).

If we match the two lists

```
[X|Y]
```

and

```
[the, little, dog]
```

then X is instantiated to `the`, and Y is instantiated to `[little, dog]`. Some other examples of attempts to unify lists are shown in Figure 20

Element order is important:

- `[foo,bar,baz]` is not the same as `[baz,bar,foo]`

7.4.2 Constructing and Destructing Lists

The basic approach:

- To take a list apart, split the list into the first element and the rest of the list.
- To construct a list from an element and a list, insert the element at the front of the list.

List Destruction:

```
[A|B] = [1,2,3,4]
```

```
A = 1
```

```
B = [2,3,4]
```

The first element is the **head** of the list; the remainder is the **tail** of the list.

List Construction:

- Take a variable bound to a list: for example

```
OldList = [happy,sad]
```

- Add the new element to the front:

```
NewList = [grumpy|OldList]
```

Bigger Chunks:

- You can always directly access any number of elements at the head of a list.

```

?- [john,june,A] = [X,Y,tom].    A=tom X=john Y=june    yes
   (simple match of variables and constants)

?- [H|T] = [c,b,a].              H=c T=[b,a]          yes
   (splits list into head and tail)

?- [H|T] = [].                  no
   (the empty list cannot be split by the list destructor '|')

?- [A|[B,C]] = [c,b,a].          A=c B=b C=a        yes
   ([A|[B,C]] same thing as [A,B,C])

?- [a,X] = [X,b].               no
   (X cannot unify with two different constants a and b)

?- [sue,[dick,wendy],P] = [P,Q,R].    P = sue, Q = [dick,wendy],
   (P and R share)                R = sue          yes

?- [X|Y] = [a(b,c),b,c].          X = a(b,c), Y = [b,c]  yes

?- [[X],Y] = [a,b].             no
   (constant 'a' cannot match to list [X])

?- [a|X] = [A,B,y]              A = a, X = [B,y]     yes

?- [fred|[]] = [X].             X = fred            yes
   ([fred|[]] same as [fred])

?- [a,b,X,c] = [a,b,Y].         no
   (different length lists)

?- [H|T] = [tom,dick,mary,fred]. H= tom T = [dick,mary,fred]  yes

?- [[sue,tom],[10,9]] = [names,ages]. no
   (constants don't match lists)

```

Figure 20: Some examples of list matching

- To add `grumpy` and related to `OldList`:

```
NewList = [grumpy, elated|OldList]
```

- To remove three elements: suppose `OldList` is bound to a list of three or more elements, then

```
OldList = [One,Two,Three|Remainder]
```

7.5 Summary

In the next section, we go on to look at list manipulation in more detail.

7.6 Exercises

Question 7.1 How many elements are there in each of the following list structures?

e.g. the length of `[foo,bar,baz]` is 3

the length of `[foo,1,[a(X)],[1],[foo,bar]]` is 5

7.1a `[a,[a,[a,[a]]]]`

7.1b `[1,2,3,1,2,3,1,2,3]`

7.1c `[a(X),b(Y,Z),c,X]`

7.1d `[[sum(1,2)],[sum(3,4)],[sum(4,6)]]`

7.1e `[c,[d,[x]],[f(s)],[r,h,a(t)],[[a]]]`

Question 7.2 The predicate `=` takes 2 arguments and tries to unify them. For each of the following:

- specify whether the query submitted to Prolog succeeds or fails;
- if it succeeds, specify what is assigned to any variables;
- if it fails, explain why it fails.

e.g. `?- [H|T] = [1,2,3].` yes `H=1, T=[2,3].`

`?- [a, b] = [c, A].` no because a and c are constants
and constants cannot unify
with each other.

- ?- [A,john,C] = [jim,B,tom]. yes A=jim, B=john, C=tom.
- 7.2a ?- [A,B,C,D] = [a,b,c].
- 7.2b ?- bar([1,2,3]) = bar(A).
- 7.2c ?- [X, []] = [X].
- 7.2d ?- [X,Y|Z] = [a,b,c,d].
- 7.2e ?- [1,X,X] = [A,A,2].
- 7.2f ?- likes(Y,a) = likes(X,Y).
- 7.2g ?- [foo(a,b),a,b] = [X|Y].
- 7.2h ?- [b,Y] = [Y,a].
- 7.2i ?- [H|T] = [red,blue,b(X,Y)].
- 7.2j ?- [1,[a,b],2] = [[A,B],X,Y].
- 7.2k ?- test(a,L) = test(E,[b,c,d]).
- 7.2l ?- [[a,[b]],C] = [C,D].
- 7.2m ?- [fred|T]=[H|[sue,john]].

Question 7.3 Imagine that this program is consulted by the Prolog interpreter:

```
foo([], []).
```

```
foo([H|T], [X|Y]):-
    H = X,
    foo(T,Y).
```

[Note: this program tests if two lists unify by testing if the heads unify then recursing on the tails]

What will be the outcome of each of the following queries?

7.3a ?- foo([a,b,c], A).

7.3b ?- foo([c,a,t], [c,u,t]).

7.3c ?- foo(X, [b,o,o]).

7.3d ?- foo([p|L], [F|[a,b]]).

7.3e ?- foo([X,Y], [d,o,g]).

7.7 Practical 4: Simple Sentence Generation

1. The program shown in Figure 21 is stored in the file

```
/home/infteach/prolog/code/sent1
```

Copy it to your area.

2. Get into Prolog. Consult the file `sent1` (or whatever you have called it in your area):

```
| ?- consult(sent1).
```

List it. Predict what the result will be if you type the following goal:

```
| ?- sentence.
```

Play around with it and check you understand how it works.

3. Using the editor, modify the program to be as shown in Figure 22. It's a good idea to copy the file and alter the copy; that way you still have the original in case something goes wrong.

What is the outcome of the following goal?

```
| ?- sentence(S).
```

Explain how this solution is achieved. With the same goal, redo to find all solutions.

Trace through how this works by typing:

```
| ?- trace, sentence(S).
```

and then stepping through the program by pressing `<RETURN>` in response to each `'?'`.

4. The `make` subgoal of the `sentence` clause, and the `make` clause itself, are actually redundant in this example. Instead of unifying `X` and `[N1,V1,N2]` in the `make` clause, it can be done in the head of the `sentence` clause itself, as shown in Figure 23.

Edit your program to be the same as this version, and test it out. Trace it as before to see how it works.

5. In these examples, lists of items are printed. Write a higher level clause `go` that uses the `sentence` clause and another predicate, `prlist`. The `prlist` predicate should write out all items in the list that is unified with the variable `S` in the solution of the goal `sentence(S)`.

Modify your version of `prlist` to deal with embedded lists; so, for example, you should see behaviour like the following:

```
| ?- prlist([a,b,[c,d],e,[f,[g]],h]).
```

```
a b c d e f g h
yes
```

```
sentence:-
    noun(N1), write(' '), write(N1),
    verb(V1), write(' '), write(V1),
    noun(N2), write(' '), write(N2).

noun(fred).
noun(beer).
noun(doris).
noun(gin).

verb(likes).
verb(drinks).
```

Figure 21: A simple sentence generator

```
sentence(X):-
    noun(N1),
    verb(V1),
    noun(N2),
    make([N1,V1,N2],X).

make(Anything,Anything).

noun(fred).
noun(beer).
noun(doris).
noun(gin).

verb(likes).
verb(drinks).
```

Figure 22: A revised sentence generator

```
sentence([N1,V1,N2]):-  
    noun(N1),  
    verb(V1),  
    noun(N2).  
  
noun(fred).  
noun(beer).  
noun(doris).  
noun(gin).  
  
verb(likes).  
verb(drinks).
```

Figure 23: A further revised sentence generator

Hint: to `prlist` a list, `prlist` the head of the list then `prlist` the tail of the list (i.e., instead of recursing on the tail alone, recurse on both the head and the tail of the list).

6. Expand the vocabulary of the program and experiment with it further.

8 Manipulating lists

8.1 Introduction

In this section we will look more closely at manipulating lists.

8.2 The predicate `member/2`

8.2.1 Building the predicate

Suppose we have a list of names like the following:

```
[fred, john, ann, mary].
```

and we want to know if a given person is in this list.

We first ask if the person has the same name as the head of the list:

- if so, we have succeeded;
- if not, then we see if the person is one of the rest of the list, that is, if the person is in the tail of the list.
- So, we take the tail of the list and ask if the person has the same name as the head of this list:
 - if so, we have succeeded;
 - if not, then we see if the person is in the tail of this list.
 - So, we take the tail of the list ...

And so on, until we run out of list and fail to find the person.

So we need to write a set of clauses to test membership of a list. We will call it the `member` predicate. It will need two arguments, the item to be looked for and the list to look for it in. We'll refer to this predicate as `member/2`.

In this predicate,

- either the item will be the head of the list;
- or it will be in the tail of the list;
- or there will be no match and it will fail.

So we need to consider these three possibilities, or cases. The first clause we can specify, in English, as:

X is a member of a list if X is the same as the head of the list.

The second clause will be:

X is a member of a list if it is a member of the tail of the list.

Consider how we write these in Prolog. The first argument of the `member/2` predicate is the element to be found, and the second is the list:

```
member(X, AList):-
```

But we have to somehow get at the head of this list. We use the list destructor and unification to do this for us:

```
member(X, [H|T]):-
```

So, the first case becomes

```
member(X, [H|T]):-  
    X = H.
```

and the second becomes:

```
member(X, [H|T]):-  
    X \== H,  
    member(X, T).
```

Note that here `member/2` has the subgoal `member/2`, so by definition it is **recursive**: it calls itself.

8.2.2 Cleaning Up

We have some redundancy in these two clauses.

Let's take the first clause first:

```
1:    member(X, [H|T]):-  
        X = H.
```

We want to test whether or not X and H unify. It is easier to test that by calling them both the same; that is, by making them the same variable. Note that doing this makes the subgoal redundant:

```
1:    member(X, [X|T]).
```

If this succeeds then we never test the second clause. Put another way, if we reach the second clause, the goal `X = H` must have failed. So by the time we get to the second clause, we know that `X \== H` and so we do not need to test it again. The second clause then becomes:

```
2:    member(X, [H|T]):-  
        member(X, T).
```

8.2.3 Recursion in List Processing

Any program that calls itself as a subgoal is **recursive**. A lot of other list processing programs are recursive. They have the following general pattern:

- split the list;
- do something to the head of the list (test it or process it); and
- recurse on the tail of the list.

Or, in a schematic Prolog form:

```
recpred([H|T]):-  
    test(H),  
    recpred(T).
```

In recursive predicates there are usually at least two cases:

1. the **base case** or **boundary condition**, which stops the recursion (eventually): this could be when some condition is true (for example, when the name sought is the head of the list, or when we are left with the empty list);
2. the **recursive** case, which has the goal that recurs, often calling subgoals with shorter and shorter lists (the tail of original list, the tail of the tail of the original list, and so on) as arguments.

An example of a recursive program we have already encountered is `likes/2`, in the case where some person *likes* another if both the person and the other *likes* the same drink.

8.2.4 And/Or Trees for Recursive Predicates

Here is the predicate `member/2` with an example goal and the AND/OR tree to illustrate the execution of this goal:

First, the `member/2` predicate:

```
1:    member(X, [X|T]).  
2:    member(X, [H|T]):-  
        member(X, T).
```

Suppose we present Prolog with the following goal:

```
| ?- member(ann, [fred, john, ann, mary]).
```

The behaviour of Prolog that follows is represented in the AND/OR tree in Figure 24.




Figure 24: An AND/OR tree representation of `member`

8.3 Other List Processing Predicates

8.3.1 Printing a list of elements

Suppose we have a list `[a, b, c, d]`, and we want to print it out. We can define a predicate `prlist/1`, which takes a list as its argument and simply uses `write/1` to write it out:

```
prlist(X):-
    write(X).
```

For our example list above, this would produce the output

```
| ?- prlist([a, b, c, d]).
[a,b,c,d]
yes
| ?-
```

However, suppose we want to print each element of the list on a separate line. This means we have to print the elements one at a time. We can define a recursive procedure to do this:

To `prlist` a list of elements:

- write the head of the list; and
- `prlist` the tail of the list.

The Prolog code to do this is as follows:

```
prlist([H|T]):-
    write(H), nl,
    prlist(T).
```

We also need to know when to stop, of course. The printing should stop when we have no more elements to write, i.e., when the list to be `prlisted` is the empty list. So, we also need the following clause:

```
prlist([]).
```

The full program is then as follows:

```
1:    prlist([H|T]):-
        write(H), nl,
        prlist(T).
2:    prlist([]).
```

8.3.2 Checking that no element of a list of letters is a consonant

Suppose we want to be able to test each of the elements of a list to check that none of them are consonants. We can define a predicate `no_cons/1`, which has the same basic pattern as before: we test or process the head of the list, and then recurse on the tail of the list. The empty list provides the base case.

The test we'll use here is actually whether the element is a vowel, rather than whether it isn't a consonant. The code is then as follows; note that we have to include facts that tell Prolog which characters are vowels.

```
1:    no_cons([]).
2:    no_cons([H|T]):-
        vowel(H),
        no_cons(T).
3:    vowel(a).
4:    vowel(e).
5:    vowel(i).
6:    vowel(o).
7:    vowel(u).
```

Here are some example goals presented to Prolog with this database loaded:

```
| ?- no_cons([a,e,e,o,u]).
yes
| ?- no_cons([a,r,e]).
no
| ?-
```

```
max([H|T], Answer):-
    max(T, H, Answer).

max([], Answer, Answer).
max([H|T], Temp, Answer):-
    H > Temp,
    max(T, H, Answer).
max([H|T], Temp, Answer):-
    max(T, Temp, Answer).
```

Figure 25: The max predicate

8.4 More List Processing Predicates

8.4.1 Finding the maximum of a list of numbers

The predicate `max/2` has two arguments: the first is a list of numbers, and the second is the maximum of that list. If the predicate is called with the second argument uninstantiated, then this argument will become instantiated to the maximum of the list; this might be thought of as returning a single element as the result of processing the complete list.

As before, the head of the list is processed, and the program then recurses on the rest of the list. The Prolog code is shown in Figure 25.

Note that the predicate `max/3` is called by the predicate `max/2`. The additional middle argument is the current (temporary) maximum, instantiated each time a higher value is found as the program recurses down the list. Initially this is set to the head of the list.

`max/3` has three cases:

- The first case of `max/3` is the base case: if the empty list is being processed, then the current maximum is the final one.
- In the second case, if the head of the list is greater than the current maximum then it becomes the new current maximum, and the rest of the list is recursed on.
- In the third case, if the head of the list is not greater than the current maximum then the current maximum stays the same, and again the rest of the list is recursed on.

8.4.2 Building new list structures

As well as recursing down a list, testing all the elements (as we did in `member/2` and `no_cons/1`), or returning a single element (as we did in `max/2`), we may want to process each element and build a completely new list data structure.

Suppose, for example, we want to generate a sentence. We could simply pattern match for each word category (to begin with, *noun* and *verb*) and write the resulting word: such a generator might look like that shown in Figure 26.

```

sentence:-
    noun(N1), write(' '), write(N1),
    verb(V1), write(' '), write(V1),
    noun(N2), write(' '), write(N2).

noun(fred).
noun(beer).
noun(doris).
noun(gin).

verb(likes).
verb(drinks).

```

Figure 26: A simple sentence generator

Note that this code provides no way of saving the sentence generated as a whole. To get around this, we could make each new word a successive element of a new list. We can do this in the head of the `sentence` clause directly:

```

sentence([N1,V1,N2]):-
    noun(N1),
    verb(V1),
    noun(N2).

```

In a later section, we will look at the special mechanism Prolog provides for writing grammar rules.

8.5 Exercises

Question 8.1 The predicate `member/2` succeeds if the first argument matches an element of the list represented by the second argument.

e.g. `?- member(1,[2,3,1,4]).` `yes`

`member/2` is defined as:

1. `member(E1,[E1|T]).`
2. `member(E1,[H|T]):-`
`member(E1,T).`

For each of the following:

- specify whether the query submitted to Prolog succeeds or fails;
- if it succeeds, specify what is assigned to any variables.

8.1a ?- member(a,[c,a,b]).


8.1b ?- member(a,[d,o,g]).

8.1c ?- member(one,[one,three,one,four]).

8.1d ?- member(X,[c,a,t]).

8.1e ?- member(tom,[[jo,alan],[tom,anne]]).

8.1f Complete the AND/OR tree below which represents the execution of the query:



Question 8.2 The predicate no_cons/1 succeeds if all elements of the list represented by the one argument are vowels (as specified by vowel/1).

e.g. ?- no_cons([a,e,i]).
yes

? no_cons([a,b,c]).
no

no_cons/1 is defined as:

1. no_cons([]).
2. no_cons([H|T]):-vowel(H),no_cons(T).
3. vowel(a).
4. vowel(e).
5. vowel(i).
6. vowel(o).
7. vowel(u).


For each of the following:

- specify whether the query submitted to Prolog succeeds or fails;
- if it succeeds, specify what is assigned to any variables.

8.2a ?- no_cons([a,a,e,i]).

8.2b ?- no_cons([A,e,e]).

8.2c Complete the AND/OR tree below which represents the execution of the query:




Question 8.3 The predicate max/2 succeeds if the first argument is a list of numbers and second argument unifies with the maximum value in that list.


e.g. ?- max([2,4,6,8],M).
 M = 8
 yes

max/2 is defined as:

0. max([H|T],Max):-
 max(T,H,Max).
1. max([H|T],Temp,Max):-
 H>Temp,
 max(T,H,Max).
2. max([H|T],Temp,Max):-
 max(T,Temp,Max).
3. max([],Finalmax,Finalmax).

8.3 Complete the AND/OR tree below which represents the execution of the query:





Question 8.4 The predicate `prlist/1` is supposed to write out the elements of a list structure, regardless of the levels of embedding that are present in the list.

e.g. intended behaviour:

```
?- prlist([a,b,[c,d,e],f,[g]]).
abcdefg
yes
```

Instead, the predicate as defined below has the following behaviour:

```
?- prlist([a,b]).
ab[]
yes

?- prlist([a,b,[c,d],e]).
abcd[]e[]
yes

prlist/2 is defined as:
1. prlist([H|T]):-
    prlist(H),
    prlist(T).
2. prlist(X):-
    write(X).
```

Explain why this predicate produces this behaviour (instead of the intended behaviour), using an AND/OR tree or a trace in your explanation.

Question 8.5 The predicate `checkvowels` takes a list representing a word, checks each letter to see whether it is a vowel, and if it is it writes out the vowel.

```
checkvowels([H|T]):-
    vowel(H),
    write(H), nl,
    checkvowels(T).
```

```
checkvowels([H|T]):-
    checkvowels(T).
```

```
checkvowels([]).
```

```
vowel(a).
vowel(e).
vowel(i).
vowel(o).
vowel(u).
```

```
e.g.    ?- checkvowels([c,a,t,i]).
        a
        i
        yes
```

8.5a Using this as a model, write a predicate `results/1` that takes a list of names, checks whether each is a pass (using a predicate that you also must define, `pass/1`) and writes out the names if they pass.

For example, the following query:

```
?- results([tom,bob,sue,jane]).
```

should give the output:

```
bob
sue
yes
```

8.5b Modify this program so that instead of 'writing out' the name of each person who passes it should output a list of them. (You may need to read ahead to answer this part).

```
?- results2([tom,bob,sue,jane],Passlist).
Passlist = [bob,sue]
yes
```

9 How Programs Work

9.1 Tracing

We can follow through the execution of a goal in Prolog by **tracing** it. The trace will show:

- which goal is currently being called, with what arguments;
- whether the goal succeeds or has to be redone (i.e., whether another match has to be looked for);
- whether the goal fails; and
- if the goal succeeds, whether other clauses might also match.

For example, given the predicate `no_cons/2`, the goal:

```
| ?- no_cons([a,e,o]).
```

can be traced by calling the conjunction of the goal `trace` and the goal itself; the code for `no_cons/2` and the results of tracing are shown in Figure 27.

Similarly we can trace the goal:

```
| ?- trace, max([7,3,9,2,6], Max).
```

giving the code and output as shown in Figure 28.

Some things to note here:

1. The number to the left of the goal indicates the level in the tree from which the goal is being called (the top level goal being 0). This makes it easier to see which goals are subgoals of the same clause, and also which are recursive goals.
2. The leftmost number is the number of the goal that is being called: each new goal that is created is given a number which is incremented for the next new goal. This makes it easier to see exactly which goal is being called, being redone, failing, or succeeding.

9.2 The Byrd Box Model of Execution

The model on which the tracing is based is known as the 'Byrd Box model of Execution'.

A program trace:

```
| ?- listing(parent).
```

```
parent(a, b).  
parent(c, d).
```

```
no_cons([]).

no_cons([H|T]):-
    vowel(H),
    no_cons(T).

| ?- trace, no_cons([a,e,o]).
{The debugger will first creep -- showing everything (trace)}
  1 1 Call: no_cons([a,e,o]) ?
  2 2 Call: vowel(a) ?
  2 2 Exit: vowel(a) ?
  3 2 Call: no_cons([e,o]) ?
  4 3 Call: vowel(e) ?
  4 3 Exit: vowel(e) ?
  5 3 Call: no_cons([o]) ?
  6 4 Call: vowel(o) ?
  6 4 Exit: vowel(o) ?
  7 4 Call: no_cons([]) ?
  7 4 Exit: no_cons([]) ?
  5 3 Exit: no_cons([o]) ?
  3 2 Exit: no_cons([e,o]) ?
  1 1 Exit: no_cons([a,e,o]) ?

yes
{trace}
| ?-
```

Figure 27: Tracing the behaviour of the predicate no_cons

```

max([H|T], Answer):-
    max(T, H, Answer).

max([], Answer, Answer).
max([H|T], Temp, Answer):-
    H > Temp,
    max(T, H, Answer).
max([H|T], Temp, Answer):-
    max(T, Temp, Answer).

| ?- trace, max([7,3,9,2,6], Max).
{The debugger will first creep -- showing everything (trace)}
  1 1 Call: max([7,3,9,2,6],_112) ?
  2 2 Call: max([3,9,2,6],7,_112) ?
  3 3 Call: 3>7 ?
  3 3 Fail: 3>7 ?
  3 3 Call: max([9,2,6],7,_112) ?
  4 4 Call: 9>7 ?
  4 4 Exit: 9>7 ?
  5 4 Call: max([2,6],9,_112) ?
  6 5 Call: 2>9 ?
  6 5 Fail: 2>9 ?
  6 5 Call: max([6],9,_112) ?
  7 6 Call: 6>9 ?
  7 6 Fail: 6>9 ?
  7 6 Call: max([],9,_112) ?
  7 6 Exit: max([],9,9) ?
  6 5 Exit: max([6],9,9) ?
  5 4 Exit: max([2,6],9,9) ?
  3 3 Exit: max([9,2,6],7,9) ?
  2 2 Exit: max([3,9,2,6],7,9) ?
  1 1 Exit: max([7,3,9,2,6],9) ?

Max = 9 ?

yes
{trace}
| ?-

```

Figure 28: Tracing the behaviour of max

```

yes
| ?- trace, parent(X,Y), X = f.
{The debugger will first creep
-- showing everything (trace)}
  1 1 Call: parent(_44,_58) ?
  1 1 Exit: parent(a,b) ?
  2 1 Call: a=f ?
  2 1 Fail: a=f ?
  1 1 Redo: parent(a,b) ?
  1 1 Exit: parent(c,d) ?
  2 1 Call: c=f ?
  2 1 Fail: c=f ?
  1 1 Redo: parent(c,d) ?
  1 1 Fail: parent(_44,_58) ?

no
{trace}
| ?-

```


- Each box represents one invocation of a **procedure**.
- Use nested boxes for the bodies of rules.
- Each box has four ports:

Call: used first time we look for a solution

Exit: used if the procedure succeeds

Redo: used when backtracking

Fail: used if we can't satisfy the goal



Not all systems provide tracers that use all four ports, i.e. some tracers do not explicitly use the 'redo' port, but use 'call' again when backtracking.

9.3 Debugging using the Tracer

spy(predicate_name) Mark any clause with the given predicate_name as “spyable”. Does not work for built-in predicates; can take a list of predicates as argument.

debug If a spied predicate is encountered, switch on the tracer.

nodebug Remove all spypoints. The tracer will therefore not be invoked.

nospy(predicate_name) Undo the effect of spy—*i.e.* remove the spy point.

debugging Shows which predicates are marked for spying plus some other information.

trace Switches on the tracer.

notrace Switches the tracer off. Does not remove spypoints.

Debugging options:

<cr>	creep	c	creep
l	leap	s	skip
r	retry	r <i>	retry i
d	display	p	print
w	write		
g	ancestors	g <n>	ancestors n
n	nodebug	=	debugging
+	spy this	-	nospy this
a	abort	b	break
@	command	u	unify
<	reset printdepth	< <n>	set printdepth
^	reset subterm	^ <n>	set subterm
?	help	h	help

Useful Debugging Options:

creep: This is the single stepping command. Use `<RETURN>` to **creep**. The tracer will move on to the next port.

skip: This moves from the `CALL` or `REDO` ports to the `EXIT` or `FAIL` ports. If one of the subgoals has a spypoint then the tracer will ignore it.

leap: Go from the current port to the next port of a spied predicate.

Tracing in SICStus Prolog, given the following program:

```
son(A, B) :-
    parent(B, A),
    male(A).
```

```
daughter(A, B) :-
```

```

    parent(B, A),
    female(A).

parent(A, B) :-
    father(A, B).
parent(A, B) :-
    mother(A, B).

male(tom).
male(jim).
male(cecil).
male(fred).

female(mary).
female(jane).
female(sue).

father(tom, jim).
father(tom, sue).
father(fred, cecil).

mother(mary, tom).
mother(mary, jane).
mother(mary, fred).

| ?- trace,daughter(X,Y).
{The debugger will first creep -- showing everything (trace)}
  1 1 Call: daughter(_51,_67) ?
  2 2 Call: parent(_67,_51) ? s
  2 2 Exit: parent(tom,jim) ?
  4 2 Call: female(jim) ? g
Ancestors:
  1 1 daughter(jim,tom)
  4 2 Call: female(jim) ?
  4 2 Fail: female(jim) ?
  2 2 Redo: parent(tom,jim) ? s
  2 2 Exit: parent(tom,sue) ?
  4 2 Call: female(sue) ? g
Ancestors:
  1 1 daughter(sue,tom)
  4 2 Call: female(sue) ? s
  4 2 Exit: female(sue) ?
  1 1 Exit: daughter(sue,tom) ?

X = sue,
Y = tom ?

yes

```



```
{trace}
| ?-
```

We have 'skipped' each time that we called `parent/2` so we do not see whether it is satisfied by using the `father/2` or `mother/2` rule: we only see the outcome of the subgoal. We have also asked to see the ancestors of the subgoal `female(jim)` i.e. what it was the immediate subgoal of.

We might choose to trace a particular predicate such as `father/2` by placing a spy point on it. Whenever we 'leap' in the tracing thereafter, the tracer will jump to the next call to this goal.

```
| ?- spy(father).
{Spypoint placed on user:father/2}
{The debugger will first leap -- showing spypoints (debug)}
```

```
yes
{debug}
| ?- daughter(X,Y).
+ 3 3 Call: father(_81,_65) ?
+ 3 3 Exit: father(tom,jim) ?
  2 2 Exit: parent(tom,jim) ?
  4 2 Call: female(jim) ? s
  4 2 Fail: female(jim) ?
  2 2 Redo: parent(tom,jim) ? l
+ 3 3 Redo: father(tom,jim) ?
+ 3 3 Exit: father(tom,sue) ?
  2 2 Exit: parent(tom,sue) ?
  4 2 Call: female(sue) ? l
```

```
X = sue,
Y = tom ?
```

```
yes
{debug}
```

9.4 Loading Files

`consult` or `consultall` can be used for loading files into the prolog interpreter. If the filename has punctuation characters in it, remember to enclose it in single quotes. More than one file may be consulted at once.

- ```
| ?- consult(parents).
 {consulting parents...}
 {parents consulted, 10 msec 287 bytes}
```

```
yes
| ?-
```

- | ?- [parents].
- | ?- consult('/u/ai/s2/ai2/aifoo/program').
- | ?- consult('foo.pl').
- | ?- consult([foo,baz,'foobaz.pl']).
- | ?- [foo,baz].

Avoid splitting predicate definitions between files. For example, if you define part of your family program in one file and part in another (perhaps you decided to put different families in different files?) the following may happen:

```
| ?- consult(file1).
{consulting file1...}
{file1 consulted, 10 msec 287 bytes}
```

yes

```
| ?- consult(file2).
{consulting file2...}
```

The procedure parent/2 is being redefined.

Old file: file1

New file: file2

Do you really want to redefine it? (y, n, p, or ?) ?

y    redefine this procedure

n    don't redefine this procedure

p    redefine this procedure and don't ask again

?    print this information

(y, n, p, or ?)

## 9.5 Some common mistakes

1. | ?- member(a, [a,b,c]).

```
** , | or] expected in list **
```

```
member (a , [a , b , c
```

```
** here **
```

```
) .
```

Leaving off the closing list bracket causes a syntax error.

2. | ?- parent(a,b) X = f.

```
** variable follows expression **
```

```
parent (a , b)
```

```
** here **
```

```
X = f .
```

Subgoals must be separated by commas.

```
3. | ?- parent(a b),X = f.
```

```
** atom follows expression **
```

```
parent (a
** here **
b) , X = f .
```

Arguments must also be seperated by commas.

```
4. | ?- parent(a, b, X = f.
```

```
** , or) expected in arguments **
```

```
parent (a , b , X = f
** here **
```

```
.
```

Closing predicate bracket omitted.

```
5. | ?- parent (a,b).
```

```
** bracket follows expression **
```

```
parent
** here **
(a , b) .
| ?-
```

Extra space between the predicate name and the opening bracket.

## 9.6 Summary

You should be able to:

- model the behaviour of Prolog programs using the Byrd Box Model
- be able to use the basic functionality of Prolog's debugging tools
- be able to identify common mistakes in typing Prolog queries

## 10 Further list processing predicates

### 10.1 Changing one sentence into another: the predicate `alter/2`

Suppose you want to write a simple predicate that enables you to alter an input sentence to get a new output sentence (Eliza style). An example of the sort of dialogue produced might be:

```
You: you are a computer
Prolog: i am not a computer

You: do you speak french
Prolog: no i speak german
```

How do we go about this? First we break the task down into steps.

1. Accept the sentence typed in;
2. If there are any *you*'s change them to *i*'s
3. and change *are* to *am not*
4. and change *french* to *german*
5. and change *do* to *no*

[this may lead to some obvious problems in other examples, but we will ignore those for now]

The predicate we will use will be called `alter/2`. It will need two arguments, both of which will be lists.

An example goal might be:

```
| ?- alter([do,you,know,french],Rep).
Rep=[no,i,know,german]
```

How is `alter/2` defined? Think of cases to be dealt with:

- the list to be altered
- the empty list

Take the latter:

Alter the empty list to the empty list

```
alter([],[]).
```

This will give us the boundary condition and will stop the recursion.

What about the rest of the list to be altered?

- change one word at a time (or leave it if not to be changed)
- build a new list of changed and unchanged words (=reply)

We do this by changing the head of the list and then recursing on the tail, building a new list as we go and stopping when we run out of list.

1.
  - Change the head of the input list (represented by the first argument) into another word and
  - let the head of the output list be the same as that word (by matching)
2.
  - Alter the tail of the input list and
  - let the tail of the output list be the same as the altered tail
3. If the end of the input list is reached then there is no more to do.

We now have to deal with changing one word into another. We will define a predicate `change/2` that will take two arguments: the element to be changed, and the element that it is to be changed to:

```
change(you,i).
change(are,[am,not]).
change(french,german).
.....
.....
change(X,X).
```

As these are all single list elements, any replacement of more than one word will have to be a sub-list. Note the *catchall* in the last clause of `change/2`. This will take care of all the words that we don't want changed.

So, the two clauses that make up the predicate `alter/2` are:

```
1: alter([],[]).
2: alter([H|T],[X|Y]):-
 change(H,X),
 alter(T,Y).
```

If we now put the whole program together with predicates for *changing* and *altering* we get the following:

```
1: alter([],[]).
```

---

```

| ?- trace,alter([i,like,your,shirt],P).
{The debugger will first creep -- showing everything (trace)}
 1 1 Call: alter([i,like,your,shirt],_99) ?
 2 2 Call: change(i,_237) ?
 2 2 Exit: change(i,you) ?
 3 2 Call: alter([like,your,shirt],_238) ?
 4 3 Call: change(like,_505) ?
 4 3 Exit: change(like,like) ?
 5 3 Call: alter([your,shirt],_506) ?
 6 4 Call: change(your,_772) ?
 6 4 Exit: change(your,my) ?
 7 4 Call: alter([shirt],_773) ?
 8 5 Call: change(shirt,_1038) ?
 8 5 Exit: change(shirt,shirt) ?
 9 5 Call: alter([],_1039) ?
 9 5 Exit: alter([],[]) ?
 7 4 Exit: alter([shirt],[shirt]) ?
 5 3 Exit: alter([your,shirt],[my,shirt]) ?
 3 2 Exit: alter([like,your,shirt],[like,my,shirt]) ?
 1 1 Exit: alter([i,like,your,shirt],[you,like,my,shirt]) ?

P = [you,like,my,shirt] ?

yes

(the trace is indented for additional clarity)

```

Figure 29: Tracing the behaviour of the predicate alter/2

---

```

2: alter([H|T],[X|Y]):-
 change(H,X),
 alter(T,Y).
1: change(i,you).
2: change(me,you).
3: change(your,my).
4: change(their,our).
5: change(X,X).

```

An example trace of the program is given below in Figure 29.

We now will consider a few more examples of list processing predicates, looking at some of the general techniques used in them.

## 10.2 Deleting the first occurrence of an element from a list: the predicate `delete/2`

What do we mean? The predicate `delete/3` will need 3 arguments:

- the element `E` to be deleted
- the list `L` from which it is to be deleted
- the new list `NewL` which has the item deleted

We assume we know the first 2 and want to build the third, so the goal might be:

```
?- delete(a,[c,a,m,e,l],Ans).
```

So we would expect the result:

```
Ans=[c,m,e,l]
Yes
```

What do we do? We want to look at each element of the list `L` in turn to see if it is the element to be deleted, so need to access the head of `L` recursively (as in `member/2`). This means breaking `L` into a Head `HL` and a tail `TL` (using matching). We want to process the whole list and to build a new list at the same time.

The two cases we need to consider are:

1. when the head of the list that we are looking at is the one we want to delete
2. when it is not

In the first case we then need to save the rest of the list:

deleting the element `E` from the list with head `E` and tail `TL` will give the list `TL`

In prolog:

```
delete(E,[E|TL],TL).
```

In the second case we need to save the head as well as the rest of the list (we don't want to delete all the other elements):

deleting the element `E` from the list with head `HL` and tail `TL` will give the list with head `HL` and tail `NL` if deleting `E` from the list `TL` gives the list `NL`

```
delete(E,[HL|TL],[HL|NL]):-
 delete(E,TL,NL).
```

---

```

| ?- trace,delete(a,[c,a,p],Nlist).
| ?- delete(a,[c,a,p],Nlist).
 1 1 Call: delete(a,[c,a,p],_100) ?
 2 2 Call: delete(a,[a,p],_241) ?
 2 2 Exit: delete(a,[a,p],[p]) ?
 1 1 Exit: delete(a,[c,a,p],[c,p]) ?

Nlist = [c,p] ?

yes

```

Figure 30: Tracing the behaviour of the predicate `delete/3`

---

So, the complete program is:

```

delete(E,[E|TL],TL).

delete(E,[HL|TL],[HL|NL]):-
 delete(E,TL,NL).

```

And example goals would be:

```

| ?- delete(a,[c,a,p],Nlist).

 Nlist=[c,p]
yes

| ?- delete(e,[f,e,e,d],Ans).

 Ans=[f,e,d]
yes

```

An example trace of this is shown in Figure 30.

Note: we can use this also to help build a predicate for deleting ALL the elements E in the list L - consider what changes we need to the first clause.....

### 10.3 Reversing a list: the predicates `rev/2` and `rev/3`

We are going to write a predicate to reverse a list here. The predicate will be `rev/3`. For convenience it can be called by a simpler predicate `rev/2`.

```

| ?- listing(rev).

```



---

```

| ?- rev([a,b,c],Rev1).
 1 1 Call: rev([a,b,c],_86) ?
 2 2 Call: rev([a,b,c],[],_86) ?
 3 3 Call: rev([b,c],[a],_86) ?
 4 4 Call: rev([c],[b,a],_86) ?
 5 5 Call: rev([], [c,b,a],_86) ?
 5 5 Exit: rev([], [c,b,a], [c,b,a]) ?
 4 4 Exit: rev([c], [b,a], [c,b,a]) ?
 3 3 Exit: rev([b,c], [a], [c,b,a]) ?
 2 2 Exit: rev([a,b,c], [], [c,b,a]) ?
 1 1 Exit: rev([a,b,c], [c,b,a]) ?

Rev1 = [c,b,a] ?

yes

```

Figure 31: Tracing the behaviour of the predicate `rev/3`

---

```

1: rev(L,Rev1) :-
 rev(L, [],Rev1).

1: rev([],L,L).

2: rev([H|List],Acc,Rev1) :-
 rev(List, [H|Acc],Rev1).

```

We 'pour' each element into the accumulator one at a time, so that the new list builds up with each successive head at the front (and the first head being in first is now last).

When all are 'poured in' we copy the accumulator list across to the answer.

An example trace of `rev/3` is shown in Figure 31.

## 10.4 Joining two lists together: the predicates `append/3`

The predicate `append/3` enables us to join two lists together into one list.

```

| ?- listing(append).

1: append([],L,L).

2: append([H|L],M,[H|N]) :-
 append(L,M,N).

```

1. The list `L` is the same as the list `L` appended to the empty list.

---

```

| ?- append([a],[b],New1).
 1 1 Call: append([a],[b],_88) ?
 2 2 Call: append([], [b],_224) ?
 2 2 Exit: append([], [b],[b]) ?
 1 1 Exit: append([a],[b],[a,b]) ?

New1=[a,b]
yes

```

Figure 32: Tracing the behaviour of the predicate `append/3`

---

```

| ?- append([1,2],[3,4],X).
 1 1 Call: append([1,2],[3,4],_128) ?
 2 2 Call: append([2],[3,4],_264) ?
 3 3 Call: append([], [3,4],_363) ?
 3 3 Exit: append([], [3,4],[3,4]) ?
 2 2 Exit: append([2],[3,4],[2,3,4]) ?
 1 1 Exit: append([1,2],[3,4],[1,2,3,4]) ?

X=[1,2,3,4]
yes

```

Figure 33: A further example of the behaviour of the predicate `append/3`

---

2. The list with head H and tail N is the same as the list with head H and tail L appended to the list M if the list N is the list L appended to the list M.

An example trace of `append/3` is shown in Figure 32.

A further trace of `append/3` using a different goal is shown in Figure 33.

An example trace of using `append/3` with different instantiation patterns in the goal is shown in Figure 34.

What about

```

?- append(F, [3,4], [1,2,3,4]).

 F=[1,2]
yes

```

## 10.5 Exercises

**Question 10.1** The predicate `delete/3` succeeds if deleting the element represented by the first argument, from the list represented by the second argument, results in a list represented

---

```

| ?- append([1,2],R,[1,2,3,4]).
 1 1 Call: append([1,2],_96,[1,2,3,4]) ?
 2 2 Call: append([2],_96,[2,3,4]) ?
 3 3 Call: append([],_96,[3,4]) ?
 3 3 Exit: append([], [3,4], [3,4]) ?
 2 2 Exit: append([2], [3,4], [2,3,4]) ?
 1 1 Exit: append([1,2], [3,4], [1,2,3,4]) ?

R=[3,4]
yes

```

Figure 34: A trace of a different calling pattern of the predicate `append/3`

---

by the third argument.

```

e.g. ?- delete(a,[a,p,p,l,e],A).
 A=[p,p,l,e]
 yes

```

`delete/3` is defined as:

```

1. delete(E1,[E1|T],T).
2. delete(E1,[H|T],[H|NT]):-
 delete(E1,T,NT).

```

For each of the following:

- specify whether the query submitted to Prolog succeeds or fails;
- if it succeeds, specify what is assigned to any variables.

10.1a ?- `delete(X,[pat, john, paul],Ans)`.

10.1b ?- `delete(A,L,[t,o,p])`.

10.1c ?- `delete(a,[c,a,b],Ans)`.

10.1d ?- `delete(e,[d,o,g],P)`.

10.1e ?- `delete(e,[f,e,e,t],Ans)`.

**Question 10.2** The predicate `deleteall/3` succeeds if deleting all occurrences of the element represented by the first argument from the list represented by the second argument results in a list represented by the third argument.

e.g. `?- delete(p,[a,p,p,l,e],A). A=[a,l,e] yes`

deleteall/3 is defined as:

1. `deleteall(E1,[],[]).`
2. `deleteall(E1,[E1|T],NT):- deleteall(E1,T,NT).`
3. `deleteall(E1,[H|T],[H|NT]):- deleteall(E1,T,NT).`


For each of the following:

- specify whether the query submitted to Prolog succeeds or fails;
- if it succeeds, specify what is assigned to any variables.

10.2a `?- deleteall(e,[f,e,e,t],Ans).`

10.2b `?- deleteall(p,[d,o,g],X).`

10.2c Complete the AND/OR tree below which represents the execution of the query:



10.2d Suppose that the predicate deleteall/3 is defined incorrectly as:

1. `delall(E,[],[]).`
2. `delall(E,[E|T],Y):- delall(E,T,Y).`
3. `delall(E,[H|T],Y):- delall(E,T,[H|Y]).`

resulting in the behaviour:

i. `?- delall(e,[f,e,e,t],Ans).`  
no

However, the following query succeeds, as intended:

```
ii. ?- delall(a,[a,a,a],Ans).
 Ans=[]
 yes
```

Explain why the program does not give the intended answer to query i. using an AND/OR tree or a trace to illustrate your answer.

**Question 10.3** The predicate `repall/4` succeeds if replacing all occurrences of the element represented by the first argument, by the element represented by the second argument, in the list represented by the third argument, results in a list represented by the fourth argument.

```
e.g. ?- repall(p,b,[a,p,p,l,e],A).
 A=[a,b,b,l,e]
 yes
```

`repall/4` is defined as:

```
1. repall(E1,Rel,[],[]).
2. repall(E1,Rel,[E1|T],[Rel|NT]):-
 repall(E1,Rel,T,NT).
3. repall(E1,Rel,[H|T],[H|NT]):-
 repall(E1,Rel,T,NT).
```

For each of the following:

- specify whether the query submitted to Prolog succeeds or fails;
- if it succeeds, specify what is assigned to any variables.

```
10.3a ?- repall(e,a,[f,e,a,t],Ans).
```

```
10.3b ?- repall(P,r,[s,o,u,p],Ans).
```

**Question 10.4** The predicate `whowants/3` has three arguments. The first is intended to represent a type of food; the second a list of people; and the third another list, representing those people on the 2nd argument list who want the type of food specified in the 1st argument (where `want/2` is defined separately, with two arguments representing who wants what food).

```
e.g. ?- whowants(bean,[jo,tom,ann],Who).
 Who = [jo,ann]
 yes
```

```
1. whowants(Food,[Name|Rest],[Name|Others]):-
 wants(Name,Food),
 whowants(Food,Rest,Others).
```

2. `whowants(Food, [Name|Rest], Others):-  
    whowants(Food, Rest, Others).`
3. `whowants(Food, [], []).`
4. `wants(jo, chips).`
5. `wants(jo, beans).`
6. `wants(jo, eggs).`
7. `wants(ann, beans).`
8. `wants(ann, bacon).`
9. `wants(tom, eggs).`
10. `wants(tom, chips).`
11. `wants(rick, bacon).`

10.4a Give either the AND/OR tree or a trace which represents the execution of the query:

```
?- whowants(beans, [jo, tom, ann], Who).
```

For each of the following:

- specify whether the query submitted to Prolog succeeds or fails;
- if it succeeds, specify what is assigned to any variables.

10.4b `?- whowants(chips, [ann, rick], Ans).`

10.4c `?- whowants(bacon, [tom, ann], A).`

10.4d `?- whowants(eggs, Diners, Ans).`

10.4e Using `whowants/3` as a model, write a predicate `overage/3` that takes an age limit and a list of names, checks the age of each person named (using a predicate `age/2`) and returns a list of names of those people who are over the age limit.

For example, the query below should give the output shown:

```
?- overage(18, [sally, alice, bill], Ans).
Ans=[alice, bill]
yes
```

where:

```
age(sally, 15).
age(mark, 26).
age(bill, 20).
age(alice, 37).
```

10.4f If the predicate `whowants/3` had been (incorrectly) defined as:

1. `whowants(Food, [Name|Rest], [Name|Others]) :-`  
`wants(Name, Food),`  
`whowants(Food, Rest, Others).`
2. `whowants(Food, [Name|Rest], Others) :-`  
`whowants(Food, Rest, Others).`

(i.e. no 3rd clause) predict the outcome of the query in 4c.

```
?- whowants(bacon, [tom, ann], A).
```

and explain why this is the case, using an AND/OR tree or a trace in your explanation.

**Question 10.5** The predicate `deletefirst/3` is supposed to succeed if deleting the element represented by the first argument, from the list represented by the second argument, results in a list represented by the third argument, as `delete/3` defined as above (B.)

e.g. intended behaviour:

```
?- deletefirst(i, [b,i,b,s], A).
A=[b,b,s]
yes
```

Instead, it produces the following behaviour:

```
?- deletefirst(i, [b,i,b,s], A).
A=[s]
yes
```

```
?- deletefirst(a, [b,a,l,l], X).
no
```

`deletefirst/3` is defined as:

1. `deletefirst(E1, [E1|T], T).`
2. `deletefirst(E1, [H|T], NT) :-`  
`deletefirst(E1, T, [H|NT]).`

Explain why this predicate produces this behaviour (instead of the intended behaviour), using an AND/OR tree or a trace in your explanation.

## 10.6 Practical 5: List Processing

---

### 10.6.1 Introduction

This practical covers a lot of ground. You should attempt to get through at least the *Basics* section, and only go on to the material in the *More Advanced List Processing* once you are happy with the basics.

### 10.6.2 Basic List Processing

1. Copy the file `/home/infteach/prolog/code/useful.pl`, the contents of which are shown below, to your area. Get into Prolog and consult your copy of the file. List it and try and work out what the predicates are supposed to be used for.

```
(a) member(X, [X|_]).
 member(X, [_|T]) :-
 member(X, T).

(b) no_cons([]).
 no_cons([H|T]) :-
 vowel(H),
 no_cons(T).

 vowel(a).
 vowel(e).
 vowel(i).
 vowel(o).
 vowel(u).

(c) prlist([]).
 prlist([H|T]) :-
 prlist(H),
 nl,
 prlist(T).

 prlist(X) :-
 write(X).

(d) all_over_21([]).
 all_over_21([H|T]) :-
 21 < H,
 all_over_21(T).

(e) max([H|T], Answer) :-
 max(T, H, Answer).
```



```

max([], Answer, Answer).
max([H|T], Temp, Answer):-
 H > Temp,
 max(T, H, Answer).
max([H|T], Temp, Answer):-
 H =< Temp,
 max(T, Temp, Answer).

(f) last(X, [X]).
last(X, [H|T]):-
 last(X, T).

(g) write_type([]).
write_type([H|T]):-
 vowel(H),
 write(vowel),
 nl,
 write_type(T).
write_type([H|T]):-
 write(consonant),
 nl,
 write_type(T).

(h) rev(List, Revl):-
 revlist(List, [], Revl).

 revlist([H|L], Acc, Answer):-
 revlist(L, [H|Acc], Answer).
 revlist([], Answer, Answer).

(i) append([], Answer, Answer).
append([X|L1], L2, [X|L3]):-
 append(L1, L2, L3).

```

2. Work out what the result of each of the following goals should be. Type each and test it.

- (a) | ?- member(a, [c, a, b, b, a, g, e]).
- (b) | ?- member(X, [t, o, p]).
- (c) | ?- no\_cons([a, e, e, i]).
- (d) | ?- no\_cons([a, d, e]).
- (e) | ?- prlist([a, simple, list]).
- (f) | ?- prlist([a, [more, embedded], [nested, [list]]]).
- (g) | ?- all\_over\_21([32, 36, 97, 25]).
- (h) | ?- all\_over\_21([26, 7, 18]).
- (i) | ?- max([7, 3, 8, 2, 16], A).
- (j) | ?- last([which, is, the, last, word], Last).
- (k) | ?- write\_type([a, v, o, w, e, l]).

- (l) | ?- rev([ann,saw,john],R).
- (m) | ?- rev([1,2,3,4,5],Somelist).
- (n) | ?- append([the,dog],[chased,tom],Ans).
- (o) | ?- append(Who,[chased,the,dog],[the,cat,chased,the,dog]).

3. Now try and write predicates to do the following:

- (a) `next_to` where `X` is the element next to `Y` in the list `[A,...,X,Y,...,Z]`; for example

```
| ?- nexto(a,Y,[c,a,t]).
```

```
Y = t
```

```
| ?- nexto(Y,a,[c,a,t]).
```

```
Y = c
```

- (b) `nth` where `E1` is the `N`th element of a list; for example

```
| ?- nth(5,[w,o,r,s,e,n],E1).
```

```
E1 = e
```

```
| ?- nth(N,[f,r,e,d],r).
```

```
N = 2
```

- (c) `delall` where all elements of the list that match the first argument are deleted to produce a new list; for example

```
| ?- delall(a,[b,a,n,a,n,a],C).
```

```
C = [b,n,n]
```

- (d) `replall` where all elements of the list that match the first argument are replaced by the second argument to produce a new list: for example

```
| ?- replall(a,the,[a,cat,bit,a,dog],X).
```

```
X = [the,cat,bit,the,dog]
```

### 10.6.3 More Advanced List Processing

Once you are happy with all the material covered above, you can try writing some more complex list processing predicates. Try defining the following set and list utilities.

1. `subset(X,Y)` where set `X` is a subset of set `Y` (i.e., every element that is a member of `X` is a member of `Y`).
2. `equal(X,Y)` if `X` and `Y` are the same set (i.e., if both `X` and `Y` have the same elements, although not necessarily in the same order).
3. `union(A,B,C)` where set `C` is the union of set `A` and `B` (i.e., `C` contains all the elements of `A` and all the elements of `B`).

4. `intersection(A,B,C)` where set `C` is the intersection of sets `A` and `B` (i.e., `C` contains only those elements which are members of both `A` and `B`).
5. `difference(A,B,C)` where set `C` contains those members of set `A` which do not occur in set `B`.
6. `subst(A,B,C,D)` where list `D` is list `C` with element `A` substituted for all occurrences of element `B`.

# 11 Parsing in Prolog

## 11.1 Introduction

In this section, we introduce the facilities that Prolog provides for parsing. This is done through the idea of a parse tree as applied to a simple model for the construction of English sentences.

We describe Prolog's inbuilt mechanism for encoding a parser via **grammar rules**. We then explain how to extract the parse tree and show how to extend a parser using arbitrary Prolog code.

## 11.2 Simple English Syntax

First, what do we want the parser to do? We would like to know that a sentence is correct according to the (recognised) laws of English grammar: so, *The ball runs fast* is syntactically correct while *The man goes pub* is not, since the verb *go* (usually) does not take a direct object.

Secondly, we may want to build up some structure which describes the sentence—so it would be worth returning, as a result of the parse, an expression which represents the syntactic structure of the successfully parsed sentence.

Of course, we are not going to try to extract the **meaning** of the sentence so we will not consider attempting to build any **semantic** structures.

The components of this simple syntax will be such categories as sentences, nouns, verbs etc. Here is a (top down) description:

|                   |             |                                              |
|-------------------|-------------|----------------------------------------------|
| Unit:             | sentence    |                                              |
| Constructed from: |             | noun phrase followed by a verb phrase        |
| Unit:             | noun phrase |                                              |
| Constructed from: |             | proper noun or determiner followed by a noun |
| Unit:             | verb phrase |                                              |
| Constructed from: |             | verb or verb followed by noun phrase         |
| Unit:             | determiner  |                                              |
| Examples:         |             | a, the                                       |
| Unit:             | noun        |                                              |
| Examples:         |             | man, cake                                    |
| Unit:             | verb        |                                              |
| Examples:         |             | ate                                          |




Figure 35: A parse tree

### 11.3 The Parse Tree

Figure 35 shows the parse tree for the sentence *the man ate the cake* with some common abbreviations in brackets.

We must remember that many sentences are ambiguous—i.e., they result in different parse trees.

### 11.4 Prolog Grammar Rules

In the earlier section, we saw the beginnings of a parser which used list processing techniques to build up a string of words corresponding to a sentence. In this section, we describe Prolog's inbuilt mechanism for building grammar rules.

Prolog, as a convenience, will do most of the tedious work for you. This is how we can define the simple grammar which is accepted 'as is' by Prolog.

|             |     |                           |
|-------------|-----|---------------------------|
| sentence    | --> | noun_phrase, verb_phrase. |
| noun_phrase | --> | determiner, noun.         |
| verb_phrase | --> | verb, noun_phrase.        |
| determiner  | --> | [a].                      |
| determiner  | --> | [the].                    |
| noun        | --> | [man].                    |
| noun        | --> | [cake].                   |
| verb        | --> | [ate].                    |

It is very easy to extend if we want to include adjectives.

```
noun_phrase --> determiner, adjectives, noun.
adjectives --> adjective.
adjectives --> adjective, adjectives.
adjective --> [young].
```

This formulation is sometimes known as a **Definite Clause Grammar** (DCG).

We might later think about the ordering of these rules and whether they really capture the way we use adjectives in general conversation, but we won't pursue this here.

We could have interpreted these rules in English as follows:

```
Predicate noun_phrase/1
 noun_phrase(X) means that: X is a sequence of words
 forming a noun_phrase
```

In which case, we could have used the predicate `append/3` to implement this:

```
noun_phrase(X) :-
 append(Y,Z,X),
 det(Y),
 noun(Z).

det([the]).
noun([cat]).
noun([dog]).

append([],L,L).
append([H|L1],L2,[H|L3]) :-
 append(L1,L2,L3).

?- noun_phrase([the,cat]).
yes
```

Instead, our definition for `noun_phrase/2` is:

```
noun_phrase(X,Y) is true if:
 there is a noun phrase at the beginning of sequence X
 and the part of the sequence left after the nounphrase is Y.
```

In Prolog we could write this as:

```
noun_phrase(X,Y) :-
 det(X,Z),
 noun(Z,Y).
```

```
det([the|S],S).
noun([dog|S],S).
```

```
?- noun_phrase([the,dog,bit,the,cat],[bit,the,cat]).
yes
```

Essentially, the Prolog Grammar Rule formulation is **syntactic sugaring**. This means that Prolog enables you to write in:

```
sentence --> noun_phrase, verb_phrase.
```

and Prolog turns this into:

```
sentence(S,S0):-
 noun_phrase(S,S1),
 verb_phrase(S1,S0).
```

and

```
adjective --> [young].
```

into

```
adjective(A,A0):-
 'C'(A,young,A0).
```

where 'C'/3 is a built in Prolog predicate which is defined as if:

```
'C'([H|T],H,T).
```

## 11.5 Using the Grammar Rules

Set a goal of the form

```
sentence([the,man,ate,a,cake],[])
```

and *not* as

```
sentence.
```

or

```
sentence([the,man,ate,a,cake])
```

## 11.6 How to Extract a Parse Tree

We can add an extra argument which can be used to return a result.

```
sentence([[np,NP],[vp,VP]]) --> noun_phrase(NP), verb_phrase(VP).
noun_phrase([[det,Det],[noun,Noun]])--> determiner(Det), noun(Noun).
determiner(the) --> [the].
and so on
```

What we have done above is declare predicates `sentence/3`, `noun_phrase/3`, `verb_phrase/3`, `determiner/3`, and so on. The explicit argument is the first and the two others are added when the clause is read in by Prolog. Basically, Prolog expands a grammar rule with  $n$  arguments into a corresponding clause with  $n + 2$  arguments.

So what structure is returned from solving the goal?

```
sentence(Structure,[the,man,ate,a,cake],[])
```

The result is:

```
[[np,[[det,the],[noun,man]]],[vp,[...
```

Not too easy to read!

We can improve on this representation if we are allowed to use Prolog terms as arguments. For example, in `foo(happy(fred),12)` the term `happy(fred)` is one of the arguments of `foo/2`. Such a term is known as a *compound term*.

With the help of compound terms, we could tidy up our representation of sentence structure to something akin to:

```
sentence([np([det(the),noun(man))],vp[...
```

## 11.7 Adding Arbitrary Prolog Goals

Grammar rules are simply expanded to Prolog goals. We can also insert arbitrary Prolog subgoals on the right hand side of a grammar rule, but we must tell Prolog that we do not want them expanded. This is done with the help of **braces**—i.e., ‘{’ and ‘}’.

For example, here is a grammar rule which parses a single character input as an ASCII code and succeeds if the character represents a digit. It also returns the digit found.

```
digit(D) -->
 [X],
 { X >= 48,
```



```
X =< 57,
D is X-48 }.
```

The grammar rule looks for a character at the head of a list of input characters and succeeds if the Prolog subgoals

```
{ X >= 48,
 X =< 57,
 D is X-48 }.
```

succeed. Note that we assume we are working with ASCII codes for the characters and that the ASCII code for “0” is 48 and for “9” is 57. Also note the strange way of signifying “equal to or less than” as “=<”.

See further details on the use of Definite Clause Grammars in the Natural Language notes.

## 11.8 Practical 6: Definite Clause Grammars

---

### 11.8.1 Introduction

This practical is intended to give you some experience in writing Definite Clause Grammars. You are asked to extend a basic grammar in various ways.

### 11.8.2 The Basic Grammar

1. Copy the file `/home/infteach/prolog/code/grammar1.pl`, the contents of which are shown in Figure 36, to your area. Get into Prolog and consult your copy of the file.

Note here that the symbols `vt` and `vi` are being used to mean **transitive verb** and **intransitive verb** respectively. A transitive verb is one which takes an object noun phrase; an intransitive verb is one which does not take an object noun phrase.

2. You might find it interesting to do a `listing` of the grammar; you will see from this that Prolog adds extra arguments in the conversion from the definite clause grammar formalism to normal Prolog clauses.

```
| ?- [grammar1].
{consulting grammar1.pl...}
{grammar consulted, 50 msec 2650 bytes}
```

```
yes
| ?- listing.
```

```
n(A, B) :-
 'C'(A, dog, B).
```

```
n(A, B) :-
 'C'(A, cat, B).
```

```
s(A, B) :-
 np(A, C),
 vp(C, B).
```

```
...
```

```
yes
| ?-
```

3. To test the grammar, you have to provide calls to `s` with two arguments, since you are calling the translated form. The first argument is the list of words whose sentencehood you want to check, and the second argument is the empty list:

```
| ?- s([a,cat,chases,a,dog], []).
```

```
yes
| ?-
```

---

```
s --> np, vp.

np --> det, n.
np --> pn.

vp --> vt, np.
vp --> vi.

det --> [every].
det --> [a].

vt --> [chases].
vi --> [miaows].

n --> [dog].
n --> [cat].

pn --> [fido].
pn --> [tigger].
```

Figure 36: A simple Definite Clause Grammar

---

Note that you can also use this grammar to randomly generate sentences by providing a variable as the first argument to `s`:

```
| ?- s(Sentence, []).

Sentence = [every, dog, chases, every, dog] ? ;

Sentence = [every, dog, chases, every, cat] ? ;

Sentence = [every, dog, chases, a, dog] ? ;

...
```

Test the grammar with the following sentences and make sure you understand its behaviour; before trying each sentence, try to work out what Prolog will do and why.

```
the dogs miaow
the cat miaows
fido chases a cat
every dog chases
tigger miaows a dog
a tigger miaows
```

Try other random sentences.

### 11.8.3 Structure Building

A string of words is in the language defined by a grammar if a tree corresponding to the structure of the string can be built. Parsing a sentence in Prolog consists in building such a tree implicitly. We can make this tree explicit using another facility provided by the DCG notation: grammar symbols may be given arguments, in exactly the same way as Prolog goals. To build the tree, we associate with each non-terminal symbol an argument which represents its structure, as in the following example:

```
s([s,[NP,VP]]) --> np(NP), vp(VP).
```

For this exercise, you should augment the grammar given in Figure 36 with structure-building arguments. You should achieve the following result.

```
| ?- s(ParseTree,[tigger,miaows],[]).

ParseTree = [s,[[np,[[pn,[tigger]]]], [vp,[[verb,[miaows]]]]]] ? ;

no
| ?-
```

So, for each node  $n$  in the tree, the result should contain a list whose first element is the name of the node, and whose second element is a list of elements that correspond to the nodes that are daughters of  $n$ . Each terminal node should be represented by the word that lies at that terminal node.

Note that, since we are adding an additional argument to each predicate, calls to `s` must now contain this new argument; it appears *before* the arguments required by the DCG translation process.

Also notice that, in the example above, we have typed a `;` to see if Prolog can offer any more parses. When more than one parse is available, we say that the string is **syntactically ambiguous**.

### 11.8.4 Adding Number Agreement

Above, we saw how an extra argument was added to the non-terminal symbols of a grammar to build a syntactic structure. Any number of arguments may be added in this way. The DCG translator just passes through the argument structure and adds the two string handling arguments at the end.

This aspect of DCGs can be used to improve the coverage of a grammar. For instance, we can add an argument whose values range over `{singular, plural}` to represent the feature **number**. By introducing the appropriate values for this feature in the lexicon and then percolating them up and identifying the values on say, subject and verb phrase, we can guarantee number agreement. Here's an example:

```
noun([n, [dog]],plural) --> [dogs].
```

```

np([np, [Det, Noun]], Num) --> det(Det, Num), noun(Noun, Num) .
s(s(NP, VP)) --> np(NP, Num), vp(VP, Num) .

```

For this part of the exercise, you should augment all of the appropriate rules in the grammar to take account of number agreement information, so that your grammar should rule out sentences like the following:

1. the dogs miaows
2. the cat miaow
3. fido chase a cat
4. every dogs chases a cat
5. tigger chases a dogs

To make this work, you will have to attend to the following:

- Note that, in English, the subject noun phrase and the verb phrase in a sentence must agree in number; similarly, in a noun phrase, the determiner and noun must agree in number. However, the number of the object noun phrase in a sentence is not relevant to the grammaticality of the sentence as a whole.
- Each **lexical rule** (the rules that have words as their right hand sides) will need to be augmented with number information, which will then percolate up via unification to higher level nodes in the trees.

### 11.8.5 Extending the Coverage of the Grammar

You should now try to extend the coverage of the grammar to include the following phenomena. In each case you will have to add at least one new grammar rule and some lexical rules.

**Relative clauses** as in *A cat that chases Tigger miaows*. You should incorporate number agreement arguments to rule out sentences like *\*A cat that chase Tigger miaows*.

**Prepositional phrases** as in

A cat with a hangup chases Tigger.  
 A cat with a hangover miaows in discomfort.

Note here that prepositional phrases may appear to be attached to either noun phrases, as in the first case, or to verb phrases, as in the second case.

Your grammar should provide two parses for the sentence *A cat chases the dog with a mousetrap*.

Ultimately your grammar should build structure for these syntactic constructions, but you may find it easier in the first instance to add the additional grammar rules required to the version of the grammar that doesn't build structure.

## 12 Input/Output

### 12.1 Basic input/output facilities

Prolog provides the system predicates `read` and `write` for reading and writing atoms:

```
| ?- read(X).
|: 2.
```

X = 2 ?

```
yes
| ?- read(X).
|: 'This is just a few words between single quotes'.
```

X = This is just a few words between single quotes ?

```
yes
| ?- write(foo).
foo
yes
| ?- write('Where *did* you buy that jacket?').
Where *did* you buy that jacket?
yes
| ?-
```

We might start with a basic program that computes the cube of a number:

```
cube(N,C) :-
 C is N * N * N.
```

Using the program:

```
| ?- cube(1, X).
```

X = 1 ?

```
yes
| ?- cube(5, Y).
```

Y = 125 ?

```
yes
| ?- cube(12, Z).
```

Z = 1728 ?

```
yes
| ?-
```

We can then make modification so that the program will read the numbers itself:

```
cube :-
 read(X),
 process(X).

process(stop):- !.

process(N) :-
 C is N * N * N,
 write(C), ttyflush,
 cube.
```

```
| ?- cube.
|: 2.
8
|: 5.
125
|: 1728.
864813056
|: stop.
```

```
yes
| ?-
```

We might think that this could be simplified:

```
cube :-
 read(stop).

cube :-
 read(N),
 C is N * N * N,
 write(C), ttyflush,
 cube.
```

However, we get the following behaviour:

```
| ?- cube.
|: 34.
|: 2.
8
|: stop.
```

What happens is that `read` cannot be redone, so when the first `read` in the first `cube` rule is called, the number '34' is input. It fails to match to 'stop', so this rule fails and the next `cube` rule is tried. This calls `read` and waits for a further input, '2', which it then 'cubes' and writes out: the first value input has been lost. We can illustrate this by tracing the program.

```
| ?- trace, cube.
 1 1 Call: cube ?
 2 2 Call: read(stop) ?
|: 34.
 2 2 Fail: read(stop) ?
 2 2 Call: read(_165) ?
|: 2.
 2 2 Exit: read(2) ?
 3 2 Call: _170 is 2*2*2 ?
 3 2 Exit: 8 is 2*2*2 ?
 4 2 Call: write(8) ?
8 4 2 Exit: write(8) ?
 5 2 Call: ttyflush ?
 5 2 Exit: ttyflush ?
 6 2 Call: cube ?
 7 3 Call: read(stop) ?
|: stop.
 7 3 Exit: read(stop) ?
 6 2 Exit: cube ?
 1 1 Exit: cube ?
```

yes

- `get(X)` unifies `X` with next non blank printable character (in ASCII code) from current input stream
- `get0(X)` unifies `X` with next character (in ASCII) from current input stream
- `put(X)` puts a character on to the current output stream; `X` must be bound to a legal ASCII code

For example:

```
| ?- get0(C).
|:
C = 10 ?
```

yes

```
| ?- get0(C).
|: f
```



```
C = 102 ?
yes
| ?- get0(C).
|:
```

```
C = 32 ?
yes
| ?-
```

```
| ?- get(C).
|:
|:
|:
|: .
```

```
C = 46 ?
yes
| ?- get(C).
|: y
```

```
C = 121 ?
yes
| ?- put(87).
W
yes
| ?- put(32).
```

```
yes
| ?- put(10).
```

```
yes
| ?-
```

## 12.2 File input/output

A wee program:

```
double(X, Y):-
 Y is 2*X.

test:-
 read(X),
 double(X,Y),
 write(Y),
 nl.
```

```
| ?- test.
|: 2.
4
yes
| ?-
```

We might want to put this into a test loop, for testing a number of values.

```
double(X, Y):-
 Y is 2*X.

test:-
 read(X),
 \+(X = -1),
 double(X,Y),
 write(Y),
 nl,
 test.

test.
```

Rather than keep having to type in the test values, we might want to store them in a file, and read them into the program each time an input is called for.

```
go:-
 see(datafile),
 test,
 seen,
 write('Okay, all done.').

double(X, Y):-
 Y is 2*X.

test:-
 read(X),
 \+(X = -1),
 double(X,Y),
 write(Y),
 nl,
 test.

test.
```

So if the contents of the file were:

```
3.
54.
-1.
```

and we ran the program, we would get:

```
| ?- go.
6
108
Okay, all done.
yes
| ?-
```

The useful predicates here for reading in from a file are:

- `see/1`
- `seen/0`
- `seeing/1`

where `see/1` specifies which file to read from ('datafile' in the above example); `seen/0` closes this file and returns to the default input from the terminal, and `seeing/1` enables the user to find out where the input is coming from at any time.

We might also want to consider writing the output from our testing to a file:

```
go:-
 tell(resultsfile),
 see(datafile),
 test,
 seen,
 told,
 write('Okay, all done.').

double(X, Y):-
 Y is 2*X.

test:-
 read(X),
 \+(X = -1),
 double(X,Y),
 write(Y),
 nl,
 test.

test.
```

The system predicates provided here are:

- `tell/1`
- `told/0`

- `telling/1`

where `tell/1` specifies the file to write to; `told/0` closes this file and returns to the default keyboard output, and `telling/1` allows the user to query where the output is being sent to at any time.

Instead of defining an arbitrary character to indicate when we have read all the values we need from a file, we can use the pre-defined `end_of_file` marker.

```
go:-
 tell(resultsfile),
 see(datafile),
 test,
 seen,
 told,
 write('Okay, all done.').

double(X, Y):-
 Y is 2*X.

test:-
 read(X),
 \+(X = end_of_file),
 double(X,Y),
 write(Y),
 nl,
 test.

test.
```

### 12.3 Translating atoms and strings

The `name` predicate defines the relation that holds between an atom and the list of characters that make it up.

```
| ?- name(cat,Y).
```

```
Y = [99,97,116] ?
```

```
yes
```

```
| ?- name(X,[98,101,114,116]).
```

```
X = bert ?
```

```
yes
```

```
| ?-
```

At least one of `name`'s arguments must be instantiated. It can be used to avoid ASCII codes:

| ?- [98,101,114,116] = "bert".

yes

| ?- [X] ="a".

X = 97 ?

yes

| ?-

This will be used in the later section on Morphology.

## 12.4 Practical 7: Input/Output

---

### 12.4.1 Introduction

This practical is intended to give you some experience in using Prolog's built-in input/output predicates. From the notes above you should have some understanding of the operation of each of the following predicates:

| Predicate               | Behaviour                                                         |
|-------------------------|-------------------------------------------------------------------|
| <code>read/1</code>     | read a term from the current input stream                         |
| <code>write/1</code>    | write a term to the current output stream                         |
| <code>nl/0</code>       | write a newline to the current output stream                      |
| <code>tab/1</code>      | write a specified number of spaces to the current output stream   |
| <code>put/1</code>      | write a specified ASCII character to the current output stream    |
| <code>get/1</code>      | read a printable ASCII character from the current input stream    |
| <code>get0/1</code>     | read an ASCII character from the current input stream             |
| <code>see/1</code>      | make the specified file be the current input stream               |
| <code>seeing/1</code>   | determine the current input stream                                |
| <code>seen/0</code>     | close the current input stream and reset it to <code>user</code>  |
| <code>ttyflush/0</code> | flush the output buffer                                           |
| <code>tell/1</code>     | make the specified file be the current output stream              |
| <code>telling/1</code>  | determine the current output stream                               |
| <code>told/0</code>     | close the current output stream and reset it to <code>user</code> |
| <code>name/2</code>     | arg 1 (an atom) is made of the ASCII characters listed in arg 2   |

In this practical, you will use some of these predicates to write a number of programs that make use of input and output. More information on each of these predicates can be found in the SICStus manual.

There is a lot of material here: you should not expect to complete all the exercises during the practical, but see how far you can get.

### 12.4.2 Basic Terminal Input/Output

1. Write a predicate `echo/0` that reads a number from the terminal and writes it out again. Your program should keep going until it reads the number 0. So, for example, you should see behaviour like the following:

```
| ?- echo.
|: 3.
3
|: 4.
4
|: 5.
5
```

```
|: 1278.
1278
|: 0.

yes
| ?-
```

Note that each number typed in must be terminated by a full stop.

2. Modify this program so that it prints a prompt for input, and prints a response message, along the following lines:

```
| ?- echo.
Give me a number: 4.
Your number was 4.
Give me a number: 129.
Your number was 129.
Give me a number: 0.
Okay, see you later.
yes
| ?-
```

Note that you will have to use `ttyflush` to print the prompt in the way it is done here; alternatively, you might look at what the manual has to say about the built-in predicate `prompt/2`.

3. Write a predicate `limit/1` which takes a number as argument, and then reads numbers typed at the terminal, for each number saying whether it is greater than, equal to, or less than the number provided as argument. The program should stop when it reads a 0. The program's behaviour should be something like the following:

```
| ?- limit(41).
Give me a number: 28.
Your number was below the limit.
Give me a number: 41.
Your number was the same as the limit.
Give me a number: 71891.
Your number was above the limit.
Give me a number: 0.
Okay, see you later.
yes
| ?-
```

### 12.4.3 File Input/Output

1. Write a predicate called `numberfile/1` that takes as argument the name of a file. The named file should be one you have already constructed, containing a number terminated by a full stop on each line, like the following:

```
1.
1898.
34.
```

1891238912398.

12.

18.

0.

The last number in the file should be a 0; this is how you will terminate the reading process.

The predicate should open the file and read numbers from the file one at a time, writing them out to the screen, until it reads the 0. It should then close this input stream.

2. Augment the program above so that it will write out *any* number it finds, including 0. To do this, you should modify the program so that it terminates on reaching the `end_of_file` marker.
3. Write a predicate `findnumber/2` that takes as arguments a number and the name of a file. The predicate should open the file and read numbers from the file one at a time, doing nothing until it finds a number that matches the first argument; it should then announce this fact and continue with the rest of the file.

4. Modify the `echo` predicate you defined earlier so that it now takes two arguments, the first the name of an input file and the second the name of an output file.

The predicate should open the input file and read numbers from the file one at a time, writing them out to the output file. Use the `end_of_file` marker to terminate the process, at which point you should make sure you reset both the input and output streams to `user`.

5. Modify `echo` again, this time to read the names of the input and output files from the terminal. You should print appropriate prompts in each case.
6. Modify the predicate `limit` you defined earlier, to read input from a file and to write the numbers out to different places as follows:

- if the number is the same as the limit, announce this on the terminal;
- if the number is less than the limit, write it to a file called `lower`; and
- if the number is greater than the limit, write it to a file called `higher`.

Once done, check the contents of the two files to make sure your program did what it should have done.

#### 12.4.4 Atoms and Strings

To do these problems, you will need to make use of the built-in predicate `name/2`.

1. Define a predicate `starts/2` that takes as arguments an atom and a character, and checks whether the atom starts with the character. So, your predicate should behave in the following manner:



```
| ?- starts(foo,f).
```

```
yes
```

```
| ?- starts(baz,g).
```

```
no
```

```
| ?-
```

2. Define a predicate called `plural/2` which will convert nouns into their plural forms: for example

```
| ?- plural(cake,X).
```

```
X = cakes
```

```
yes
```

```
| ?- plural(xylophone, X).
```

```
X = xylophones
```

```
yes
```

```
| ?-
```

3. Modify `plural` so that it reads words from a file and writes the output to the screen.

## 13 Morphology: A List Processing Application

Most natural languages show a regularity in the way words decompose into units of meaning: for example, in English most plurals are formed by adding the letter *s*. This suggests a possible economy for NLP systems: instead of storing every form (singular, plural, etc.) of each word, we can store just the base form plus some rules for building the derived forms. Question: how would we do this in Prolog?

We are going to look at:

- Turning atoms into strings and back again.
- Using `append` to concatenate strings.
- Adding exceptions to general rules.

Just to remind you about lists:

```
[1] = [1|[]]
[1, 2] = [1|[2]]
[1, 2, 3] = [1|[2,3]]
 = [1|[2|[3]]]
 = [1,2|[3]]
```

and about using the predicate `append`:

```
append([],L,L).
append([H|T], L1, [H|L2]):-
 append(T, L1, L2).
```

Given the goal:

```
?- append([a,b], [c,d], L).
```

the following trace results:

```
| ?- trace, append([a,b],[c,d],L).
{The debugger will first creep
-- showing everything (trace)}
 1 1 Call: append([a,b],[c,d],_84) ?
 2 2 Call: append([b],[c,d],_238) ?
 3 3 Call: append([], [c,d], _345) ?
 3 3 Exit: append([], [c,d], [c,d]) ?
 2 2 Exit: append([b], [c,d], [b,c,d]) ?
 1 1 Exit: append([a,b], [c,d], [a,b,c,d]) ?
```

```
L = [a,b,c,d] ?
```

```
yes
{trace}
| ?-
```

So now we have `name` which gives us a way of converting an atom into a list of characters, and `append` which gives us a way of concatenating lists. So, we have a way of concatenating atoms:

```
| ?- name(black,L1), name(bird,L2),
 append(L1,L2,L3), name(Word,L3).

L1 = [98,108,97,99,107],
L2 = [98,105,114,100],
L3 = [98,108,97,99,107,98,105,114,100],
Word = blackbird ?

yes
| ?-
```

If we then think about the simple plural rule in English:

To make the plural form of a singular noun, add an *s*.

for example:

| <u>Singular Noun</u> | <u>Plural Noun</u> |
|----------------------|--------------------|
| terminal             | terminals          |
| tree                 | trees              |
| cube                 | cubes              |

How can we perform this mapping in Prolog? We could define a predicate

```
plural(SingularNoun, PluralNoun)
```

with the following behaviour:

```
| ?- plural(cube, Plural).

Plural = cubes ?

yes
| ?- plural(keyboard, Plural).
Plural = keyboards ?

yes
| ?-
```

How do we do this?

```
% plural(Sing, Plu)
```

```
plural(Sing, Plu):-
 name(Sing, SingChs),
 name(s, EndChs),
 append(SingChs, EndChs, PluChs),
 name(Plu, PluChs).
```

Tracing an example:

```
| ?- trace, plural(cube, Plural).
Call: plural(cube,_58) ?
Call: name(cube,_210) ?
Exit: name(cube,[99,117,98,101]) ?
Call: name(s,_216) ?
Exit: name(s,[115]) ?
Call: append([99,117,98,101],[115],_223) ?
Call: append([117,98,101],[115],_706) ?
Call: append([98,101],[115],_814) ?
Call: append([101],[115],_921) ?
Call: append([], [115],_1027) ?
Exit: append([], [115], [115]) ?
Exit: append([101], [115], [101,115]) ?
Exit: append([98,101], [115], [98,101,115]) ?
Exit: append([117,98,101], [115], [117,98,101,115]) ?
Exit: append([99,117,98,101], [115], [99,117,98,101,115]) ?
Call: name(_58,[99,117,98,101,115]) ?
Exit: name(cubes,[99,117,98,101,115]) ?
Exit: plural(cube,cubes) ?
```

Plural = cubes ?

yes

There are other morphological transformations in English: for example, we generally add *ed* to get the past tense of verbs:

| Present Tense | Past Tense |
|---------------|------------|
| collect       | collected  |
| screen        | screened   |
| attend        | attended   |

We can build a more general procedure

```
% generate_morph(BaseForm, Suffix, DerivedForm)

generate_morph(BaseForm, Suffix, DerivedForm):-
 name(BaseForm, BaseFormChs),
 name(Suffix, SuffixChs),
```

```
append(BaseFormChs, SuffixChs, DerivedFormChs),
name(DerivedForm, DerivedFormChs).
```

A problem: there are exceptions to the morphological rules we have seen.

For example:

- To make the plural of a word like *knife*, we not only add an *s* but we change the *f* to a *v*.
- To make the past tense of a word like *create*, we add only a *d*, instead of *ed*.

How do we deal with this? First, we define a predicate `morph/3`: this is just `append/3` under another name.

```
morph([], Suffix, Suffix).
morph([H|T], Suffix, [H|L2]):-
 morph(T, Suffix, L2).
```

We call this inside `generate_morph`:

```
% generate_morph(BaseForm, Suffix, DerivedForm)

generate_morph(BaseForm, Suffix, DerivedForm):-
 name(BaseForm, BaseFormChs),
 name(Suffix, SuffixChs),
 morph(BaseFormChs, SuffixChs, DerivedFormChs),
 name(DerivedForm, DerivedFormChs).
```

A few other things you should remember about strings:

```
| ?- X = "fe".
```

```
X = [102,101] ?
```

```
yes
```

```
| ?- X = "s".
```

```
X = [115] ?
```

```
yes
```

```
| ?- X = "ves".
```

```
X = [118,101,115] ?
```

```
yes
```

```
| ?- X = "e".
```

```
X = [101] ?
yes
| ?- X = "ed".
```

```
X = [101,100] ?
yes
| ?-
```

Note that the exceptions we indicated are also **rules**: there are classes of words that have this behaviour (in other words, they are *not* irregular forms).

So, we can add special base cases for these more specific rules:

```
morph([102,101], [115], [118,101,115]).
morph([101], [101,100], [101,100]).
morph([], Suffix, Suffix).
```

```
morph([H|T], Suffix, [H|L2]):-
 morph(T, Suffix, L2).
```

But we don't need to use ASCII codes explicitly:

```
morph("fe", "s", "ves").
morph("e", "ed", "ed").
morph([], Suffix, Suffix).
```

```
morph([H|T], Suffix, [H|L2]):-
 morph(T, Suffix, L2).
```

## 13.1 Summary

- Understand how to use `name` to turn atoms into strings and back again.
- Understand how to use `append` to concatenate strings.
- Understand how to add exceptions to general rules.

## 13.2 Prolog Practical 8: Morphology: An Exercise in List Processing

---

### 13.2.1 Introduction

This practical is intended to give you some practice in using list processing. The practical is built around the use of the built-in predicate `name/2` and the `generate_morph/3` predicate explained above.

### 13.2.2 Basic Operations on Atoms and Strings

The exercises in this practical are carried over from the previous practical. To do these problems, you will need to make use of the built-in predicate `name/2`.

1. Define a predicate `starts/2` that takes as arguments an atom and a character, and checks whether the atom starts with the character. So, your predicate should behave in the following manner:

```
| ?- starts(foo,f).
```

```
yes
```

```
| ?- starts(baz,g).
```

```
no
```

```
| ?-
```

2. Define a predicate called `plural/2` which will convert nouns into their plural forms: for example

```
| ?- plural(cake,X).
```

```
X = cakes
```

```
yes
```

```
| ?- plural(xylophone, X).
```

```
X = xylophones
```

```
yes
```

```
| ?-
```

3. Modify `plural` so that it reads words from a file and writes the output to the screen.

### 13.2.3 Morphological Processing

Figure 37 shows the basic elements required for morphological processing.

This is used as follows:

---

```

generate_morph(BaseForm, Suffix, DerivedForm):-
 name(BaseForm, BaseFormChs),
 name(Suffix, SuffixChs),
 morph(BaseFormChs, SuffixChs, DerivedFormChs),
 name(DerivedForm, DerivedFormChs).

morph("fe", "s", "ves").
morph("e", "ed", "ed").
morph([], Suffix, Suffix).

morph([H|T], Suffix, [H|L2]):-
 morph(T, Suffix, L2).

```

Figure 37: Predicates for morphological processing

---

```

| ?- generate_morph(knife,s,W).

W = knives ?

yes
| ?- generate_morph(terminal,s,W).

W = terminals ?

yes
| ?- generate_morph(attempt,ed,W).

W = attempted ?

yes
| ?- generate_morph(inundate,ed,W).

W = inundated ?

yes
| ?-

```

For this part of the practical, you have to extend this program in various ways.

1. First, make sure you understand thoroughly how the program works. To do this you should type it in and trace its behaviour.
2. The program as it stands includes, effectively, a rule that pluralises words ending in *-fe* by changing the *f* to a *v* before adding the *s*. There are other rules required for other plurals. Add additional clauses to obtain the following behaviour from your program:

```

| ?- generate_morph(box,s,W).

```



```

W = boxes ?

yes
| ?- generate_morph(fox,s,W).

W = foxes ?

yes
| ?- generate_morph(spy,s,W).

W = spies ?

yes
| ?- generate_morph(fly,s,W).

W = flies ?

yes
| ?-

```

3. Think about how would you extend the program above so it *also* deals with the following case:

```

| ?- generate_morph(ox,s,W).

W = oxen ?

yes
| ?-

```

You don't have to implement this: the objective is to understand the nature of the problem.

4. Finally, add clauses to your program so that it produces the following behaviour:

```

| ?- generate_morph(slow,ly,W).

W = slowly ?

yes
| ?- generate_morph(full,ly,W).

W = fully ?

yes
| ?-

```

# 14 Top-down design: The Missionaries and Cannibals Problem

## 14.1 The Problem

Given the following ...

- Three missionaries and three cannibals are trying to cross a river: they want to get from the left bank to the right bank.
- There is one boat, which can take two people at most.
- If the cannibals outnumber the missionaries on a bank, then the missionaries get eaten. We'll assume that the boat being at a bank counts as it being on the bank.

... we want to find the schedule of crossings that will get all six across safely.

We'll use this problem to demonstrate the **top-down design** of a solution.

## 14.2 Viewing the Problem as State Space Search

In abstract terms:

- We can represent the states as follows:

|               |           |
|---------------|-----------|
| Initial State | MMMCCCB — |
| Goal State    | — MMMCCCB |

- We can represent the possible moves as follows:

|      |      |
|------|------|
| M →  | ← M  |
| MM → | ← MM |
| MC → | ← MC |
| CC → | ← CC |
| C →  | ← C  |

A strategy for solving the problem, given a current state:

- Try each move in turn.
  - Check if it is a safe move to make.
  - If it is not, try another move.
- If there are no more moves to try, go back to where we had a choice of moves and try again.
- Once we've made a move, check to see if we are at the goal state; if not, repeat the process.

## 14.3 A Prolog Program to Solve the Problem

What we have to do:

1. Choose a representation.
2. Decide on the algorithm to be used.
3. Choose predicates to represent each step.
4. Work out the detail in a top-down fashion.

**Choosing a Representation:** How will we represent the situation on each side of the river?

Some possibilities:

- Use a list of the entities that are on that side of the river:

```
[m,m,m,c,c,c,b]
```

- Use separate terms for each type of entity:

```
missionaries(3)
cannibals(3)
```

- Use one term with three arguments:

```
leftside(3,3,1)
```

- Use a list with three elements:

```
[3,3,1]
```

where

- missionaries = {0,1,2,3}
- cannibals = {0,1,2,3}
- boat = {0,1}

We'll use the last of these; so:

|                       |       |   |         |
|-----------------------|-------|---|---------|
| <b>Initial state:</b> | Left  | = | [3,3,1] |
|                       | Right | = | [0,0,0] |
| <b>Goal state:</b>    | Left  | = | [0,0,0] |
|                       | Right | = | [3,3,1] |

## 14.4 Problem Solution

The general idea is to make moves from one bank to the other until we reach the goal state.

To make a move from one bank to the other:

1. Make a move and get the new states of the banks.
2. Check if the new states of the banks are safe (i.e., make sure the missionaries won't get eaten); if they are not, repeat Step 1 for a different move.
3. Repeat the entire process until we reach the goal state.

We have to choose some predicates to represent these steps, then work top-down.

**The Top Level Predicate:** The main predicate `gofrom/2`:

- make a move and get the new states of the banks
- if `safe(Left)` and `safe(Right)` then `gofrom` the new states.

In Prolog:

```
gofrom(Left, Right):-
 applymove(Left, Right, NewLeft, NewRight),
 safe(Left),
 safe(Right),
 gofrom(NewLeft, NewRight).
```

We need to check if the goal state has been reached:

```
gofrom([0,0,0],[3,3,1]).
```

Remember we have to put this before the recursive `gofrom/2` clause.

**Applying a Move** The predicate `applymove/4` gives us new states of the two banks by applying a move.

- If the boat is already on the left bank, the pattern when called will be:

```
applymove([M1,C1,1],[M2,C2,0],...)
```

- If the boat is on the right, the pattern will be:

```
applymove([M1,C1,0],[M2,C2,1],...)
```

To save having to do two sets of clauses, one for each side, we can check which side the boat is on then apply the same operator:

- If the boat is on the left:

```

applymove(Left, Right, NewLeft, NewRight):-
 boathere(Left),
 moveload(Left, Right, NewLeft, NewRight).

```

- If the boat is on the right:

```

applymove(Left, Right, NewLeft, NewRight):-
 boathere(Right),
 moveload(Right, Left, NewRight, NewLeft).

```

This uses the same `boathere/1` and `moveload/4`. `boathere/1` is defined very simply:

```
boathere([M,C,1]).
```

**Moving a Load** Now we need to define the predicate `moveload/4`.

The algorithm:

- select a move (`move/1`)
- check if it is possible (`possibletomove/2`)
- make the move (`performmove/5`)

So:

```

moveload(Source, Target, NewSource, NewTarget):-
 move(BoatLoad),
 possibletomove(Source, BoatLoad),
 performthemove(Source, Target,
 BoatLoad, NewSource, NewTarget).

```

Note that we use `Source` and `Target` since we may be moving from the left bank to the right, or from the right bank to the left.

**Choosing a Move** We have five possibilities for boatloads: one missionary, two missionaries, one missionary and one cannibal, two cannibals, or one cannibal.

How do we represent the moves?

- we could represent the possible moves as a list, and repeatedly select elements from this list; or
- we could represent each move as a separate clause and choose each clause in turn.

With the latter, we can use Prolog's backtracking: get a move, then if it fails, backtrack and get another move.

`move/1` is called with its argument uninstantiated:

```
move(Boatload)
```

This matches to potential moves and instantiates `Boatload` in the process. For example:

```
move([1,1,1]).
```

means one missionary, one cannibal, and one boat.

You have to define the other moves yourself.

**Checking Whether a Move is Possible** Defining `possibletomove/2`:

- Suppose the current state of the `Source` is `[2,0,1]`.
- Suppose the move selected is `move([1,1,1])`.

This move is not possible, since there are no cannibals to move.

So, there must be, at the most, the number of missionaries, cannibals and boats in the move as in the `Source` state.

```
possibletomove([M,C,B], [MoveM,MoveC,MoveB]):-
 M >= MoveM,
 C >= MoveC,
 B >= MoveB.
```

We've selected a move and checked if it is possible. Now we have to perform the move.

**Performing the Chosen Move** We want to define `performthemove/5`.

What we have to do:

- we are given the current `Source` and `Target` banks
- we know the move to be made
- we want to find out the new states of the `Source` and `Target` banks after the move.

```
performthemove(Source, Target,
 Move, NewSource, NewTarget)
```

The definition is as follows:

```
performthemove([SM,SC,SB], [TM,TC,TB], [MM,MC,MB],
 [NSM,NSC,NSB], [NTM,NTC,NTB]):-
 NSM is SM - MM,
 NSC is SC - MC,
 NSB is SB - MB,
 NTM is SM + MM,
 NTC is SC + MC,
 NTB is SB + MB.
```

**Safe States** We still have to define `safe/1`, which determines whether a state is safe.

A state is safe provided there are not more cannibals than missionaries, or if there are no missionaries in that state (then it doesn't matter how many cannibals there are). So:

```
safe([2,3,1])
```

should be false, since there are two missionaries and three cannibals.

You have to define this predicate. `safe/1` should succeed if the state is safe, and fail if it is not.

**The Problem of Looping** We don't want to waste effort visiting states already tried.

- How do we know if we've tried a state before? We keep a list of all the states visited.
- Do we need to remember the whole state? No: The left (or right) bank alone will do.

So:

1. Each time we make a move, add the old left side state to a list of previous left side states.
2. Before committing ourselves to the next move, after checking the resulting state is safe, we check if we are looping, i.e., if the new left state is a member of the list of previously visited left states. If it is not, then we are okay.

You should add the code to do this check.

## 14.5 Where We Are So Far

go:-

```
gofrom([3,3,1], [0,0,0], [[3,3,1]]).
```

```
gofrom([0,0,0], [3,3,1], PreviousLeftStates).
```

```
gofrom(Left, Right, PreviousLeftStates):-
 applymove(Left, Right, NewLeft, NewRight),
 safe(NewLeft),
 safe(NewRight),
 \+looping(NewLeft, PreviousLeftStates),
 gofrom(NewLeft, NewRight,
 [NewLeft|PreviousLeftStates]).
```

```
applymove(Left, Right, NewLeft, NewRight):-
 boathere(Left),
 moveload(Left, Right, NewLeft, NewRight).
```

```

applymove(Left, Right, NewLeft, NewRight):-
 boathere(Right),
 moveload(Right, Left, NewRight, NewLeft).

boathere([Missionaries, Cannibals, 1]).

moveload(Source, Target, NewSource, NewTarget):-
 move(BoatLoad),
 possibletomove(Source, BoatLoad),
 performthemove(Source, Target,
 BoatLoad, NewSource, NewTarget).

possibletomove([M,C,B], [MoveM,MoveC,MoveB]):-
 M >= MoveM,
 C >= MoveC,
 B >= MoveB.

performthemove([SourceM, SourceC, SourceB],
 [TargetM, TargetC, TargetB],
 [MoveM, MoveC, MoveB],
 [NewSourceM, NewSourceC, NewSourceB],
 [NewTargetM, NewTargetC, NewTargetB]):-
 NewSourceM is SourceM - MoveM,
 NewSourceC is SourceC - MoveC,
 NewSourceB is SourceB - MoveB,
 NewTargetM is TargetM + MoveM,
 NewTargetC is TargetC + MoveC,
 NewTargetB is TargetB + MoveB.

```



## 14.6 Prolog Practical 9: The Missionaries and Cannibals Problem

---

### 14.6.1 Introduction

This practical is based on the Missionaries and Cannibals Problem described above.

In summary, the problem is as follows

- Three missionaries and three cannibals are trying to cross a river: they want to get from the left bank to the right bank.
- There is one boat, which can take two people at most.
- If the cannibals outnumber the missionaries on a bank, then the missionaries get eaten. We'll assume that the boat being at a bank counts as it being on the bank.

We want to find the schedule of crossings that will get all six across safely.

1. Copy the file `/home/infteach/prolog/code/mandc.pl` into your area.
2. Look at it and try and work out how it works. Work out which predicates calls which other ones, and what purpose each serves in solving the problem.
3. Go into Prolog and consult the file.
4. Try out some of the individual predicates and see if they work as you thought they did.

### 14.6.2 Making it Work

There are three predicate definitions missing: `safe/1`, `move/1` and `looping/2`. Your task is to write these missing definitions. Note that:

- In the code, you'll see the form `\+looping(NewLeft, PreviousLeftStates)`. The `\+` means 'not': if a goal  $G$  is true, then `\+G` will be false, and *vice versa*. So, when you write the definition for `looping/2`, it should check for looping being the case; the goal `\+looping(NewLeft, PreviousLeftStates)` will then succeed if there is *no* looping.
- The file contains the `member/2` predicate: you will need to make use of this.

Edit the file and add in the missing clauses, then get into Prolog again, consult the file and test it.

### 14.6.3 Adding a Commentary

You should add `write` statements in sensible places so that the program writes out the sequence of moves it tries, and the successful set of moves that gives the solution.

---

```

go:-
 gofrom([3,3,1], [0,0,0], [[3,3,1]]).

gofrom([0,0,0], [3,3,1], PreviousLeftStates).

gofrom(Left, Right, PreviousLeftStates):-
 applymove(Left, Right, NewLeft, NewRight),
 safe(NewLeft),
 safe(NewRight),
 \+looping(NewLeft, PreviousLeftStates),
 gofrom(NewLeft, NewRight, [NewLeft|PreviousLeftStates]).

applymove(Left, Right, NewLeft, NewRight):-
 boathere(Left),
 moveload(Left, Right, NewLeft, NewRight).

applymove(Left, Right, NewLeft, NewRight):-
 boathere(Right),
 moveload(Right, Left, NewRight, NewLeft).

boathere([Missionaries, Cannibals, 1]).

moveload(Source, Target, NewSource, NewTarget):-
 move(BoatLoad),
 possibletomove(Source, BoatLoad),
 performthemove(Source, Target, BoatLoad, NewSource, NewTarget).

possibletomove([M,C,B], [MoveM,MoveC,MoveB]):-
 M >= MoveM, C >= MoveC, B >= MoveB.

performthemove([SourceM, SourceC, SourceB],
 [TargetM, TargetC, TargetB],
 [MoveM, MoveC, MoveB],
 [NewSourceM, NewSourceC, NewSourceB],
 [NewTargetM, NewTargetC, NewTargetB]):-
 NewSourceM is SourceM - MoveM,
 NewSourceC is SourceC - MoveC,
 NewSourceB is SourceB - MoveB,
 NewTargetM is TargetM + MoveM,
 NewTargetC is TargetC + MoveC,
 NewTargetB is TargetB + MoveB.

member(X, [X|_]).
member(X, [_|Z]):-
 member(X, Z).

```

Figure 38: The Code Provided

---

# 15 Eliza in Prolog

## 15.1 Implementing Eliza

The general algorithm:

- Read a sentence.
- Perform some transformation on the sentence to produce a response.
- Write out the response.

### A Rudimentary Eliza

go:-

```
 read_in(Input),
 process(Input).
```

process([W1,W2]):-

```
 member(W1, [bye, goodbye, quit, halt, chow]),
 write('I hope I was of some help. '), nl.
```

process(Input):-

```
 build_response(Input, Output),
 reply(Output), nl,
 go.
```

build\_response(Input, Output):-

```
 rule(Input, Output).
```

rule([Greeting|Rest], [hi|Rest]):-

```
 member(Greeting, [hi, hello]).
```

rule([i,hate,X, '.'],

```
 [do, you, really, hate, X, ?]).
```

rule([X,you,Y,me, '?''],

```
 [why, do, you, ask, if, 'I', Y, you, ?]).
```

rule([i,like,X, '.'],

```
 [does, anyone, else, in, your, family, like, X, ?]).
```

rule([are, you, X, '?''],

```
 [what, makes, you, think, 'I', am, X, ?]).
```

rule([you, are, X, '.'],

```
 [does, it, please, you, to, believe, 'I', am, X, ?]).
```

rule(X,

```
 [please, go, on, '.']).
```

A Consultation

| ?- go.  
|: Hello there.  
Hi there.

|: How are you?  
Please go on.

|: Can you help me?  
Why do you ask if I help you?

|: Because you're supposed to.  
Please go on.

|: Don't you like me?  
Why do you ask if I like you?

|: You're not being very helpful.  
Please go on.

|: Are you deaf?  
What makes you think I am deaf?

|: You don't help much.  
Please go on.

|: Oh, I give up.  
Please go on.

|: Bye.  
I hope I was of some help.

## 15.2 Some Comments on Prolog Layout

Program headers:

```
% Program: pract1.pl
% Author: fredcl
% Updated: 9 March 1991
% Purpose: AI1 Prolog Practical
% Uses: append/3 from utils.pl

% Defines:
% foo(+,-)
% baz(+,+)
```

Code:

separate different predicate definitions by at least one blank line

General structure:

```
% Predicate Header Documentation
Head of Clause1:-
 Indented Subgoal1
 ...
 Indented Subgoaln
Head of Clause1:-
 ...
Head of Clause1:-
 Indented Subgoal1
 ...
 Indented Subgoaln
% Predicate Header Documentation
...
```

- Precede each predicate with some description.
- Use end of line comments to explain what's happening.

```
append([], A, A). % The base case
append([A|B], C, [A|D]):- % The recursive case
 append(B, C, D).
```

## 15.3 Prolog Practical 10: Eliza

---

This practical is based on the elementary Eliza program described above.

1. Copy the file `/home/infteach/prolog/code/elizette.pl` into your area.
  2. The file contains a lot of code that is concerned with input and output. For your purposes, however, the important predicates are shown in Figure 39.  
Look at the code and make sure you understand how it works.
  3. Go into Prolog and consult the file. Experiment a little. An example consultation with the program is shown in Figure 40.
  4. The dialogue contributions that the program is capable of making are not particularly impressive. Think about ways in which you could extend the coverage of the program, and try some of these out.
- 

```
go:-
 read_in(Input),
 process(Input).

process([W1,W2]):-
 member(W1, [bye, goodbye, quit, halt, chow]),
 write('I hope I was of some help. '), nl.

process(Input):-
 build_response(Input, Output),
 reply(Output), nl,
 go.

build_response(Input, Output):-
 rule(Input, Output).

rule([Greeting|Rest], [hi|Rest]):-
 member(Greeting, [hi, hello]).
rule([i,hate,X,'.'], [do, you, really, hate, X, ?]).
rule([X,you,Y,me,'?'], [why, do, you, ask, if, 'I', Y, you, ?]).
rule([i,like,X,'.'], [does, anyone, else, in, your, family, like, X, ?]).
rule([are, you, X,'?'], [what, makes, you, think, 'I', am, X, ?]).
rule([you, are, X, '.'], [does, it, please, you, to, believe, 'I', am, X, ?]).
rule(X, [please, go, on, '.']).
```

Figure 39: The basics of a rudimentary Eliza

---

---

| ?- go.  
|: Hello there.  
Hi there.

|: How are you?  
Please go on.

|: Can you help me?  
Why do you ask if I help you?

|: Because you're supposed to.  
Please go on.

|: Don't you like me?  
Why do you ask if I like you?

|: You're not being very helpful.  
Please go on.

|: Are you deaf?  
What makes you think I am deaf?

|: You don't help much.  
Please go on.

|: Oh, I give up.  
Please go on.

|: Bye.  
I hope I was of some help.

yes  
| ?-

Figure 40: A sample consultation

---

# 16 Problem Solving in Prolog: The Monkey and the Bananas

Here we take a standard example AI problem: modelling problem-solving behaviour. We look at representing the world as symbolic descriptions, using predicates and arguments for relations and objects and also for actions. We use operators to represent actions in the world—a good approach where the search space is potentially infinite, but a finite set of potential actions and outcomes can be identified.

## 16.1 The Problem

A hungry monkey is in a cage. Suspended from the roof, just out of his reach, is a bunch of bananas. In the corner of the cage is a box. After several unsuccessful attempts to reach the bananas, the monkey walks to the box, pushes it under the bananas, climbs on to it, picks the bananas and eats them.

How do we write a program that can build a plan which, if executed, would model the behaviour of the monkey?

## 16.2 The General Approach to a Solution

We express the actions that can be performed as **operators** which act on the world and change its state.

For example:

- the monkey can move objects;
- the monkey can move from place to place;
- the monkey can stand on the floor or climb on the box;
- the monkey can grab bananas.

**Operators:** this basic techniques we'll use come from an early AI program called STRIPS.

- Each operator has outcomes: it affects the state of the world.
- Each operator can only be applied in certain circumstances: these circumstances are the **preconditions** of the operator.

## 16.3 Representational Considerations

An informal solution would be:



The monkey pushes the box under the bananas, climbs on it and grabs the bananas.

Questions:

- How do we represent the state of the world?
- How do we represent operators?
- Does our representation make it easy to:
  - check preconditions;
  - alter the state of the world after performing actions; and
  - recognise the goal state?

**Representing the World:** we have:

- a monkey, a box, the bananas, a floor;
- places in the room—call them  $a$ ,  $b$ , and  $c$ ;
- relations of objects to places:
  - the monkey being at location  $a$ ;
  - the monkey being on the floor;
  - the bananas hanging;
  - the box being at the same place as the bananas.

We use appropriately chosen predicates and arguments:

```
at(monkey, a)
on(monkey, box)
status(bananas, hanging)
at(box,X), at(bananas,X)
```

The Initial State of the World:

```
on(monkey, floor),
on(box, floor),
at(monkey, a),
at(box, b),
at(bananas, c),
status(bananas, hanging)
```

The Goal State:

```

on(monkey, box),
on(box, floor),
at(monkey, c),
at(box, c),
at(bananas, c),
status(bananas, grabbed)

```

**Representing Operators:** We make some assumptions about the use of the operators and what needs to be stated explicitly:

**Moving around in the world:** only the monkey can move: `go(X,Y)`—e.g., `go(a,b)`

**Pushing things around:** Again, only the monkey does this: `push(B,X,Y)`—e.g., `push(box,a,b)`

**Climbing on objects:** `climb_on(X)` Note: we don't allow the monkey to get off the box!

**Grabbing objects:** `grab(X)`

Each operator has preconditions and effects on the world:

| Operator                 | Preconditions                                                                                        | Effects                                      |                                              |
|--------------------------|------------------------------------------------------------------------------------------------------|----------------------------------------------|----------------------------------------------|
|                          |                                                                                                      | Delete                                       | Add                                          |
| <code>go(X,Y)</code>     | <code>at(m,X)</code><br><code>on(m,fl)</code>                                                        | <code>at(m,X)</code>                         | <code>at(m,Y)</code>                         |
| <code>push(B,X,Y)</code> | <code>at(m,X)</code><br><code>at(B,X)</code><br><code>on(m,fl)</code><br><code>on(B,fl)</code>       | <code>at(m,X)</code><br><code>at(B,X)</code> | <code>at(m,Y)</code><br><code>at(B,Y)</code> |
| <code>climb_on(B)</code> | <code>at(m,X)</code><br><code>at(B,X)</code><br><code>on(m,fl)</code><br><code>on(B,fl)</code>       | <code>on(m,fl)</code>                        | <code>on(m,B)</code>                         |
| <code>grab(B)</code>     | <code>on(m,box)</code><br><code>at(box,X)</code><br><code>at(B,X)</code><br><code>status(B,h)</code> | <code>status(B,h)</code>                     | <code>status(B,g)</code>                     |


where:

```

m = monkey
fl = floor
h = hanging
g = grabbed

```

## How the World Changes



## The General Solution

1. Look at the state of the world:
  - Is it the goal state? If so, the list of operators so far is the plan to be applied.
  - If not, go to Step 2.
2. Pick an operator:
  - Check it has not already been applied (i.e., check for looping).
  - Check if it can be applied (ie that the preconditions are satisfied).

If either of these checks fails, backtrack to get another operator.
3. Apply the operator:
  - Make changes to the world: delete from and add to the world state.
  - Add the operator to the list of operators to be applied.
  - Go to Step 1.

## 16.4 Doing All This in Prolog

The top level predicate will be `solve/3`, whose arguments are the initial state, the goal state, and the eventual plan.

Operators will be represented by the predicate `opn/4` whose arguments are

- the operator name and arguments;
- the list of preconditions;
- what to delete from the world state;
- what to add to the world state

### The Top Level in Prolog:

```
% solve(+State, +Goal, -Plan).
% Given starting State and final Goal state,
% returns a Plan consisting of a list of
% operations for transforming State into
% Goal (N.B. operators in plan are in reverse
% order of application).
```

```
solve(State, Goal, Plan):-
 solve(State, Goal, [], Plan).
```

### The Operators in Prolog:

```
opn(go(X,Y),
 [at(monkey,X), on(monkey,floor)],
 [at(monkey,X)],
 [at(monkey,Y)]).
```

```
opn(push(B,X,Y),
 [at(monkey,X), at(B,X),
 on(monkey,floor), on(B,floor)],
 [at(monkey,X), at(B,X)],
 [at(monkey,Y), at(B,Y)]).
```

```
opn(climbon(B),
 [at(monkey,X), at(B,X),
 on(monkey,floor), on(B,floor)],
 [on(monkey,floor)],
 [on(monkey,B)]).
```

```
opn(grab(B),
 [on(monkey,box), at(box,X),
 at(B,X), status(B,hanging)],
 [status(B,hanging)],
 [status(B,grabbed)]).
```

**The Main Predicate:** The work is done by the `solve/4` predicate. There are two cases:

- if we've reached the goal state; and
- if we haven't reached the goal state.

The arguments to `solve/4` are the current state, the goal state, the sequence of operations so far, and the final plan.

- If at the goal state, the plan is the sequence of operators so far;
- If not at the goal state:
  - select an operator: (match to `opn/4`)
  - check if not a member of list so far (use `member`)
  - check if preconditions hold in world (i.e., preconditions list should be a subset of world state)
  - delete from world state what is no longer true (use `dellist`)
  - add to world state what is now true (use `append`)
  - recurse on `solve` to get the next state

The Prolog Code for `solve/4` is:

```
solve(State, Goal, Plan, Plan):-
 is_subset(Goal, State).

solve(State, Goal, Sofar, Plan):-
 opn(Op, Preconditions, Delete, Add),
 \+ member(Op, Sofar),
 is_subset(Preconditions, State),
 delete_list(Delete, State, Remainder),
 append(Add, Remainder, NewState),
 solve(NewState, Goal, [Op|Sofar], Plan).
```

**Utility Predicates are:**


```
is_subset([H|T], Set):-
 member(H, Set),
 is_subset(T, Set).
is_subset([], _).

delete_list([H|T], List, Final):-
 remove(H, List, Remainder),
 delete_list(T, Remainder, Final).
delete_list([], List, List).


remove(X, [X|T], T).
remove(X, [H|T], [H|R]):-
 remove(X, T, R).
```

## 16.5 Using Strips in Another Domain

The initial state:



The goal state:



An Operator Definition:

| Operator          | Preconditions                              | Effects                    |                            |
|-------------------|--------------------------------------------|----------------------------|----------------------------|
|                   |                                            | Delete                     | Add                        |
| $move(Obj, A, B)$ | $on(Obj, A)$<br>$clear(Obj)$<br>$clear(B)$ | $on(Obj, A)$<br>$clear(B)$ | $on(Obj, B)$<br>$clear(A)$ |

## 16.6 Prolog Practical 11: A Simple Version of STRIPS

---

### 16.6.1 Basics

The file `/home/infteach/prolog/code/simstrips.pl` contains a simple version of a STRIPS-type means-ends analysis program, coded to solve the monkey and bananas problem. This is the program discussed above. The main predicates are shown in Figure 41; the utility predicates used are shown in Figure 42.

Copy the file to your area. Have a look at the file and make sure you understand the structure of the program. Think back to how the program is supposed to work (operators with preconditions, add and delete lists applied to some world state); try and match this to the program. Run the program by calling the top level goal

```
test(P).
```

The `test/1` predicate has a subgoal `solve/3`. The first argument of `solve` represents the initial state of the world; the second represents the goal state to be achieved; and the third will become instantiated to the plan that, when applied to the initial state, will achieve the goal state.

### 16.6.2 Extensions

Consider the blockworld scenario shown in Figure 43. You might represent this using predicates like the following:

```
on(brick, floor), on(cube, brick), on(pyramid, cube),
colour(cube, green), colour(brick, red), colour(pyramid, blue),
clear(pyramid), clear(floor).
```

Suppose the configuration of objects in Figure 43 is the initial state, and the configuration in Figure 44 is the desired goal state. When you think that you understand how the `simstrips` program works, modify the code to build a plan that will achieve the goal state. You should express the goal state as the following:

```
on(X,Y), colour(X,red), colour(Y,green)
```

or, in English: *put the red thing on the green thing.*

You will need to change the operators and the initial and goal state representations.

Note that you do not need to change the `solve/3` predicate or the utilities. Run your program and test it.

---

```

test(Plan):-
 solve([on(monkey,floor),on(box,floor),at(monkey,a),at(box,b),
 at(bananas,c),status(bananas,hanging)],
 [on(monkey,box),on(box,floor),at(monkey,c),at(box,c),
 at(bananas,c),status(bananas,grabbed)],
 Plan).

solve(State, Goal, Plan):-
 solve(State, Goal, [], Plan).

solve(State, Goal, Plan, Plan):-
 is_subset(Goal, State).
solve(State, Goal, Sofar, Plan):-
 opn(Op, Preconditions, Delete, Add),
 \+ member(Op, Sofar),
 is_subset(Preconditions, State),
 delete_list(Delete, State, Remainder),
 append(Add, Remainder, NewState),
 solve(NewState, Goal, [Op|Sofar], Plan).

opn(go(X,Y),
 [at(monkey,X), on(monkey,floor)],
 [at(monkey,X)],
 [at(monkey,Y)]).

opn(push(B,X,Y),
 [at(monkey,X), at(B,X), on(monkey,floor), on(B,floor)],
 [at(monkey,X), at(B,X)],
 [at(monkey,Y), at(B,Y)]).

opn(climbon(B),
 [at(monkey,X), at(B,X), on(monkey,floor), on(B,floor)],
 [on(monkey,floor)],
 [on(monkey,B)]).

opn(grab(B),
 [on(monkey,box), at(box,X), at(B,X), status(B,hanging)],
 [status(B,hanging)],
 [status(B,grabbed)]).

```

Figure 41: The simstrips program

---



---

```
is_subset([H|T], Set):-
 member(H, Set),
 is_subset(T, Set).
is_subset([], _).

delete_list([H|T], List, Final):-
 remove(H, List, Remainder),
 delete_list(T, Remainder, Final).
delete_list([], List, List).

remove(X, [X|T], T).
remove(X, [H|T], [H|R]):-
 remove(X, T, R).

append([H|T], L1, [H|L2]):-
 append(T, L1, L2).
append([], L, L).

member(X, [X|_]).
member(X, [_|T]):-
 member(X, T).
```

Figure 42: The utilities used in the `simstrips` program

---




Figure 43: The initial blockworld state

---




Figure 44: The goal state

---

# 17 Answers

## 17.1 Chapter 2

**Question 2.1** The predicate `=/2` takes 2 arguments and tries to unify them. For each of the following:

- specify whether the query submitted to Prolog succeeds or fails;
- if it succeeds, specify what is assigned to any variables.

2.1a     ?- Pear = apple.  
Pear = apple     yes

2.1b     ?- car = beetle.  
no

2.1c     ?- likes(beer(murphys),john) = likes(Who,What).  
What = john,     Who = beer(murphys)     yes

2.1d     ?- f(1) = F.  
f(1) = F             yes

2.1e     ?- name(Family) = smith.  
no

2.1f     ?- times(2,2) = Four.  
Four = times(2,2)     yes

2.1g     ?- 5\*3 = 15.  
no

2.1h     ?- f(X,Y) = f(P,P).  
X = P   Y = P             yes

2.1i     ?- a(X,y) = a(y,Z).  
X = y   Z = y             yes

2.1j     ?- a(X,y) = a(z,X).  
no

**Question 2.2** The following Prolog program is consulted by the Prolog interpreter.

```
vertical(seg(point(X,Y),point(X,Y1))).
horizontal(seg(point(X,Y),point(X1,Y))).
```

What will be the outcome of each of the following queries?

```
2.2a ?- vertical(seg(point(1,1),point(1,2))).
 yes
```

Note that whilst the constants 1 and 2 match to variables in the program, the Prolog interpreter only returns the values of any variables in the query – none in this case.

```
2.2b ?- vertical(seg(point(1,1),point(2,Y))).
 no
```

The Y in the query is independent of the variable Y in the program. In fact, when the program is interpreted the variables are renamed anyway to avoid any confusion.

```
2.2c ?- horizontal(seg(point(1,1),point(2,Y))).
 Y = 1 yes
```

```
2.2d ?- vertical(seg(point(2,3),P)).
 P = point(2,_A) yes
```

The second argument of `point/2`, in the second argument of `seg`, has not been instantiated by the match, so is returned as a variable.

```
2.2e ?- vertical(S), horizontal(S).
 S = seg(point(_B,_A),point(_B,_A)) yes
```

Here is an example of renaming of variables: what were X, Y, and Y1 in the program are renamed `_A` and `_B`

**Question 2.3** The following Prolog program is consulted by the Prolog interpreter.

```
parent(pat,jim).
parent(pam,bob).
parent(bob,ann).
parent(bob,pat).
parent(tom,liz).
parent(tom,bob).
```

What will be the outcome of each of the following queries?

```
2.3a ?- parent(bob,pat).
 yes
```

```
2.3b ?- parent(liz,pat).
 no
```

- 2.3c     ?- parent(tom,ben).  
no
- 2.3d     ?- parent(Pam,Liz).  
Liz = jim Pam = pat     yes
- 2.3e     ?- parent(P,C),parent(P,C2).  
C = jim C2 = jim P = pat     yes

**Question 2.4** The following Prolog program is consulted by the Prolog interpreter.

```
colour(b1,red).
colour(b2,blue).
colour(b3,yellow).
shape(b1,square).
shape(b2,circle).
shape(b3,square).
size(b1,small).
size(b2,small).
size(b3,large).
```

What will be the outcome of each of the following queries?

- 2.4a     ?- shape(b3,S).  
S=square     yes
- 2.4b     ?- size(W,small).  
W=b1     yes
- 2.4c     ?- colour(R,blue).  
R=b2     yes
- 2.4d     ?- shape(Y,square),colour(Y,blue).  
no
- 2.4e     ?- size(X,large),colour(X,yellow).  
X=c3     yes
- 2.4f     ?- shape(BlockA,square),shape(BlockB,square).  
BlockA=b1 BlockB=b1     yes
- 2.4g     ?- size(b2,S),shape(b2,S).  
no
- 2.4h     ?- colour(b1,Shape),size(X,small),shape(Y,circle).  
Shape=red X=b1 Y=b2     yes

**Question 2.5** The following Prolog program is consulted by the Prolog interpreter.

```
film(res_dogs,dir(tarantino),stars(keitel,roth),1992).
film(sleepless,dir(ephron),stars(ryan,hanks),1993).
film(bambi,dir(disney),stars(bambi,thumper),1942).
film(jur_park,dir(spielberg),stars(neill,dern),1993).
```

What will be the outcome of each of the following queries?

2.5a     ?- film(res\_dogs,D,S,1992).  
          D=dir(tarantino) S=stars(keitel,roth)   yes

2.5b     ?- film(F,dir(D),stars(Who,hanks),Y).  
          F=sleepless D=ephron Who=ryan Y=1993   yes

2.5c     ?- film(What,Who,stars(thumper),1942).  
          no

2.5d Write the query that would answer the question:

”Who directed Jurassic Park (jur\_park)?”

and give the outcome of the query.

```
?- film(jur_park,dir(Director),stars(X,Y),Z).
 Director=spielberg X=neill Y=dern Z=1993 yes
```

2.5e Write the query that would answer the question:

”What film did hanks appear in in 1993 and who was the other star?”

and give the outcome of the query.

```
?- film(Film,D,stars(Other,hanks),1993).
 Film=sleepless D=dir(ephron) Other=ryan
[might also try ?- film(Film,D,stars(hanks,Other),1993).
with outcome no if did not know order of stars]
```

## 17.2 Chapter 4

**Question 4.1** The following Prolog program is consulted by the Prolog interpreter.

```
big(bear).
big(elephant).
small(cat).
```

```
brown(bear).
black(cat).
grey(elephant).
dark(Animal):- black(Animal).
dark(Animal):- brown(Animal).
```

What will be the outcome of each of the following queries?

- 4.1a     ?- dark(X), big(X).  
          X = bear            yes
- 4.1b     ?- big(X), grey(Y).  
          X = bear    Y = elephant            yes
- 4.1c     ?- dark(D), small(D).  
          D = cat            yes
- 4.1d     ?- big(Animal), black(Animal).  
          no
- 4.1e     ?- small(P), black(P), dark(P).  
          P = cat            yes

**Question 4.2** The following Prolog program is consulted by the Prolog interpreter.

```
knows(A,B):-
 friends(A, B).

knows(A,B):-
 friends(A, C),
 knows(C, B).

friends(john, alice).
friends(alice, tom).
friends(sue, john).
friends(sue, clive).
friends(fred, tom).
friends(tom, sue).
```

State whether the following queries succeed or fail. If a query fails, explain why.

- 4.2a     ?- knows(alice, john).  
          yes
- 4.2b     ?- knows(clive, sue).  
          yes

4.2c     ?- knows(alice, fred).  
no, loops

4.2d     ?- knows(sue, john).  
no, relationship not defined  
(and cannot change order of arguments).

## 17.3 Chapter 5

For each of the following programs, say if the query given fails or succeeds. Give any bindings made as a consequence.

### Question 5.1

```
a:-b,c.
b.
c:-d.
d:-e.
```

```
?- a.
```

```
fails (a if b and c.
 b succeeds.
 c if d.
 d if e.
 e fails.
 so d fails. so
 c fails.
 so a fails.)
```

### Question 5.2

```
a:-b,c.
c:-e.
b:-f,g.
b:-n.
e.
f.
n.
```

```
?- a.
```

```
succeeds (a if b and c.
 b if f and g.
```



```

 f succeeds. g fails.
 can't redo f so f fails.
redo b. b if n.
 n succeeds.
b succeeds.
c if e.
 e succeeds.
c succeeds.
a succeeds.)

```

[Unfortunately sicstus prolog gives the same answer to both 5.1 and 5.2 here:

```

{EXISTENCE ERROR: g: procedure user:g/0 does not exist}

```

on the assumption that if you have predicates called with no facts for them then maybe you made an error, and you probably meant to have facts for e in 5.1 and g in 5.2. Particularly this is unfortunate because the sicstus interpreter does not go on to prove b if n, but stops to tell you that g does not exist. I think this is a bad design decision: you should have got a warning here, as with singleton variables, not an "existence error".]

### Question 5.3

```

do(X):-a(X),b(X).
a(X):-c(X),d(X).
a(X):-e(X).
b(X):-f(X).
b(X):-c(X).
b(X):-d(X).
c(1).
c(3).
d(3).
d(2).
e(2).
f(1).

```

```

5.3a ?- do(1). no
(do(1) if a(1) and b(1)
 a(1) if c(1) and d(1)
 c(1) succeeds
 d(1) fails
redo a(1) if e(1)
 e(1) fails
 a(1) fails
do(1) fails)

```

```

5.3b ?- do(2). yes
(do(2) if a(2) and b(2)
 a(2) if c(2) and d(2)
 c(2) fails
redo a(2) if e(2)
 e(2) succeeds
 a(2) succeeds
 b(2) if f(2)
 f(2) fails
redo b(2) if c(2)
 c(2) fails
redo b(2) if d(2)
 d(2) succeeds
 b succeeds
do(2) succeeds)

5.3c ?- do(3). yes
(do(3) if a(3) and b(3)
 a(3) if c(3) and d(3)
 c(3) succeeds
 d(3) succeeds
 a(3) succeeds
 b(3) if f(3)
 f(3) fails
redo b(3) if c(3)
 c(3) succeeds
 b(3) succeeds
do(3) succeeds)

5.3d ?- do(A). A = 3 yes
(do(A) if a(A) and b(A)
 a(A) if c(A) and d(A)
 c(1) succeeds
 d(1) fails
redo c(A)
 c(3) succeeds
 d(3) succeeds
 a(3) succeeds
 b(3) if f(3)
 f(3) fails
redo b(3) if c(3)
 c(3) succeeds
 b(3) succeeds
do(3) succeeds)

```

## 17.4 Chapter 7

**Question 7.1** How many elements are there in each of the following list structures?

7.1a `[a, [a, [a, [a]]]]` = 2

7.1b `[1,2,3,1,2,3,1,2,3]` = 9

7.1c `[a(X), b(Y,Z), c, X]` = 4

7.1d `[[sum(1,2)], [sum(3,4)], [sum(4,6)]]` = 3

7.1e `[c, [d,[x]], [f(s)], [r,h,a(t)], [[[a]]]]` = 5

[Note: extra spaces used to illustrate each element]

**Question 7.2** The predicate `=/2` takes 2 arguments and tries to unify them. For each of the following:

- specify whether the query submitted to Prolog succeeds or fails;
- if it succeeds, specify what is assigned to any variables;
- if it fails, explain why it fails.

7.2a `?- [A,B,C,D] = [a,b,c].` no lists are different length.

7.2b `?- bar([1,2,3]) = bar(A).` yes `A = [1,2,3].`

7.2c `?- [X, []] = [X].` no lists are different length.

7.2d `?- [X,Y|Z] = [a,b,c,d].` yes `X=a, Y=b, Z=[c,d].`

7.2e `?- [1,X,X] = [A,A,2].` no `A` bound to 1, cannot then be bound to 2.

7.2f `?- likes(Y,a) = likes(X,Y).` yes `X=a, Y=a.`  
(`X` and `Y` share)

7.2g `?- [foo(a,b),a,b] = [X|Y].` yes `X=foo(a,b), Y=[a,b].`

7.2h `?- [b,Y] = [Y,a].` no `Y` cannot be bound to `b` and `a`.

7.2i `?- [H|T] = [red,blue,b(X,Y)].` yes `H=red, T=[blue,b(X,Y)].`

7.2j `?- [1,[a,b],2] = [[A,B],X,Y].` no lists cannot unify with atoms.

7.2k    ?- test(a,L) = test(E,[b,c,d]).    yes        E = a, L = [b,c,d]

7.2l    ?- [[a,[b]],C] = [C,D].            yes        C = [a,[b]], D =[a,[b]]

7.2m    ?- [fred|T]=[H|[sue,john]].        yes        H = fred T = [sue,john]

**Question 7.3** Imagine that this program is consulted by the Prolog interpreter:

```
foo([], []).
```

```
foo([H|T], [X|Y]):-
 H = X,
 foo(T,Y).
```

[Note: this program tests if two lists unify by testing if the heads unify then recursing on the tails]

What will be the outcome of each of the following queries?

7.3a    ?- foo([a,b,c], A).                yes        A=[a,b,c].

7.3b    ?- foo([c,a,t], [c,u,t]).        no

7.3c    ?- foo(X, [b,o,o]).                yes        X=[b,o,o].

7.3d    ?- foo([p|L], [F|[a,b]]).        yes        F=p, L=[a,b].

7.3e    ?- foo([X,Y], [d,o,g]).            no

## 17.5 Chapter 8

**Question 8.1** The predicate member/2 succeeds if the first argument matches an element of the list represented by the second argument.

e.g.    ?- member(1,[2,3,1,4]).            yes

```
member/2 is defined as:
1. member(E1,[E1|T]).
2. member(E1,[H|T]):-
 member(E1,T).
```

For each of the following:

- specify whether the query submitted to Prolog succeeds or fails;

- if it succeeds, specify what is assigned to any variables.

8.1a ?- member(a,[c,a,b]). yes

8.1b ?- member(a,[d,o,g]). no


8.1c ?- member(one,[one,three,one,four]). yes

8.1d ?- member(X,[c,a,t]). X=c yes

8.1e ?- member(tom,[[jo,alan],[tom,anne]]). no

8.1f Complete the AND/OR tree below which represents the execution of the query:

Tree:



Trace:

- 1 1 Call: member(a,[b,r,e,a,d]) ?
- 2 2 Call: member(a,[r,e,a,d]) ?
- 3 3 Call: member(a,[e,a,d]) ?
- 4 4 Call: member(a,[a,d]) ?
- 4 4 Exit: member(a,[a,d]) ?
- 3 3 Exit: member(a,[e,a,d]) ?
- 2 2 Exit: member(a,[r,e,a,d]) ?
- 1 1 Exit: member(a,[b,r,e,a,d]) ?

yes

**Question 8.2** The predicate no\_cons/1 succeeds if all elements of the list represented by the one argument are vowels (as specified by vowel/1).

e.g. ?- no\_cons([a,e,i]).  
yes

```
? no_cons([a,b,c]).
no
```

no\_cons/1 is defined as:

```
1. no_cons([]).
2. no_cons([H|T]):-
 vowel(H),
 no_cons(T).
```

```
3. vowel(a).
4. vowel(e).
5. vowel(i).
6. vowel(o).
7. vowel(u).
```


For each of the following:

- specify whether the query submitted to Prolog succeeds or fails;
- if it succeeds, specify what is assigned to any variables.

8.2a ?- no\_cons([a,a,e,i]).                    yes

8.2b ?- no\_cons([A,e,e]).                    A=a    yes

8.2c Complete the AND/OR tree below which represents the execution of the query:



yes


**Question 8.3** The predicate `max/2` succeeds if the first argument is a list of numbers and second argument unifies with the maximum value in that list.

e.g. `?- max([2,4,6,8],M).`  
`M = 8`  
 yes

`max/2` is defined as:

0. `max([H|T],Max):-`  
`max(T,H,Max).`
1. `max([H|T],Temp,Max):-`  
`H>Temp,`  
`max(T,H,Max).`
2. `max([H|T],Temp,Max):-`  
`max(T,Temp,Max).`
3. `max([],Finalmax,Finalmax).`

8.3 Complete the AND/OR tree below which represents the execution of the query:



Ans = 4  
yes

**Question 8.4** The predicate prlist/1 is supposed to write out the elements of a list structure, regardless of the levels of embedding that are present in the list.

e.g. intended behaviour:

```

?- prlist([a,b,[c,d,e],f,[g]]).
abcdefg
yes

```

Instead, the predicate as defined below has the following behaviour:

```

?- prlist([a,b]).
ab[]
yes

?- prlist([a,b,[c,d],e]).
abcd[]e[]
yes

```

prlist/2 is defined as:

```

1. prlist([H|T]):-

```



```

 prlist(H),
 prlist(T).
2. prlist(X):-
 write(X).

```

Explain why this predicate produces this behaviour (instead of the intended behaviour), using an AND/OR tree or a trace in your explanation.

Tree:

```

 ?- prlist([a,b]).
 /---\
 1 / \
 prlist(a) prlist([b])
 / | /---\
 1 / |2 1 / \
 fails write(a) prlist(b) prlist([])
 / \ / \
 1 / \2 /1 \2
 fails write(b) fails write([])

```

ab []

The problem here is that whatever its value, the head and tail of the list on each recursion are prlisted, and when the head is no longer a list it is written. So the empty list also gets written. An extra clause, `prlist([])` is needed to prevent this.

**Question 8.5** The predicate `checkvowels` takes a list representing a word, checks each letter to see whether it is a vowel, and if it is it writes out the vowel.

```

checkvowels([H|T]):-
 vowel(H),
 write(H), nl,
 checkvowels(T).

checkvowels([H|T]):-
 checkvowels(T).

checkvowels([]).

vowel(a).
vowel(e).
vowel(i).
vowel(o).
vowel(u).

```

```
e.g. ?- checkvowels([c,a,t,i]).
 a
 i
 yes
```

8.5a Using this as a model, write a predicate `results/1` that takes a list of names, checks whether each is a pass (using a predicate that you also must define, `pass/1`) and writes out the names if they pass.

For example, the following query:

```
?- results([tom,bob,sue,jane]).
```

should give the output:

```
bob
sue
yes
```

Solution:

```
pass(bob).
pass(sue).
results([H|T]):-
 pass(H), write(H), nl,
 results(T).
results([H|T]):-
 results(T).

results([]).
```

8.5b Modify this program so that instead of 'writing out' the name of each person who passes it should output a list of them.

```
?- results2([tom,bob,sue,jane],Passlist).
Passlist = [bob,sue]
yes

results2([H|T],[H|P1]):-
 pass(H),
 results2(T,P1).
results2([H|T],P1):-
 results2(T,P1).
results2([],[]).
```

## 17.6 Chapter 10

**Question 10.1** The predicate `delete/3` succeeds if deleting the element represented by the first argument, from the list represented by the second argument, results in a list represented by the third argument.

e.g. `?- delete(a,[a,p,p,l,e],A).`  
`A=[p,p,l,e]`  
`yes`

`delete/3` is defined as:

1. `delete(E1,[E1|T],T).`
2. `delete(E1,[H|T],[H|NT]):-`  
`delete(E1,T,NT).`

For each of the following:

- specify whether the query submitted to Prolog succeeds or fails;
- if it succeeds, specify what is assigned to any variables.

10.1a `?- delete(X,[pat,john,paul],Ans).`    `X = pat`    `Ans = [john,paul]`    `yes`

10.1b `?- delete(A,L,[t,o,p]).`    `A = _A`    `L = [_A,t,o,p]`    `yes`  
or `L = [A,t,o,p]`

10.1c `?- delete(a,[c,a,b],Ans).`    `Ans = [c,b]`    `yes`

10.1d `?- delete(e,[d,o,g],P).`    `no`

10.1e `?- delete(e,[f,e,e,t],Ans).`    `Ans = [f,e,t]`    `yes`

**Question 10.2** The predicate `deleteall/3` succeeds if deleting all occurrences of the element represented by the first argument from the list represented by the second argument results in a list represented by the third argument.

e.g. `?- delete(p,[a,p,p,l,e],A).`    `A=[a,l,e]`    `yes`

`deleteall/3` is defined as:

1. `deleteall(E1,[],[]).`
2. `deleteall(E1,[E1|T],NT):-`  
`deleteall(E1,T,NT).`
3. `deleteall(E1,[H|T],[H|NT]):-`  
`deleteall(E1,T,NT).`


For each of the following:

- specify whether the query submitted to Prolog succeeds or fails;
- if it succeeds, specify what is assigned to any variables.

10.2a `?- deleteall(e,[f,e,e,t],Ans).`                      `Ans = [f,t] yes`


10.2b `?- deleteall(p,[d,o,g],X).`                              `X = [d,o,g] yes`

10.2c Complete the AND/OR tree below which represents the execution of the query:



Solution:

Tree:



Trace:

```

1 1 Call: deleteall(e,[f,e,e,t],_115) ?
2 2 Call: deleteall(e,[e,e,t],_288) ?
3 3 Call: deleteall(e,[e,t],_288) ?
4 4 Call: deleteall(e,[t],_288) ?
5 5 Call: deleteall(e,[],_626) ?
5 5 Exit: deleteall(e,[],[]) ?
4 4 Exit: deleteall(e,[t],[t]) ?
3 3 Exit: deleteall(e,[e,t],[t]) ?
2 2 Exit: deleteall(e,[e,e,t],[t]) ?
1 1 Exit: deleteall(e,[f,e,e,t],[f,t]) ?
Ans = [f,t] ?
yes

```

10.2d Suppose that the predicate deleteall/3 is defined incorrectly as:

```

1. delall(E,[],[]).
2. delall(E,[E|T],Y):-
 delall(E,T,Y).
3. delall(E,[H|T],Y):-
 delall(E,T,[H|Y]).

```

resulting in the behaviour:

```

i. ?- delall(e,[f,e,e,t],Ans).
no

```

However, the following query succeeds, as intended:

```

ii. ?- delall(a,[a,a,a],Ans).
Ans=[]
yes

```

Explain why the program does not give the intended answer to query i. using an AND/OR tree or a trace to illustrate your answer.

```

| ?- delall(e,[f,e,e,t],Ans).
+ 1 1 Call: delall(e,[f,e,e,t],_95) ?
+ 2 2 Call: delall(e,[e,e,t],[f|_95]) ?
+ 3 3 Call: delall(e,[e,t],[f|_95]) ?
+ 4 4 Call: delall(e,[t],[f|_95]) ?
+ 5 5 Call: delall(e,[],[t,f|_95]) ?
+ 5 5 Fail: delall(e,[],[t,f|_95]) ?
+ 4 4 Fail: delall(e,[t],[f|_95]) ?
+ 4 4 Call: delall(e,[t],[e,f|_95]) ?
+ 5 5 Call: delall(e,[],[t,e,f|_95]) ?
+ 5 5 Fail: delall(e,[],[t,e,f|_95]) ?
+ 4 4 Fail: delall(e,[t],[e,f|_95]) ?

```

```

+ 3 3 Fail: delall(e,[e,t],[f|_95]) ?
+ 3 3 Call: delall(e,[e,t],[e,f|_95]) ?
+ 4 4 Call: delall(e,[t],[e,f|_95]) ?
+ 5 5 Call: delall(e,[],[t,e,f|_95]) ?
+ 5 5 Fail: delall(e,[],[t,e,f|_95]) ?
+ 4 4 Fail: delall(e,[t],[e,f|_95]) ?
+ 4 4 Call: delall(e,[t],[e,e,f|_95]) ?
+ 5 5 Call: delall(e,[],[t,e,e,f|_95]) ?
+ 5 5 Fail: delall(e,[],[t,e,e,f|_95]) ?
+ 4 4 Fail: delall(e,[t],[e,e,f|_95]) ?
+ 3 3 Fail: delall(e,[e,t],[e,f|_95]) ?
+ 2 2 Fail: delall(e,[e,e,t],[f|_95]) ?
+ 1 1 Fail: delall(e,[f,e,e,t],_95) ?
no

```

```
{trace}
```

```

| ?- delall(a,[a,a,a],Ans).
+ 1 1 Call: delall(a,[a,a,a],_89) ?
+ 2 2 Call: delall(a,[a,a],_89) ?
+ 3 3 Call: delall(a,[a],_89) ?
+ 4 4 Call: delall(a,[],_89) ?
+ 4 4 Exit: delall(a,[],[]) ?
+ 3 3 Exit: delall(a,[a],[]) ?
+ 2 2 Exit: delall(a,[a,a],[]) ?
+ 1 1 Exit: delall(a,[a,a,a],[]) ?

```

```
Ans = [] ?
```

```
yes
```

Builds in the body rather than the head of the clause.

**Question 10.3** The predicate `repa1/4` succeeds if replacing all occurrences of the element represented by the first argument, by the element represented by the second argument, in the list represented by the third argument, results in a list represented by the fourth argument.

```

e.g. ?- repa1(p,b,[a,p,p,l,e],A).
 A=[a,b,b,l,e]
 yes

```

`repa1/4` is defined as:

```

1. repa1(E1,Rel,[],[]).
2. repa1(E1,Rel,[E1|T],[Rel|NT]):-
 repa1(E1,Rel,T,NT).
3. repa1(E1,Rel,[H|T],[H|NT]):-

```

repall(E1,Rel,T,NT).

For each of the following:

- specify whether the query submitted to Prolog succeeds or fails;
- if it succeeds, specify what is assigned to any variables.

10.3a ?- repall(e,a,[f,e,a,t],Ans).      Ans = [f,a,a,t] yes

10.3b ?- repall(P,r,[s,o,u,p],Ans).      Ans = [r,o,u,p] P = s    yes

**Question 10.4** The predicate `whowants/3` has three arguments. The first is intended to represent a type of food; the second a list of people; and the third another list, representing those people on the 2nd argument list who want the type of food specified in the 1st argument (where `want/2` is defined separately, with two arguments representing who wants what food).

e.g.      ?- whowants(beans,[jo,tom,ann],Who).  
            Who = [jo,ann]  
            yes

1. whowants(Food,[Name|Rest],[Name|Others]):-  
    wants(Name,Food),  
    whowants(Food,Rest,Others).
2. whowants(Food,[Name|Rest],Others):-  
    whowants(Food,Rest,Others).
3. whowants(Food,[],[]).
4. wants(jo,chips).
5. wants(jo,beans).
6. wants(jo,eggs).
7. wants(ann,beans).
8. wants(ann,bacon).
9. wants(tom,eggs).
10. wants(tom,chips).
11. wants(rick,bacon).

10.4a Give either the AND/OR tree or a trace which represents the execution of the query:

```
?- whowants(beans,[jo,tom,ann],Who).
 /\
 1/----\ Food/beans Name/jo Rest/[tom,ann] Who/[jo|Others]
 / \
```

```

wants(jo,beans) whowants(beans,[tom,ann],Others)
 |
 5| 1/ 2\ Name1/tom Rest1/[ann]
 | / \
succeeds wants(tom,beans) whowants(beans,[ann],Others)
 | /\
 fails 1/----\ Name2/ann Rest2/[] Others/[ann|Others1]
 / \
 wants(ann,beans) whowants(beans,[],Others1)
 | /\
 7| 1/ 2| 3\ Others1/[]
 | / | \
 succeeds fails fails succeeds
 Others1=[]

```

Who=[jo,ann]  
yes

```

| ?- trace, whowants(beans,[jo,tom,ann],Who).
+ 1 1 Call: whowants(beans,[jo,tom,ann],_75) ?
+ 2 2 Call: wants(jo,beans) ?
+ 2 2 Exit: wants(jo,beans) ?
+ 3 2 Call: whowants(beans,[tom,ann],_681) ?
+ 4 3 Call: wants(tom,beans) ?
+ 4 3 Fail: wants(tom,beans) ?
+ 4 3 Call: whowants(beans,[ann],_681) ?
+ 5 4 Call: wants(ann,beans) ?
+ 5 4 Exit: wants(ann,beans) ?
+ 6 4 Call: whowants(beans,[],_1438) ?
+ 6 4 Exit: whowants(beans,[],[]) ?
+ 4 3 Exit: whowants(beans,[ann],[ann]) ?
+ 3 2 Exit: whowants(beans,[tom,ann],[ann]) ?
+ 1 1 Exit: whowants(beans,[jo,tom,ann],[jo,ann]) ?
Who = [jo,ann] ?
yes

```

For each of the following:

- specify whether the query submitted to Prolog succeeds or fails;
- if it succeeds, specify what is assigned to any variables.

10.4b ?- whowants(chips,[ann,rick],Ans).            Ans = [] yes

10.4c ?- whowants(bacon,[tom,ann],A).            A = [ann].

10.4d ?- whowants(eggs,Diners,Ans).            fails loops



```
| ?- trace,whowants(eggs,Diners,Ans).
+ 1 1 Call: whowants(eggs,_53,_79) ?
+ 2 2 Call: wants(_677,eggs) ?
+ 2 2 Exit: wants(jo,eggs) ?
+ 3 2 Call: whowants(eggs,_678,_676) ?
+ 4 3 Call: wants(_1232,eggs) ?
+ 4 3 Exit: wants(jo,eggs) ?
+ 5 3 Call: whowants(eggs,_1233,_1231) ?
+ 6 4 Call: wants(_1787,eggs) ?
+ 6 4 Exit: wants(jo,eggs) ?
+ 7 4 Call: whowants(eggs,_1788,_1786) ?
+ 8 5 Call: wants(_2342,eggs) ?
+ 8 5 Exit: wants(jo,eggs) ?
.....
```

10.4e Using `whowants/3` as a model, write a predicate `overage/3` that takes an age limit and a list of names, checks the age of each person named (using a predicate `age/2`) and returns a list of names of those people who are over the age limit.

For example, the query below should give the output shown:

```
?- overage(18,[sally,alice,bill],Ans).
Ans=[alice,bill]
yes
where:
age(sally,15).
age(mark,26).
age(bill,20).
age(alice,37).
```

Solution:

```
overage(L,[H|T],[H|Y]):-
 age(H,X),
 X>L,
 overage(L,T,Y).
overage(L,[H|T],Y):-
 overage(L,T,Y).
overage(L,[],[]).
```

10.4f If the predicate `whowants/3` had been (incorrectly) defined as:

```
1. whowants(Food,[Name|Rest],[Name|Others]):-
 wants(Name,Food),
 whowants(Food,Rest,Others).
```

```
2. whowants(Food,[Name|Rest],Others):-
 whowants(Food,Rest,Others).
```

(i.e. no 3rd clause) predict the outcome of the query in 10.4c.

```
?- whowants(bacon,[tom,ann],A).
```

and explain why this is the case, using an AND/OR tree or a trace in your explanation.

```
| ?- trace, whowants(bacon,[tom,ann],A).
+ 1 1 Call: whowants(bacon,[tom,ann],_69) ?
+ 2 2 Call: wants(tom,bacon) ?
+ 2 2 Fail: wants(tom,bacon) ?
+ 2 2 Call: whowants(bacon,[ann],_69) ?
+ 3 3 Call: wants(ann,bacon) ?
+ 3 3 Exit: wants(ann,bacon) ?
+ 4 3 Call: whowants(bacon,[],_868) ?
+ 4 3 Fail: whowants(bacon,[],_868) ?
+ 3 3 Redo: wants(ann,bacon) ?
+ 3 3 Fail: wants(ann,bacon) ?
+ 3 3 Call: whowants(bacon,[],_69) ?
+ 3 3 Fail: whowants(bacon,[],_69) ?
+ 2 2 Fail: whowants(bacon,[ann],_69) ?
+ 1 1 Fail: whowants(bacon,[tom,ann],_69) ?
no
```

Fails because there is now no clause to match the empty list.

**Question 10.5** The predicate `deletefirst/3` is supposed to succeed if deleting the element represented by the first argument, from the list represented by the second argument, results in a list represented by the third argument, as `delete/3` defined as above (B.)

e.g. intended behaviour:

```
?- deletefirst(i,[b,i,b,s],A).
A=[b,b,s]
yes
```

Instead, it produces the following behaviour:

```
?- deletefirst(i,[b,i,b,s],A).
A=[s]
yes

?- deletefirst(a,[b,a,l,l],X).
no
```


deletefirst/3 is defined as:

1. deletefirst(E1,[E1|T],T).
2. deletefirst(E1,[H|T],NT):-  
deletefirst(E1,T,[H|NT]).

Explain why this predicate produces this behaviour (instead of the intended behaviour), using an AND/OR tree or a trace in your explanation.

Solution:

Tree:



Trace:

```
| ?- deletefirst(i,[b,i,b,s],A).
 1 1 Call: deletefirst(i,[b,i,b,s],_103) ?
 2 2 Call: deletefirst(i,[i,b,s],[b|_103]) ?
 2 2 Exit: deletefirst(i,[i,b,s],[b,s]) ?
 1 1 Exit: deletefirst(i,[b,i,b,s],[s]) ?
```

A = [s] ?        yes

Matches [b—X] from first match of [b—[i,b,s]] to [b,s] left after matching i to [i,b,s]. - so only works by coincidence of 2 occurrences here.

{trace}

```
| ?- deletefirst(a,[b,a,l,l],X).
 1 1 Call: deletefirst(a,[b,a,l,l],_103) ?
 2 2 Call: deletefirst(a,[a,l,l],[b|_103]) ?
 3 3 Call: deletefirst(a,[l,l],[a,b|_103]) ?
 4 4 Call: deletefirst(a,[l],[l,a,b|_103]) ?
 5 5 Call: deletefirst(a,[],[l,l,a,b|_103]) ?
 5 5 Fail: deletefirst(a,[],[l,l,a,b|_103]) ?
 4 4 Fail: deletefirst(a,[l],[l,a,b|_103]) ?
 3 3 Fail: deletefirst(a,[l,l],[a,b|_103]) ?
```

```
2 2 Fail: deletefirst(a,[a,l,l],[b|_103]) ?
1 1 Fail: deletefirst(a,[b,a,l,l],_103) ?
no
No match here.
```

The problem is that the head of the list that is supposed to be constructed is copied onto the front of the list that is called in the third argument of the recursive call (in the body of the clause), rather than in the query itself (in the head of the clause). So it gets built up as the program recurses, the program then fails when it cannot split the empty list. The variable represented by Ans does not get instantiated normally in this version, but with the call of:

```
?- deletefirst(i,[b,i,b,s],Ans)
```

co-incidentally at one point the program matches [b,s] to [b—Ans], causing Ans to be instantiated to [s].